
**********************************

COMMENTS

Welcome to the first issue of  the Excel Experts E-letter (or EEE), by David Hager. My current plan is to make EEE a semi-monthly publication. If you received this file as an attachment to an e-mail message sent on ~ 16Mar1999, that means you are among the 100 or so people I collected e-mail addresses for as a starting point to send EEE to the top Excel experts worldwide. I looked in newsgroups, forums, web pages, books, magazines and e-lists to find you, but I did not find everyone. Also, there are quite a few Excel experts out there that I have names for, but no e-mail addresses. So, if you know of someone whom you consider to be an Excel expert and they are not currently receiving EEE, ask them to send their name and e-mail address to me at dchager@compuserve.com.

My vision for EEE is for it to be based on meaty content. I have collected a lot of material over the last 5 years on Excel, and I will try to present what I feel is of interest to the majority of people on this list. In most cases, I will use tidbits that were gleaned from postings on Excel newsgroups, forums and e-lists, and I will attempt to give credit to the creator of the tip. If you object to the use of your name, let me know. On the flip side, if you have any new and great ideas, send them to me and I will include them in a future issue. I don't plan to issue any challenges to solve Excel problems in EEE, but it is likely that many of you have made some interesting discoveries during the course of your work that you have never had the chance to share with anyone. Let EEE be the vehicle for the presentation of that Excel gem!

**********************************

TOP EXCEL WEB SITES

Tied for the top spot:

www.j-walk.com

www.bmsLTD.co.uk

**********************************

WORKSHEET FORMULA TIPS

Created by Bob Umlas and David Hager

To define a global range name in Excel, go to Insert, Name, Define and, as an example, in the Names in Workbook box type "cellA1" and in the Refers To box type this formula: =OFFSET(!$A$1,,,,). Now, type =cellA1 in a cell on any worksheet in the workbook and it will return the value in A1 for that worksheet.

Created by David Hager

To add comma delimited values in a cell (such as 1,2,11,4 in cell A1) to the right of the cell containing the string, highlight cell B1 (for this case) and create the following defined name formula (called "csum"):

=EVALUATE(SUBSTITUTE(A1,",","+"))

Then, type =csum in B1 to obtain the result (18, in this case).

**********************************

POWER FORMULA TECHNIQUE

Created by Laurent Longre:

The problem - how to create an array of filtered items in a column list. The SUBTOTAL function allows you to operate on an array of this type with a limited number of worksheet functions, but it does not expose the array for formula manipulation. Laurent came up with an elegant soution to this problem, based on an obscure behavior of the OFFSET function. It turns out that when an array is used as the 2nd argument of OFFSET, such as

=OFFSET(Rge,ROW(Rge)-MIN(ROW(Rge)),,1)

an array of RANGES is returned. If the array is properly sized, as in this example, the OFFSET function return a separate single cell range for each cell in the original range(Rge). Thus, if this array of arrays is operated on by the SUBTOTAL function, each single cell range gets evaluated separately. So, the formula

=SUBTOTAL(3,OFFSET(Rge,ROW(Rge)-MIN(ROW(Rge)),,1))

evaluates as 1 for each cell that is visible and as 0 if the cell is not visible. The use of 3 as the 1st argument in SUBTOTAL counts the number of items in the visible range. Since there is only one item in each range, the answer can only be 0 or 1. Thus, this formula can be used as an array which indicates the rows in the list that are filtered and unfiltered. If you want to returns an array of items in the column list, then use:

=IF(SUBTOTAL(3,OFFSET(Rge,ROW(Rge)-MIN(ROW(Rge)),,1)),Rge,"")

**********************************

VBA CODE EXAMPLES 

Created by Rob Bovey:

Uses path as argument and it returns True if the path is empty or doesn't exist and False if the path contains files.

Function bIsEmpty(ByVal szPath As String) As Boolean

    Dim bReturn As Boolean

    Dim szTemp As String

    bReturn = True

    If Right$(szPath, 1) <> "\" Then szPath = szPath & "\"

    szTemp = Dir$(szPath & "*.*")

    If szTemp <> "" Then bReturn = False

    bIsEmpty = bReturn

End Function

Created by Jim Rech:

Removes a known directory including all of its files and any/all possible sub-directories of unknown quantity & name/s including their files.

Const FO_DELETE = &h3&

Const FOF_NOCONFIRMATION = &h10&

Private Type SHFILEOPSTRUCT

   hwnd As Long

   wFunc As Long

   pFrom As String

   pTo As String

   fFlags As Integer

   fAnyOperationsAborted As Long

   hNameMappings As Long

   lpszProgressTitle As String

End Type

Private Declare Sub CopyMemory Lib "KERNEL32" Alias "RtlMoveMemory" 

(hpvDest As Any, hpvSource As Any, ByVal cbCopy As Long)

Private Declare Function SHFileOperation Lib "Shell32.dll" Alias 

"SHFileOperationA" (lpFileOp As Any) As Long

Sub Test()

    ShellDelete "c:\aaa"

End Sub

Sub ShellDelete(SrcFile As String)

    Dim result As Long

    Dim lenFileop As Long

    Dim foBuf() As Integer

    Dim fileop As SHFILEOPSTRUCT

    lenFileop = LenB(fileop)

    ReDim foBuf(1 To lenFileop)

    With fileop

        .hwnd = 0

       .wFunc = FO_DELETE

       .pFrom = SrcFile & Chr(0) & Chr(0)

        .fFlags = FOF_NOCONFIRMATION

        .lpszProgressTitle = "" & Chr(0) & Chr(0)

    End With

    Call CopyMemory(foBuf(1), fileop, lenFileop)

    Call CopyMemory(foBuf(19), foBuf(21), 12)

    result = SHFileOperation(foBuf(1))

End Sub

Created by Bill Manville:

Checks whether a name exists in a collection. For example, If IsIn(ActiveWorkbook.Names, "ThisOne") Then ...

Function IsIn(oCollection As Object, stName As String)  As Boolean

  Dim O As Object

  On Error Goto NotIn

  Set O = oCollection(stName)

  IsIn = True
'succeeded in creating a pointer to the object so must be there

NotIn:

End Function

**********************************

POWER PROGRAMMING TECHNIQUE

Created by Stephen Bullen

The problem - you want to programatically obtain the name of the VBComponent that contains a specified procedure. Stephen's solution was to look for unique strings, since the VBIDE object model does not provide functionality for doing this directly.

Sub TestIt()

MsgBox fnThisVBComponent(ThisWorkbook, "This Unique String").Name & ", " & fnThisProcedureName(ThisWorkbook, "Another Unique String")

End Sub

Function fnThisVBComponent(oBk As Workbook, sUniqueString As String) As VBComponent

Dim oVBC As VBComponent

'Loop through the VBComponents in the given workbook's VBProject

For Each oVBC In oBk.VBProject.VBComponents

    'Using it's code module

    With oVBC.CodeModule

        'See if we can find the unique string

        If .Find(sUniqueString, 1, 1, .CountOfLines, 1000, True, True, False) Then

            'Found it, so return the VBComponent where it was found

            Set fnThisVBComponent = oVBC

            Exit For

        End If

    End With

Next

End Function

Function fnThisProcedureName(oBk As Workbook, sUniqueString As String) As String

Dim oVBC As VBComponent

Dim lStart As Long, sProcName As String, vaProcs As Variant, vProcType As Variant

'Specify the row number of where to start the find.  This is set by the Find method to give the (starting) line number where the text was found

lStart = 1

'Loop through the VBComponents in the given workbook's VBProject

For Each oVBC In oBk.VBProject.VBComponents

    'Using it's code module

    With oVBC.CodeModule

        'See if we can find the unique string

        If .Find(sUniqueString, lStart, 1, .CountOfLines, 1000, True, True, False) Then

            'We found it, so make an array of the available procedure types to check for

            vaProcs = Array(vbext_pk_Proc, vbext_pk_Get, vbext_pk_Let, vbext_pk_Set)

            'Loop through the procedure types

            For Each vProcType In vaProcs

                'Get the name of the procedure containing the line we found above

                sProcName = .ProcOfLine(lStart, CLng(vProcType))

                'Did we get a procedure name?

                If sProcName <> "" Then

                    'We did, so return it

                    fnThisProcedureName = sProcName

                    Exit For

                End If

            Next

            Exit For

        End If

    End With

Next

End Function

**********************************

EXCEL 2000 TIP

If you have had the opportunity to use beta versions of Excel 2000, then you probably realize the great potential of the web-interactive Excel file formats and their corresponding Office Web Components (OWC). Something you might not realize (or had a chance to play with yet) is that the OWC's can also be used with UserForms! It opens up a Pandora's Box of possibilities.

**********************************

Issue No.1 OF EEE (PUBLISHED 16MAR1999)

Next issue scheduled for 01APR1999.

BY David Hager

dchager@compuserve.com

**********************************

COMMENTS

Welcome to the second issue of the Excel Experts E-letter (or EEE), by David Hager. EEE is a semi-monthly publication. Feel free to distribute copies of EEE to your friends and colleagues. 

I am overwhelmed by the positive responses I received about the 1st issue of EEE and I appreciate your kind words. The mailing list for EEE is up above 170 now. I have had to look far and wide on the Internet to find you, and frankly it has taken more time to do this than writing this newsletter. I had some trouble with the stability of my mailing list on the 1st issue, and many people that I believed were on the list were not. If you would like a copy of the 1st issue, send me a note and I will be happy to pass it along. Many of you have indicated that you would like to contribute your ideas. I look forward to seeing them! 

**********************************

TOP EXCEL WEB SITES

Do you need an Excel formula?

http://home.gvi.net/~cpearson/excel.htm

**********************************

WORKSHEET FORMULA TIPS

Created by David Hager

In a lab a test is performed in triplicate. If 2 of the values are the same, those 2 values are averaged. Otherwise, all 3 of the values are averaged. An array formula that returns a result based on this protocol is: 

=AVERAGE(IF(SUM(COUNTIF(Rng,Rng))=3,Rng,IF(COUNTIF(Rng,Rng)>1,Rng,"")))

where Rng is a three cell range containing those values.

Created by Chip Pearson

The IF function has a limit of 7 nested arguments. Chip's formula circumvents that inherent limitation.

Define this formula as OneToSix:

=IF($A$4=1,11,IF($A$4=3,22,IF($A$4=5,33,IF($A$4=7,44,IF($A$4=9,55,IF($A$4=11,44,IF($A$4=13,55,IF($A$4=15,66,FALSE))))))))

and this formula as SevenToThirteen:

=IF($A$4=17,77,IF($A$4=19,88,IF($A$4=21,99,IF($A$4=23,100,IF($A$4=25,110,IF($A$4=27,120,IF($A$4=29,130,"NotFound")))))))

The combined formula looks like this:

=IF(OneToSix,OneToSix,SevenToThirteen)

**********************************

POWER FORMULA TECHNIQUE

Created by Shane Devenshire and David Hager

The problem - to change a link in a formula without changing the formula. This can be done with the INDIRECT function by creating a concatenated string with input from several worksheet cells which contain workbook (in A1) and worksheet (in B1) names.
=INDIRECT("'["&A1&"]"&B1&"'!A1")

Unfortunately, this type of formula will only work if the referenced workbook is open. Shane came up with part of the solution to this problem when he discovered that the INDEX function can return a linked cell value from a hard-coded link range. For example, if you define a range as "ref1", where the linked range formula is:

=[Book1.xls]Sheet1!$1:$65536

(A smaller range starting at A1 can also by used.)

then you can use the formula:

=IF(ISERR(INDEX(ref1,ROW(),COLUMN())),"",INDEX(ref1,ROW(),COLUMN()))

in any cell and the returned value will be from the same cell in Book1.xls on Sheet1. Then, variable links to this formula can be made by changing the link range as referred to in a named formula. This formula is of the form:

=CHOOSE(Sheet2!$A$1-29*INT((Sheet2!$A$1-1)/29),ref1,ref2,...,ref29)

where Sheet2!$A$1 is an input cell for values from 1 to n which represent a particular link stored as a defined name. As you are probably aware, the CHOOSE can only accept 29 arguments. However, there is a workaround for this limitation, and the formula in the 1st argument is part of that process. It converts the value in Sheet2!$A$1 into a number between 1 and 29. Then, if you define the preceding formula as oref1 (and other similar formulas as oref(n)), you can use the following master formula:

=CHOOSE(INT((Sheet2!$A$1-1)/29)+1,oref1,oref2,...,oref29)

Now, if you give this formula a defined name (say mref), then the resulting "omnireference" can be used in place of ref1 in Shane's formula to produce an "omnilink" that is capable of returning values from 29 x 29 (841) different links. This formula is the one that is finally entered in a worksheet cell.

=mref

Note: This technique works great as long as the linked files are not moved, renamed or deleted.

**********************************

VBA CODE EXAMPLES 

Created by Bill Manville

To create a blinking cell:

If you define a new Style (Format / Style / Flash/ Add ) and apply that style to the cells you want to flash, paste the following code into a module sheet and run the procedure Flash from Auto-Open (if desired) you will get the text flashing alternately white and red.

Dim NextTime As Date

Sub Flash()

  NextTime = Now + TimeValue("00:00:01")

  With ActiveWorkbook.Styles("Flash").Font

    If .ColorIndex = 2 Then .ColorIndex = 3 Else .ColorIndex = 2

  End With

  Application.OnTime NextTime, "Flash"

End Sub

Sub StopIt()

  Application.OnTime NextTime, "Flash", schedule:=False

  ActiveWorkbook.Styles("Flash").Font.ColorIndex = xlAutomatic

End Sub

Created by Myrna Larson 

An UDF that returns values "between" the points in the lookup table.

 Function InterpolateVLOOKUP(x As Single, Table As Range, YCol As Integer) 

   Dim TableRow As Integer, Temp As Variant

   Dim x0 As Double, x1 As Double, y0 As Double, y1 As Double

   Dim d As Double

   On Error Resume Next

   Temp = Application.WorksheetFunction.Match(x, Table.Resize(, 1), 1) 

   If IsError(Temp) Then

     InterpolateVLOOKUP = CVErr(Temp)

   Else

     TableRow = CInt(Temp)

     x0 = Table(TableRow, 1)

     y0 = Table(TableRow, YCol)

     If x = x0 Then

       InterpolateVLOOKUP = y0

     Else

       x1 = Table(TableRow + 1, 1)

       y1 = Table(TableRow + 1, YCol)

       InterpolateVLOOKUP = (x - x0) / (x1 - x0) * (y1 - y0) + y0

     End If

   End If

 End Function  

**********************************

POWER PROGRAMMING TECHNIQUE

By Laurent Longre

There is a way for an auto-start macro to read the arguments on the command line (with Excel 97).
Assume that you want to read the command line arguments from an Auto_open sub in the workbook "c:\temp\test.xls" opened by a batch file (or by a Win95 shortcut).

1. Your command line should look like this one: start excel c:\temp\test /e/param1/param2/.../paramN

i.e. : after excel.exe, the name of the workbook containing the Auto_open, then the switch /e **immediately** followed by your own arguments. These arguments should be separated by "/" and form a continuous string without spaces.

For instance, if you want to pass the arguments "c:\temp\file1.dbf", "all" and "exclusive" to Excel, your command-line should look like: start excel c:\temp\test /e/c:\temp\file1.dbf/all/exclusive

2. In Test.xls, use the API function GetCommandLine (alias GetCommandLineA in Win95) to get the contents of this command-line string.

You should then parse the string returned by GetCommandLineA, search for the separators "/" and store each argument in an array. Here is an example of a such Auto_open sub:

Option Base 1

Declare Function GetCommandLineA Lib "Kernel32" () As String

Sub Auto_open()

  Dim CmdLine As String  'command-line string

  Dim Args() As String  'array for storing the parameters

  Dim ArgCount As Integer  'number of parameters

  Dim Pos1 As Integer, Pos2 As Integer

  CmdLine = GetCommandLineA  'get the cmd-line string

  On Error Resume Next  'for the worksheet-function "Search"

  Pos1 = WorksheetFunction.Search("/", CmdLine, 1) + 1  'search "/e"

  Pos1 = WorksheetFunction.Search("/", CmdLine, Pos1) + 1  '1st parameter
  Do While Err = 0

    Pos2 = WorksheetFunction.Search("/", CmdLine, Pos1)

    ArgCount = ArgCount + 1

    ReDim Preserve Args(ArgCount)

    Args(ArgCount) = Mid(CmdLine, Pos1, IIf(Err, Len(CmdLine), Pos2) - Pos1)
    MsgBox "Argument " & ArgCount & " : " & Args(ArgCount)

    Pos1 = Pos2 + 1

  Loop

End Sub

If you use the command-line above, this Auto_open sub will automatically store the three arguments ("c:\temp\file1.dbf", "all" and "exclusive") in the Args() array and display them.

Again, be sure that you don't insert any space between /e and each argument in the command-line, otherwise it could fail (Excel can believe that these "pseudo-arguments" are the names of workbooks to open at startup...).

**********************************

EXCEL 2000 TIP

An intriguing property was added to the Range object in Excel 2000. It is the ID property. In the normal scheme of things, it assigns a string to a worksheet cell, which is used in a HTML tag when the worksheet is saved as a web page. If the worksheet is saved in a normal manner, the ID does not appear to be persistent. However, if ID's of cells are set when a workbook is opened, they can be used in some interesting ways. As an example, consider the following:

Sub Auto_Open()

    Sheets(1).Range("A1").ID = "Test"

End Sub

'in Sheet1 module

Private Sub Worksheet_SelectionChange(ByVal Target As Range)

    If ActiveCell.Address <> Target.Address Then Exit Sub

    If Target.ID <> "" Then MsgBox Target.ID

End Sub

In this case, if A1 is selected, the MsgBox dialog will appear with the text "Test". Thus, this provides a method of creating cell comments that do not have to be stored with the cells. Further, if the assignment of ID strings to cells is criteria-based rather than address-based, then this provides a dynamic method of documenting cells of varying properties. 

Another use of the ID property would be to apply the Collection class for the Tag property developed by Ken Getz and Mike Gilbert (shown in the Nov '98 issue of MOD magazine, p.36) to this system. I leave that as an exercise to the reader.

**********************************

DID YOU KNOW?...

that you can create a hyperlink from an object to a VBA procedure. In Excel 97, make a text box on a worksheet and right-click its edge. Then, select Insert, Hyperlink from the menu and type the name of the procedure in the Named Location in File box.
NOTE: The manual setting of a procedure name as a hyperlink subaddress apparently cannot be done in Excel 2000. Not only does the dialog box not allow that option, it does not allow the use of names that do not yet exist, unlike Excel 97. In my opinion, there was no reason to add an extra validation step to see if the name exists, since the option to add the name at a later time increases the flexibility of this feature. However, you can still set/change the SubAddress of the hyperlink programatically in Excel 2000 for VBA procedures.

**********************************

Issue No.2 OF EEE (PUBLISHED 01Apr1999)

Next issue scheduled for 16APR1999.

BY David Hager

dchager@compuserve.com

**********************************

COMMENTS

Welcome to the third issue of the Excel Experts E-letter (or EEE), by David Hager. EEE is a semi-monthly publication. Feel free to distribute copies of EEE to your friends and colleagues. In honor of this being the 3rd issue of EEE, I have decided to dedicate this issue to 3D solutions. As you might know, the 3D capabilities of Excel are lacking in several areas. However, this just provides fertile ground for growing those workarounds.

**********************************

TOP EXCEL WEB SITES

For the best in 3D add-in functionality, check out:

http://perso.wanadoo.fr/longre/excel/pages/Downloads.htm

Also, check out the freeware file Make Excel 3D in the CompuServe Excel forum library for a comprehensive set of 3D custom functions.

**********************************

WORKSHEET FORMULA TIPS

Excel allows you to make 3D formulas based on the following syntax:

Sheet1:Sheet4!A2:B5

However, the functions that can actually use that syntax are limited (although not as limited as the Excel documentation would lead you to believe). I put together the following list of functions that represent most, but probably not all, of those that are 3D-enabled:

AVERAGE, AVERAGEA, COUNT, COUNTA, MAX, MAXA, MIN, MINA, PRODUCT, STDEV, STDEVA, STDEVP, STDEVPA, VAR, VARA, VARP, VARPA, SMALL, LARGE, RANK, MEDIAN, PERCENTILE, QUARTILE, TRIMMEAN, SKEW, AND, OR, AVEDEV, DEVSQ, SUMSQ

BTW, If you create a defined name for this type of 3D range, be sure to group all of the sheets in that range if you decide to insert any rows into the area bounded by that range (if you want your 3D range to reflect the insertion of that row).

Created by David Hager

To make a conditional format based on the value in the previous worksheet, create the following defined name formulas.

GlobRef as:

=INDIRECT("rc",FALSE)

which gives the value from the cell it is used in. PrevShtValue as:

=INDIRECT(INDEX(GET.WORKBOOK(1),GET.DOCUMENT(87)-1)&"!"&ADDRESS(ROW(),COLUMN()))

which gives the value from the cell of the same address in the previous sheet. Then, combine these in yet another defined name formula.

GTPSV (this cell value is greater than previous sheet value) as:

=GlobRef>PrevShtValue 

which is used as the conditional formatting formula (in Excel 97 and later versions).

**********************************

POWER FORMULA TECHNIQUES

Created by David Hager

The problem - to make a 3D formula that adjusts in a z-relative manner when it is filled across worksheets. Two separate solutions to this problem that use a similar methodology are shown below. 

Making a 3D Running Total -

This example uses information entered in column A, with the 3D Running Total formula in column B. 

Define shtPos as:

This formula returns the sheet position of the active sheet as an integer.

=GET.DOCUMENT(87)

Define wsNames as:

This formula returns an array of sheet names in the active workbook.

=RIGHT(GET.WORKBOOK(1),LEN(GET.WORKBOOK(1))-FIND("]",GET.WORKBOOK(1)))

Define shtName as:

This formula returns the active sheet name.

=INDEX(wsNames,shtPos)

Note: The string returned from GET.DOCUMENT(76) could have been modified to obtain the active sheet name, but the method shown above was used since the modified array of sheet names was available (and needed for another 3D example).

Define RunningTotal as:

=EVALUATE("SUM(Sheet1:"&shtName&"!A"&ROW()&")")+NOW()*0 

This formula returns the sum for the cell in column A for each worksheet from Sheet1 to the worksheet where the formula resides. For example, if the active sheet was Sheet4 and this formula was in B2, this formula (entered as =RunningTotal) would be the equivalent of the Excel formula =SUM(Sheet1:Sheet4!A2). Since this formula incorporates xlm macro functions, it must be forced to recalculate. This is done by using the term NOW()*0, since NOW() is a volatile function.

Making a 3D Moving Average -

Define shtNamem2 as:

=INDEX(wsNames,shtPos-2)

This formula returns the sheet name of the worksheet 2 tabs to the left of the active sheet.

Define MovingAverage as:

=EVALUATE("AVERAGE("&shtNamem2&":"&shtName&"!A"&ROW()&")")+NOW()*0

This formula returns the average of the values in the sheet 2 tabs to the left of the active sheet to the active sheet. For example, if the active sheet was Sheet4 and this formula (entered as =MovingAverage) was in B2, this formula would be the equivalent of the Excel formula =AVERAGE(Sheet2:Sheet4!A2).

Created by Laurent Longre

The problem - to make a 3D worksheet array formula. What this means is to create an array representing a z-range (a range across worksheets) that evaluates in the formula bar as an array. The 3D range used in Excel, i.e.

Sheet1:Sheet4!A2:B5

does not behave that way. I suspect that nearly everyone on the EEE list has tried to do this and found that it was not possible. However, Laurent found that it was possible, given some advanced formula tricks. The INDIRECT function can return a 3D reference if it is operated on by the N function. An illustration of this type of formula is shown below.

3D Diagonal Formula - 

=SUM(N(INDIRECT("Sheet"&{1,2,3}&"!"&ADDRESS({1,2,3},{1,2,3}))))

returns the sum of Sheet1!A1, Sheet2!B2 and Sheet3!C3. How does it work?

"Sheet"&{1,2,3}&"!"&ADDRESS({1,2,3},{1,2,3}) evaluates to the array of strings.

{"Sheet1!$A$1","Sheet2!$B$2","Sheet3!$C$3"}

When the INDIRECT function operates on this array, the expected array of values appear (by highlighting in the formula bar and pressing F9), but for some reason this array cannot be used by Excel functions. The use of the N function creates an array that can be used, so that the SUM function returns the desired result.

**********************************

POWER PROGRAMMING TECHNIQUES

By Myrna Larson and David Hager

Presented below are 3 UDF's (SumProduct3D, SumIf3D, CountIf3D) that provide an useful method of returning a variety of information from 3D ranges. Each of these functions use a 3D range argument (written as per the usual Excel protocol) as a string. This string is processed by the Parse3DRange function, which returns sheet positions and the range argument in variables that are used by these functions.

 Function SumProduct3D(Range3D As String, Range_B As Range) As Variant

    Dim sRangeA As String, sRangeB As String, Sheet1 As Integer

    Dim Sheet2 As Integer, Sum As Double, n As Integer

    Application.Volatile

    If Parse3DRange(Application.Caller.Parent.Parent.Name, Range3D, Sheet1, Sheet2, sRangeA) = False Then

      SumProduct3D = CVErr(xlErrRef)

      Exit Function

    End If

    sRangeB = Range_B.Address

    Sum = 0

    For n = Sheet1 To Sheet2

      With Worksheets(n)

        Sum = Sum + Application.WorksheetFunction.SumProduct(.Range(sRangeA), .Range(sRangeB))

      End With

    Next

    SumProduct3D = Sum

  End Function

  Function SumIf3D(Range3D As String, Criteria As String, Optional Sum_Range As Variant) As Variant

    Dim sTestRange As String, sSumRange As String, Sheet1 As Integer

    Dim Sheet2 As Integer, n As Integer, Sum As Double

    Application.Volatile

    If Parse3DRange(Application.Caller.Parent.Parent.Name, Range3D, Sheet1, Sheet2, sTestRange) = False Then

      SumIf3D = CVErr(xlErrRef)

    End If

    If IsMissing(Sum_Range) Then

      sSumRange = sTestRange

    Else

      sSumRange = Sum_Range.Address

    End If

    Sum = 0

    For n = Sheet1 To Sheet2

      With Worksheets(n)

        Sum = Sum + Application.WorksheetFunction.SumIf(.Range(sTestRange), Criteria, .Range(sSumRange))

      End With

    Next n

    SumIf3D = Sum

  End Function

  Function CountIf3D(Range3D As String, Criteria As String) As Variant

    Dim Sheet1 As Integer, Sheet2 As Integer, sTestRange As String

    Dim n As Integer, Count As Long

    Application.Volatile

    If Parse3DRange(Application.Caller.Parent.Parent.Name, Range3D, Sheet1, Sheet2, sTestRange) = False Then

      CountIf3D = CVErr(xlErrRef)

      Exit Function

    End If

    Count = 0

    For n = Sheet1 To Sheet2

        With Worksheets(n)

          Count = Count + Application.WorksheetFunction.CountIf(.Range(sTestRange), Criteria)

        End With

    Next n

    CountIf3D = Count

  End Function

  Function Parse3DRange(sBook As String, SheetsAndRange As String, FirstSheet As Integer, LastSheet As Integer, sRange As String) As Boolean

    Dim sTemp As String, i As Integer, Sheet1 As String, Sheet2 As String

    Parse3DRange = False

    On Error GoTo Parse3DRangeError

    sTemp = SheetsAndRange

    i = InStr(sTemp, "!")

    If i = 0 Then Exit Function

    'next line will generate an error if range is invalid

    'if it's OK, it will be converted to absolute form

    sRange = Range(Mid$(sTemp, i + 1)).Address

    sTemp = Left$(sTemp, i - 1)

    i = InStr(sTemp, ":")

    Sheet2 = Trim(Mid$(sTemp, i + 1))

    If i > 0 Then

      Sheet1 = Trim(Left$(sTemp, i - 1))

    Else

      Sheet1 = Sheet2

    End If

    'next lines will generate errors if sheet names are invalid

    With Workbooks(sBook)

    FirstSheet = .Worksheets(Sheet1).Index

    LastSheet = .Worksheets(Sheet2).Index

    'swap if out of order

    If FirstSheet > LastSheet Then

      i = FirstSheet

      FirstSheet = LastSheet

      LastSheet = i

    End If

    i = .Worksheets.Count

    If FirstSheet >= 1 And LastSheet <= i Then Parse3DRange = True

    End With

Parse3DRangeError:

    On Error GoTo 0

    Exit Function

  End Function  'Parse3DRange

**********************************

EXCEL 2000 INFORMATION

Although the ability to manipulate data with worksheet formulas did not change in Excel 2000, the other data handling features in Excel 2000 more than made up for this. In fact, IMHO, it is now ready to take on high-level data intensive corporate projects that previously could only be accomplished with a variety of tools. Among these are interactive web pages, Office Web Components, Pivot Tables and Charts, web queries, a greatly enhanced QueryTable object model, and OLAP cube technology. I plan to provide examples of each of these in upcoming issues (with a little help from the readership, I hope).

**********************************

Issue No.3 OF EEE (PUBLISHED 15/16Apr1999)

Next issue scheduled for 01MAY1999.

BY David Hager

dchager@compuserve.com

**********************************

COMMENTS

Welcome to the fourth issue of the Excel Experts E-letter (or EEE), by David Hager. EEE is a semi-monthly publication. Feel free to distribute copies of EEE to your friends and colleagues. 

Again, I have selected a theme for this issue. This time it is dealing with unique items in a dataset.

Remember that EEE is devoted to sharing ideas across the Excel developer community, so if you have some great Excel technique that is not well-known, send it to me and I will include it in a future issue.

For back issues of EEE, check out:

http://www.j-walk.com/ss/excel/eee/index.htm

**********************************

TOP EXCEL WEB SITES

There is a wealth of information at the following site.

http://www.baarns.com/IE4/index_devonly.asp

However, new material has not been added in quite some time.

**********************************

WORKSHEET FORMULA TIPS

Created by David Hager

This array formula returns the number of unique items in a worksheet range.

=SUM(1/COUNTIF(Rng,Rng))

However, if Rng contains blank cells, this formula returns an error. In this case, use this modified version of the formula.

=SUM(COUNTIF(Rng,Rng)/IF(NOT(COUNTIF(Rng,Rng)),1,COUNTIF(Rng,Rng))^2)

Created by David Hager

This array formula returns the Nth largest unique value in a column range.

=LARGE(IF(MATCH(Rng,Rng,0)=ROW(Rng)-MIN(ROW(Rng))+1,Rng,""),N)

Created by David Hager

To apply data validation to a column which allows only unique items to be entered, highlight that column and select (in Excel 97 and later versions) Data, Validation from the menu. Choose the custom option and enter the following formula (for column A):

=COUNTIF($A$1:A1,A1)=1

Created by Laurent Longre

This formula counts the number of unique items a column range, only if the cells in the lookup range contain the specified string. 

=SUM(N(FREQUENCY(IF(lookupRange="specifStr",MATCH(colRange,colRange,0)),MATCH(colRange,colRange,0))>0))

**********************************

POWER FORMULA TECHNIQUE

Created by David Hager, Bob Umlas and Laurent Longre

The problem - to create an array containing only the unique items from an expanding column list. In other words, if items are typed down column A, what is the formula that will return the unique items? The following example further illustrates the problem.

ColA

a

b

1

b

3

In this case, the array should be {"a";"b";1;3}. Then, if additional values are added:

a

b

1

b

3

c

1

d

a

b

the array should be {"a";"b";1;3;"c";"d"}. The answer to this problem has eluded me for years, but with recent input from Bob and Laurent, I have successfully constructed a solution to this problem. The formula is somewhat long, so it is necessary to define parts of the formula to simplify the final form.

Define TheList as:

=OFFSET(Sheet1!$A$1,,,COUNTA(Sheet1!$A:$A),)

This formula creates the expanding range for the items as they are entered into column A.

Define sArray as:

=SMALL(IF(MATCH(TheList,TheList,0)=ROW(TheList),ROW(TheList),""),ROW(INDIRECT("1:"&SUM(N(MATCH(TheList,TheList,0)=ROW(TheList))))))-1

This formula contains several important elements that require explanation. The formula =IF(MATCH(TheList,TheList,0)=ROW(TheList),ROW(TheList),"") returns an array of positions for the unique items that is the same size as the TheList array, where the duplicates items are now represented by empty strings. The formula ROW(INDIRECT("1:"&SUM(N(MATCH(TheList,TheList,0)=ROW(TheList))))) returns an array of numbers from 1 to n, where n is the number of unique items in the list, as calculated by the formula SUM(N(MATCH(TheList,TheList,0)=ROW(TheList))). What is desired is an array that contains the unique positions with no empty strings. This is accomplished by the use of the SMALL function which, along with the LARGE function, is unique among Excel functions in its ability to create different sized arrays than the array used in the 1st argument if the 2nd argument is also an array. The -1 is used to adjust the item positions for use in the formula shown below.

Define TheUniqueArray as:

=IF(T(OFFSET(TheList,sArray,,1))="",N(OFFSET(TheList,sArray,,1)),T(OFFSET(TheList,sArray,,1)))

The formula OFFSET(TheList,sArray,,1) is an array of single element arrays, as explained in detail in the 1st issue of EEE. It can be converted into a normal array by using the N or T functions. Both N and T are used here since TheList can contain either text or numeric items. 

WARNING: Although this technique works, it must not be applied to situations where there are large arrays (>1000 items). Excel's calculation engine runs much too slowly on this type of formula. For example, it calculates ~100 times slower than the programming techniques for counting unique items that are shown below for an array of 5000 items.

**********************************

VBA CODE EXAMPLES 

Here are four examples of counting unique values in a list. Each of these examples creates an array of the unique items, so they can be modified to use those arrays for a purpose other than just counting the unique items.

Created by David Hager 

Sub cMethodAdvFilter()

    CountUniqueByAdvFilter Selection.Address

End Sub

Sub CountUniqueByAdvFilter(mRange As String)

    Dim TheRange As String

    Application.ScreenUpdating = False

    TheRange = "'[" & ActiveWorkbook.Name & "]" & ActiveSheet.Name & "'!" & mRange
    Workbooks.Add    

    Range(TheRange).AdvancedFilter Action:=xlFilterCopy, CopyToRange:=Range("A1"), Unique:=True

    MsgBox Application.WorksheetFunction.CountA(Range("A:A"))

    ActiveWorkbook.Close False
    Application.ScreenUpdating = True

End Sub

Created by Keyuan Jiang 

    Sub cMethodDAO()

    Dim strDBFullName As String

    Dim dbData As Database, rstWork As Recordset, strSQL As String

    strDBFullName = ThisWorkbook.Path & "\" & ThisWorkbook.Name

    strSQL = "select distinct [your_field] from dataarea"

    'Appropriate driver needed for this statement

    Set dbData = OpenDatabase(strDBFullName, False, True, "Excel8.0;HDR=YES;")

    Set rstWork = dbData.OpenRecordset(strSQL)

    rstWork.MoveLast 

    MsgBox rstWork.RecordCount

    Set rstWork = Nothing    

    Set dbData = Nothing

    End Sub

where [your_field] is the header of the column you are interested in and the dataarea is a named area that contains all data in question (could be the single column you are interested in).

By David Hager

Sub CountUniqueByPivotTable()

    On Error GoTo uOut

    Application.ScreenUpdating = False

    Application.DisplayAlerts = False

    TheHeader = ActiveCell.Value

    ActiveSheet.PivotTableWizard SourceType:=xlDatabase, SourceData:=ActiveSheet.Name & "!" & _

    Selection.Address, TableDestination:="", TableName:="uPivotTable"

    ActiveSheet.PivotTables("uPivotTable").AddFields RowFields:=TheHeader

    ActiveSheet.PivotTables("uPivotTable").PivotFields(TheHeader).Orientation = xlDataField

    MsgBox Application.WorksheetFunction.CountA(Range("A:A")) - 3

    ActiveSheet.Delete

uOut:

    Application.ScreenUpdating = True

    Application.DisplayAlerts = True

End Sub

By John Walkenbach 

Sub cMethodByCollection()

    CountUniqueByCollection Selection.Address

End Sub

Sub CountUniqueByCollection(AllCells As String)

    Dim NoDupes As New Collection

    On Error Resume Next

    For Each Cell In Range(AllCells)

        NoDupes.Add Cell.Value, CStr(Cell.Value)


'Note: the 2nd argument (key) for the Add method must be a string

    Next Cell

    MsgBox NoDupes.Count

    On Error GoTo 0

End Sub

Although not tested extensively, it appears that the procedure that uses the Collection object produces the fastest result. 

**********************************

EXCEL 2000 TIP

Under Tools, Options, View there is a checkbox entitled "Windows in Taskbar". When it is checked, every file/window that is open in Excel has its own Taskbar button. Uncheck that box if you do not want this feature. Under View, Toolbars, Customize, Options there is a checkbox entitled "Menus show recently used commands first". When it is checked, menu items change their position based how often they are used. Uncheck that box if you do not want this feature.

**********************************

DID YOU KNOW?...

that the Office Spreadsheet Component can have up to 676 columns.

**********************************

Issue No.4 OF EEE (PUBLISHED 30Apr1999)

Next issue scheduled for 16MAY1999.

BY David Hager

dchager@compuserve.com

**********************************

COMMENTS

Welcome to the 5th issue of the Excel Experts E-letter (or EEE), by David Hager. EEE is a semi-monthly publication. Feel free to distribute copies of EEE to your friends and colleagues. Starting with issue #6, EEE will no longer be directly mailed. New issues are normally available on the 1st and 16th of each month.

It is becoming difficult to continue to credit the originator of the techniques I present in EEE, but I will continue to assign credit when I know the source. Please continue to submit your best tips, techniques and procedures so that all can benefit from your creations.

**********************************

TOP EXCEL WEB SITES

Visit Dave Steppan's web page at:

http://www.geocities.com/SiliconValley/Network/1030/ExcelTop.html

for some great tips and downloadable files.

**********************************

WORKSHEET FORMULA TIPS

Created by David Hager

Here is an array formula that will return TRUE if all of the characters in a string (in A1 in this example) are unique and return FALSE if not.

=SUM(N(MATCH(MID(A1,ROW(INDIRECT("1:"&LEN(A1))),1),MID(A1,ROW(INDIRECT("1:"&LEN(A1))),1),0)=ROW(INDIRECT("1:"&LEN(A1)))))=LEN(A1)

Created by David Hager

This array formula reverses the digits in a number.

=SUM(VALUE(MID(A1,ROW(INDIRECT("1:"&LEN(A1))),1))*10^(ROW(INDIRECT("1:"&LEN(A1)))-1))

Created by Laurent Longre

This formula reverses the characters in a string. Note that the CALL function is disabled by the Microsoft Excel CALL patch.

Under Windows, text length < 256 characters :

=CALL("Msvcrt40","_strrev","1F","String to reverse")

**********************************

POWER FORMULA TECHNIQUE

Created by David Hager and "unknown"

The problem: to create an inverted range. The following array formula does this, but in a very strange way. The example uses the information shown below.
A1:D1 = {1,2,3,4}

A5:D5 = {5,6,7,5}

=SUM({1,2,3,4}*{5,7,6,5}) = 57

Note that the 2nd array has been inverted. This formula returns an answer of 57, which is (1*5)+(2*7)+(3*6)+(4*5). The following formula also returns this result, but only when entered in a certain way. If you enter this formula in a single cell, it returns an answer of 50. However, if the same formula is array-entered in two cells, each cell will return the corrrect answer. The reason for this behavior is not known.

=SUM(A1:D1*INDEX(A5:D5,5-COLUMN(A5:D5)))

A formula that returns an inverted column range is shown below. The range being inverted in this example is iRng. 

=IF(T(OFFSET(iRng,ROWS(iRng)-ROW(OFFSET($A$1,,,ROWS(iRng),)),,,))="",N(OFFSET(iRng,ROWS(iRng)-ROW(OFFSET($A$1,,,ROWS(iRng),)),,,)),T(OFFSET(iRng,ROWS(iRng)-ROW(OFFSET($A$1,,,ROWS(iRng),)),,,)))

The key to this formula is the array of inverted cell positions created by

ROWS(iRng)-ROW(OFFSET($A$1,,,ROWS(iRng),)).

**********************************

VBA CODE EXAMPLES 

This procedure prevents the user from using File, Save As.

Private Sub Workbook_BeforeSave(ByVal SaveAsUI As Boolean, Cancel As Boolean)

    If SaveAsUI Then Cancel = True

End Sub

By Jim Rech

This procedure creates a high performance timer.

Declare Function QueryPerformanceFrequency Lib "kernel32"(lpFrequency As LARGE_INTEGER) As Long

Declare Function QueryPerformanceCounter Lib "kernel32"(lpPerformanceCount As LARGE_INTEGER) As Long

Type LARGE_INTEGER

    lowpart As Long

    highpart As Long

End Type

Sub TestHighResolutionTimer()

    Dim FirstCount As LARGE_INTEGER

    Dim SecondCount As LARGE_INTEGER

    Dim TimerOverhead As Long, Counter As Long

    QueryPerformanceCounter FirstCount

    QueryPerformanceCounter SecondCount

    TimerOverhead = SecondCount.lowpart - FirstCount.lowpart

    QueryPerformanceCounter FirstCount

    ''Procedure to time

    For Counter = 1 To 10000000

    Next

    ''End procedure to time

    QueryPerformanceCounter SecondCount

    MsgBox "Timer counts: " & Format(SecondCount.lowpart - FirstCount.lowpart - TimerOverhead, "#,##0")

End Sub

Sub GetHighResolutionTimerFrequency()

    Dim Freq As LARGE_INTEGER

    If QueryPerformanceFrequency(Freq) = 0 Then

        MsgBox "Your computer does not support the high performance timer"

    Else

        MsgBox "Your computer's high resolution timer frequency is " & Format(Freq.lowpart, "#,##0") & " counts per second"

    End If

End Sub

By Bill Manville

This procedure finds a string (from a Textbox) across worksheets in a workbook.

Sub gFindIt()

Dim strWhat As String

Dim WS As Worksheet

Dim R As Range

strWhat = txtSearchFor.Text

If strWhat = "" Then Exit Sub

For Each WS In ActiveWorkbook.Worksheets

   Set R = WS.Cells.Find(What:=strWhat, After:=WS.Range("A1"), LookIn:=xlFormulas, LookAt :=xlPart, SearchOrder:=xlByRows, SearchDirection:=xlNext, MatchCase:= False)

   If Not R Is Nothing Then

    Application.Goto R, Scroll:=True

    Exit For

   End If

Next

End Sub

**********************************

POWER PROGRAMMING TIPS

By Bill Manville

Is there any way to stop the links update box from apperaring when opening a file?

Answer:

You can open the file by Excel, specifying UpdateLinks:=0

So, you could create a new very small workbook, whose job is to open the main workbook, containing just:

Sub Auto_Open()

  Workbooks.Open ThisWorkbook.Path & "\RealOne.xls", UpdateLinks:=0

  ThisWorkbook.Close False

End Sub

By Bob Umlas

Is there a way to clear a worksheet model of data and leave the formulas intact?

Sub ResetModel()

   Range("A1").SpecialCells(xlCellTypeConstants, xlNumbers).ClearContents

End Sub

By Stephen Bullen

Can you prevent the flickering when a procedure is run in the VBE, similar to using Application.ScreenUpdating = False ?

Declare Function LockWindowUpdate Lib "user32" (ByVal hwndLock As Long) As Long

Sub SomeRoutine()

'Freeze the VBE window - same effect as ScreenUpdating

LockWindowUpdate Application.VBE.MainWindow.HWnd

‘code here

'Unfreeze the VBE window

LockWindowUpdate 0&

End Sub

**********************************

EXCEL 2000 POWER PROGRAMMING TECHNIQUE

By David Hager

The following technique allows you to highlight data in a worksheet while displaying an UserForm that presents the data in chart form when the focus returns to the UserForm. This action can be repeated as many times as desired before closing the UserForm.

Place the event procedures in the module for an UserForm that is named chartssUserForm and the MakeNewChart procedure in a module in the same workbook. The UserForm needs to have a Spreadsheet object named Spreadsheet1 (which is not visible) and a ChartSpace object named ChartSpace1 (which is visible). The two events in the UserForm module each call the same procedure. This is a workaround for the lack of an event that is triggered when an UserForm loses or gains focus. A modeless UserForm is created in Excel 2000 by setting the ShowModal property to False in the Properties box.

Private Sub ChartSpace1_Click(ByVal ChartEventInfo As OWC.WCChartEventInfo)

    MakeNewChart

End Sub

Private Sub UserForm_Click()

    MakeNewChart

End Sub

Sub MakeNewChart()

    On Error GoTo NoChart

    With chartssUserForm

        .ChartSpace1.Clear

        .ChartSpace1.Charts.Add

        .ChartSpace1.DataSource = .Spreadsheet1

        .Spreadsheet1.Cells.Clear

        Application.ScreenUpdating = False

        Selection.Copy

        Sheets.Add

        Range("A1").PasteSpecial

        Selection.Copy

        .Spreadsheet1.ActiveSheet.Range("A1").Paste

        Application.DisplayAlerts = False

        ActiveWindow.SelectedSheets.Delete

        Application.DisplayAlerts = True

        Application.ScreenUpdating = True

        Set cr = .Spreadsheet1.ActiveSheet.Cells(1, 1).CurrentRegion

        TheMax = Application.WorksheetFunction.Max(Range(cr.Address))

        TheRows = cr.Rows.Count

        TheCols = cr.Columns.Count

    End With

    With chartssUserForm.ChartSpace1.Charts(0)

        For NumSeries = 1 To TheCols - 1

            .SeriesCollection.Add

        Next

        For n = 0 To TheCols - 2

        With chartssUserForm.Spreadsheet1.ActiveSheet

            theseriesnamesrange = .Cells(1, n + 2).Address

            thecatagoriesrange = .Range(.Cells(2, 1), .Cells(TheRows, 1)).Address

            thevaluesrange = .Range(.Cells(2, n + 2), .Cells(TheRows, n + 2)).Address

        End With

        With chartssUserForm.ChartSpace1.Charts(0).SeriesCollection(n)

            .SetData chDimSeriesNames, 0, theseriesnamesrange

            .SetData chDimCategories, 0, thecatagoriesrange

            .SetData chDimValues, 0, thevaluesrange

        End With

        Next

        .HasLegend = True

        .Axes(chAxisPositionLeft).NumberFormat = "General"

        .Axes(chAxisPositionLeft).MajorUnit = 0.1 * TheMax

    End With

    chartssUserForm.Show

    Exit Sub

NoChart:

    MsgBox "Your data range is not valid!", , "Try again"

End Sub

**********************************

DID YOU KNOW?...

You can find a comprehensive list of Excel viruses at:

http://www.datafellows.com/v-descs/

by entering Excel in the Search Virus Description Database box.

**********************************

Issue No.5 OF EEE (PUBLISHED 15May1999)

Next issue scheduled for 01Jun1999.

BY David Hager

dchager@compuserve.com

*********************************

**********************************

COMMENTS

Welcome to the 6th issue of the Excel Experts E-letter (or EEE), by David Hager. EEE is a semi-monthly publication. Feel free to distribute copies of EEE to your friends and colleagues. 

My pronouncement in EEE #5 that EEE would no longer be directly mailed was premature. I have corrected some problems I was having with my e-mail software, so I will try to continue direct mailing for now.

I have received some corrections/improvements on a few of the items that have been published in EEE. I plan to start a section for this in the next issue. Another section that will debut to coincide with the release of Office 2000 in EEE #7 is the reporting of design irregularities associated with Excel 2000, and workarounds for these if they exist.

**********************************

TOP EXCEL WEB SITES - WORKSHEET FORMULA TIP

Created by David Hager

If you want to find information posted in Internet newsgroups by a particular person, you can use the following technique. First, make a 4 column list as shown below (on Sheet1).

A      B        C                D

Mike   Jomes    mjomes@abc.com   Mike Jomes mjomes@abc.com 

Kim    Jimes    kjimes@xyz.com   Kim Jimes kjimes@def.com

etc. 

where column D concatenates the information in columns A-C. Then, use that column as the list for Data Validation in a cell. Give the list a defined name such as addList if it is on another worksheet than the HYPERLINK formula shown below, as it is in this example). To make the list dynamic, use:

=OFFSET(Sheet1!$D$1,,,COUNTA(Sheet1!$D:$D),)

Create the Data Validation for cell D1 on Sheet2 (by using =addList as the lookup list). Then, type this formula in A1 on Sheet2 and fill to C1.

=OFFSET(Sheet1!A$1,MATCH($D$1,addList,0)-1,,,)

Finally, type this formula in A2 on Sheet2:

=HYPERLINK("http://www.deja.com/profile.xp?author=%22"&$A$1&"%20"&$B$1&"%22%20%3c"&$C$1&"%3e&ST=PS")

This creates a hyperlink that will return the newsgroup postings of the person you have selected from the list. I use this technique to obtain information posted by specific individuals in the Excel newsgroups.

**********************************

VBA CODE EXAMPLES

Created by Rob Bovey

Creates a list of all number formats in use in the active workbook.

Sub ListNumberFormats()

    Dim lCount As Long

    Dim lRow As Long

    Dim rngCell As Range

    Dim szSheet As String

    Dim szFormat As String

    Dim szFormatArray() As String

    Dim wksSheet As Worksheet

    Dim wkbTargetBook As Workbook

    Dim wkbReportBook As Workbook

    Application.Calculation = xlCalculationManual

    Application.ScreenUpdating = False

    ReDim szFormatArray(0 To 0)

    szFormatArray(0) = "NumberFormats in Use:"

    Set wkbTargetBook = ActiveWorkbook

    For Each wksSheet In wkbTargetBook.Worksheets

        szSheet = wksSheet.Name

        lRow = 1

        For Each rngCell In wksSheet.UsedRange

            If rngCell.Row <> lRow Then lRow = rngCell.Row

            Application.StatusBar = "Determining NumberFormats in use. " & _

                    "Checking worksheet " & szSheet & " row " & CStr(lRow)

            szFormat = rngCell.NumberFormat

    ''' If the current NumberFormat isn't already part of the array, add it.

            If IsError(Application.Match(szFormat, szFormatArray, 0)) Then

                lCount = lCount + 1

                ReDim Preserve szFormatArray(0 To lCount)

                szFormatArray(lCount) = szFormat

            End If

        Next rngCell

    Next wksSheet

    ''' Add a new workbook and dump the array into it.

    Set wkbReportBook = Workbooks.Add(xlWBATWorksheet)

    Set rngCell = wkbReportBook.Worksheets(1).Range("A1")

    For lCount = LBound(szFormatArray) To UBound(szFormatArray)

        rngCell.Offset(lCount, 0).Value = szFormatArray(lCount)

    Next lCount

    rngCell.EntireColumn.AutoFit

    Application.ScreenUpdating = True

    Application.StatusBar = False

    Application.Calculation = xlCalculationAutomatic

End Sub

[In the next issue, a method for deleting custom number formats that are not currently in use (created by Leo Hauser) will be presented.]

Created by Chip Pearson

Cleans up data by removing tabs and carriage returns in worksheet cells. 

Sub CleanUp()

Dim TheCell As Range

For Each TheCell In ActiveSheet.UsedRange

    With TheCell

    If .HasFormula = False Then .Value = Application.WorksheetFunction.Clean(.Value)

    End With

Next TheCell

End Sub

**********************************

POWER PROGRAMMING TECHNIQUE

By Victor Eldridge

The following code is written for Excel 97. It displays pop-up messages when the mouse cursor is rested over embedded charts. Unlike Excel 's built-in chart tips, XTips allows you to specify individual tips for every data point in every series. It also allows you to format the text, and the textbox. Unlike other techniques (that utilise a chart's MouseMove event), XTips avoids screen flicker when working with more than one chart.

To use it, copy the code below to a standard module and run the

   XTipsOn   subroutine to turn them on.

   XTipsOff  subroutine to turn them off.

It assumes that the Source Data for your charts is layed out vertically in columns, and further assumes that the column to the right of the source data, contains your personalized chart tips.

   For example :

     Series1  Series1             Series2    Series2     etc...

      Data       Tips             Data       Tips

       20         What             88         More

       25         ever             74         of

       33         you              63         your

       29         like             93         own

       30         goes             72         chart

       27         here             85         tips

You will also need to place a TextBox (from the Drawing ToolBar) on each worksheet that contains a chart. Format it as you wish. NOTE: NOT an ActiveX textbox.

Known problems:

   * The Worksheet window must be maximised.

   * Excel's Zoom factor must be set to 100% .

   * Windows' Font size must be set to small.

   * Overlapping plot areas may have unpredictable results.

   * The Cursor & Status Bar do not show default Excel behaviour.

   * It does not support some chart types.

   * It does not support chart sheets.

   * Split windows & frozen panes will cause problems.

   * Compared to Excel's built-in chart tips, XTips is slow.

   That 's a pretty long list but everything else seems to work OK. Remember, XTips is only an alternative.

'API function to find out the position of the cursor.

Declare Function GetCursorPos Lib "user32" (lppoint As CursorCoords) As Long

Type CursorCoords

    X As Long

    Y As Long

End Type

Dim pos As CursorCoords

'API function to find out height of the Windows caption bar.

Declare Function GetSystemMetrics Lib "user32" (ByVal nIndex As Long) As Long

Public Const SM_CYCAPTION = 4

Dim PreviousX As Long, PreviousY As Long, CurrentX As Long

Dim CurrentY As Long

Dim GetIt As Variant

Dim Yoffset As Single

Dim Xoffset As Single

Dim NextTime As Date

Dim TLTop As Single, TLLeft As Single

Dim CurrentChart As Chart

Dim chrt As Object

Dim bar As Object

Dim Fix2D As Long

Dim I As Integer

Dim Counter As Integer

Dim X As Long, Y As Long

Dim ElementID As Long, SeriesIndex As Long, PointIndex As Long

Dim F As String

Dim StartOfRange As Integer, EndOfRange As Integer

Dim EndOfWorkbook As Integer

Dim SeriesRange As String, SeriesWorkbook As String, SeriesWorksheet As String

Sub XTipsOn()

    NextTime = Now + TimeValue("00:00:01")

    With Application

        .Cursor = xlNorthwestArrow

        .StatusBar = "Ready"

        .ShowChartTipNames = False

        .ShowChartTipValues = False

        .OnTime NextTime, "XTipsOn" 'Starts a recursive loop.

    End With

    Call GetXoffset

    Call GetYoffset

    'Get the current position of the cursor.

    PreviousX = pos.X - Xoffset - (Application.Left * 1.333) - 3

    PreviousY = pos.Y - Yoffset - (Application.Top * 1.333) - 4

    GetIt = GetCursorPos(pos)

    CurrentX = pos.X - Xoffset - (Application.Left * 1.333) - 3

    CurrentY = pos.Y - Yoffset - (Application.Top * 1.333) - 4

    On Error Resume Next 'happens when no textbox is on the worksheet.

    If CurrentX <> PreviousX Or CurrentY <> PreviousY Then  'The mouse is moving.

            ActiveSheet.TextBoxes(1).Visible = msoFalse

        Else: 'The mouse is at rest.

            If ActiveSheet.TextBoxes(1).Visible = msoFalse Then DisplayTip

    End If

    On Error GoTo 0

End Sub

Sub DisplayTip()

    'Gets the Top & Left values of the cell at the top,left of the screen.

    TLTop = Cells(ActiveWindow.ScrollRow, ActiveWindow.ScrollColumn).Top

    TLLeft = Cells(ActiveWindow.ScrollRow, ActiveWindow.ScrollColumn).Left

    'Works out which chart is underneath the cursor.

    For Each chrt In ActiveSheet.ChartObjects

        If (chrt.Left - TLLeft) * 1.333 < CurrentX And _

           (chrt.Left + chrt.Width - TLLeft) * 1.333 > CurrentX And _

           (chrt.Top - TLTop) * 1.333 < CurrentY And _

           (chrt.Top + chrt.Height - TLTop) * 1.333 > CurrentY _

            Then

            Set CurrentChart = ActiveSheet.ChartObjects(chrt.Index).Chart

            Exit For

        End If

        If chrt.Index = ActiveSheet.ChartObjects.Count Then ' There is no chart underneath the cursor.

            Exit Sub

        End If

    Next

    'Makes an adjustment if the chart does not have a 3D effect.

    On Error GoTo ChartIsNot3D

    Fix2D = CurrentChart.Floor.Interior.ColorIndex

    On Error GoTo 0

    'X & Y will be passed to the GetChartElement method.

    X = CurrentX - (CurrentChart.Parent.Left - TLLeft) * 1.333

    Y = CurrentY - (CurrentChart.Parent.Top - TLTop) * 1.333

    CurrentChart.GetChartElement X, Y, ElementID, SeriesIndex, PointIndex

    If ElementID <> 3 Then Exit Sub

    'Finds the range that contains the Series' Source Data.

    F = CurrentChart.SeriesCollection(SeriesIndex).Formula

    If Mid(F, 1, 10) <> "," _

        Then 'The chart has a range specified for X-axis labels.

        F = Left(F, 9) & Mid(F, InStr(10, F, ","))

    End If

    StartOfRange = InStr(1, F, "!")

    EndOfRange = InStr(StartOfRange + 1, F, ",")

    SeriesRange = Mid(F, StartOfRange + 1, EndOfRange - StartOfRange - 1)

    'Finds the Workbook & Worksheet containing the Series' Source Data.

    EndOfWorkbook = InStr(F, "]")

    If EndOfWorkbook > 0 Then 'The Source Data is in a separate Workbook.

            SeriesWorkbook = Mid(F, 13, EndOfWorkbook - 13)

            SeriesWorksheet = Mid(F, EndOfWorkbook + 1, (StartOfRange - EndOfWorkbook - 2))

        Else: 'The Source Data is in the Active Workbook.

            SeriesWorkbook = ActiveWorkbook.Name

            SeriesWorksheet = Mid(F, 11, InStr(1, F, "!") - 11)

    End If

    'Re-position, re-write & display the text box.

    With ActiveSheet.TextBoxes(1)

        .Left = (CurrentX / 1.333) + TLLeft + 5

        .Top = (CurrentY / 1.333) + TLTop + 12

        On Error GoTo WorkbookNotOpen

        .Characters(1).Insert String:= Workbooks(SeriesWorkbook).Worksheets(SeriesWorksheet). _

             Range(SeriesRange).Offset(PointIndex - 1, 1).Resize(1, 1).Value

        On Error GoTo 0

        .AutoSize = True

        .ShapeRange.ZOrder msoBringToFront

        .Visible = msoTrue

    End With

    Exit Sub

WorkbookNotOpen: ActiveSheet.TextBoxes(1).Characters(1).Insert String:= "The workbook containing" & Chr(10) & "the source data for this" & Chr(10) & "chart needs to be open."

                Resume Next

                Exit Sub

ChartIsNot3D: CurrentX = CurrentX - 1

              CurrentY = CurrentY - 1

              Resume Next

End Sub

Sub GetYoffset()

    'Adds up the heights of all toolbars docked at the top of the screen.

    'If multiple Toolbars share the same RowIndex, only one is counted.

    Yoffset = 0

    ReDim TheArray(0)

    For Each bar In Application.CommandBars

        If bar.Visible = True And bar.Position = msoBarTop Then

            For I = 1 To UBound(TheArray)

                If TheArray(I) = bar.RowIndex Then Yoffset = Yoffset - bar.Height : Exit For

            Next I

            Yoffset = Yoffset + bar.Height

            Counter = Counter + 1

            ReDim Preserve TheArray(Counter)

            TheArray(Counter) = bar.RowIndex

        End If

    Next

    'Accounts for the height of the Windows caption bar.

    Yoffset = Yoffset + GetSystemMetrics(SM_CYCAPTION)

    'Accounts for the height of the Formula Bar.

    If Application.DisplayFormulaBar = True Then Yoffset = Yoffset + 17

    'Accounts for the height of Column Headers.

    On Error Resume Next

    If ActiveWindow.DisplayHeadings = True Then Yoffset = Yoffset + 17

    On Error GoTo 0

End Sub

Sub GetXoffset()

    'Adds up the widths of all toolbars docked at the left of the screen.

    'If multiple Toolbars share the same RowIndex, only one is counted.

    Xoffset = 0

    ReDim TheArray(0)

    For Each bar In Application.CommandBars

        If bar.Visible = True And bar.Position = msoBarLeft Then

            For I = 1 To UBound(TheArray)

                If TheArray(I) = bar.RowIndex Then Xoffset = Xoffset - bar.Width : Exit For

            Next I

            Xoffset = Xoffset + bar.Width

            Counter = Counter + 1

            ReDim Preserve TheArray(Counter)

            TheArray(Counter) = bar.RowIndex

        End If

    Next

    'Makes an adjustment if any toolbars are docked at the left.

    If Xoffset > 0 Then Xoffset = Xoffset - 1

    'Accounts for the width of Row Headers.

    On Error Resume Next

    If ActiveWindow.DisplayHeadings = True Then

        Xoffset = Xoffset + 26

        'If your charts are near row 1000 or row 10000 ,

        'you may need to adjust the values 963 & 9963 .

        If ActiveWindow.ScrollRow > 963 Then Xoffset = Xoffset + 7

        If ActiveWindow.ScrollRow > 9963 Then Xoffset = Xoffset + 7

    End If

    On Error GoTo 0

End Sub

Sub XTipsOff()

    With Application

        .OnTime NextTime, "XTipsOn", schedule:=False

        .Cursor = xlDefault

        .StatusBar = False

        .ShowChartTipNames = True

        .ShowChartTipValues = True

    End With

    ActiveSheet.TextBoxes(1).Visible = msoTrue

End Sub

**********************************

DID YOU KNOW?...

that if you upgrade only to Excel 2000 instead of Office 2000 that you cannot create interactive web pages. Unless you manipulate vast amounts of data, the creation of interactive web pages is really the only major change in Excel 2000 vs. Excel 97. Most users/power users might think that this is not a compelling reason to upgrade. However, even if you don't currently make web pages, you likely will soon since this interactivity allows for the placement of virtually any calculation model into an online environment, making everyone's worksheet constructs available to everyone on the company Intranet. Thus, the upgrade to Office 2000 is a must!

**********************************

Issue No.6 OF EEE (PUBLISHED 01Jun1999)

Next issue scheduled for 16Jun1999.

BY David Hager

dchager@compuserve.com

**********************************

COMMENTS

Welcome to the 7th issue of the Excel Experts E-letter (or EEE), by David Hager. EEE is a semi-monthly publication. Feel free to distribute copies of EEE to your friends and colleagues.

*********************************

EXCEL 2000 DESIGN IRREGULARITIES

Considering the large amount of private and public beta testing of Excel 2000 and the relatively small number of changes that were made, you might have thought that this version would be basically error-free. Well, at least Microsoft is doing a good job of reporting problems. Here are some of the problems associated with copy/pasting.

When you copy and paste cells, copied formulas are pasted as static values; the formulas are not copied. This problem occurs when you copy cells that are not one contiguous range of cells. For example, you select the range A1:A5, C1:C5, copy the cells, and paste them all as one block in cell D1. This problem also existed in Excel 97. See:

http://support.microsoft.com/support/kb/articles/q210/7/25.asp

Since the new Office Clipboard stores only values, you cannot use it to copy/paste formulas. However, it may appear that you can, but the formulas are actually coming from the Windows Clipboard. See:

http://support.microsoft.com/support/kb/articles/q209/2/84.asp

There are problems with the cut/pasting of formulas containing 3D references. See:

http://support.microsoft.com/support/kb/articles/q215/2/17.asp

**********************************

POWER FORMULA TECHNIQUE

Created by David Hager

The goal is to create a way to transform a string into a sorted string. This can be easily done with an user-defined function, as shown below.

Option Base 1

Function SortStr(uSortStr As String) As String

    Dim sArr()

    Dim newStr As String

    Dim store As String

    Dim strlen As Integer 

    strlen = Len(uSortStr)

    ReDim sArr(strlen)

    For s = 1 To strlen

        sArr(s) = Mid(uSortStr, s, 1)

    Next

    For i = 1 To UBound(sArr) - 1

        For j = i + 1 To UBound(sArr)

            If sArr(i) > sArr(j) Then

                store = sArr(i)

                sArr(i) = sArr(j)

                sArr(j) = store

            End If

        Next

    Next

    newStr = ""

    For r = 1 To strlen

        newStr = newStr & sArr(r)

    Next

    SortStr = newStr

End Function

The SortStr function returns a string sorted in ascending order, but it could be easily modified with a second argument to choose ascending or descending order. Although this can be done with an UDF, the challenge is there to accomplish the same goal by just using worksheet formulas. Part of the solution shown below is somewhat kludgy, due to the lack of an Excel function that concatenates elements of an array into a string (perhaps this can be done with the CALL function, though). The following defined name formula transforms a string into a sorted array of characters that comprise the string (the active cell must be B1 during the creation of this formula). 

Define sArr as:

=CHAR(SMALL(CODE(MID(A1,ROW(INDIRECT("1:"&LEN(A1))),1)),ROW(INDIRECT("1:"&LEN(!A1)))))

The MID function creates the array of characters. The CODE function returns the ASCII code number for each character in the array. The SMALL function sorts the array of code numbers in ascending order. Finally, the CHAR function returns the ASCII character for each code number in the array. In order to convert the array into a string, the following defined name formula was created for each character in the string. 

Define zz1 as:

=IF(ISERROR(INDEX(sArr,1)),"",INDEX(sArr,1))

where the number argument in the INDEX function indicates the character position in the array.

These formulas are concatenated by the following defined name formula.

Define SortString as:

=zz1&zz2&zz3&zz4&zz5&...etc

Of course, this will only work for strings that <= the # of formulas that have been concatenated. Now, if you type =SortStr in a cell to the right of a cell containing a string, the sorted string will be returned. I don't know if there is a burning need for the preceding techniques, but it has been an interesting exercise.

**********************************

VBA CODE EXAMPLES 

Created by Rob Bovey

Here is a data encryption/decryption method for strings.

Option Explicit

Sub Test()

    Dim szTest As String

    szTest = "My dog has fleas."

    ''' Encrypt the string

    EncryptDecrypt szTest

    MsgBox szTest

    ''' Decrypt the string

    EncryptDecrypt szTest

    MsgBox szTest

End Sub

'This procedure is a quick and dirty encryption/decryption device. It will process as much text as you can load into a string variable and it is *very* fast. I've encrypted entire documents worth of text with it. You can store the encrypted text in a text file or the registry for later retrieval and decryption.

''' szData  The string you want to encrypt/decrypt.

'''         Pass the string through once to encrypt it.

'''         Pass it through a second time to decrypt it.

'''

Sub EncryptDecrypt(ByRef szData As String)

    Const lKEY_VALUE As Long = 215

    Dim bytData() As Byte

    Dim lCount As Long

    bytData = szData

    For lCount = LBound(bytData) To UBound(bytData)

        bytData(lCount) = bytData(lCount) Xor lKEY_VALUE

    Next lCount

    szData = bytData

End Sub

Sub ViewDecrEncr()

    EncryptDecrypt "This is a test."

    MsgBox szData

End Sub

**********************************

POWER PROGRAMMING TECHNIQUE

By Leo Heuser

This procedure provides a workaround for the glaring lack of accessibility in VBA for manipulating custom number formats. To do this, it hacks into the Number Format dialog box with SendKeys. It loops through each item, including those custom number formats that have been orphaned from the worksheet. The dialog box flickers upon each opening, but it works! If anyone comes up with a way to eliminate the flicker, let me know.

Sub DeleteUnusedCustomNumberFormats()

    Dim Buffer As Object

    Dim Sh As Object

    Dim SaveFormat As Variant

    Dim fFormat As Variant, nFormat() As Variant, xFormat As Long

    Dim Counter As Long, Counter1 As Long, Counter2 As Long

    Dim StartRow As Long, EndRow As Long

    Dim Dummy As Variant

    Dim pPresent As Boolean

    Dim NumberOfFormats As Long

    Dim Answer

    Dim c As Object

    Dim DataStart As Long, DataEnd As Long

    Dim AnswerText As String

    NumberOfFormats = 1000

    ReDim nFormat(0 To NumberOfFormats)

    AnswerText = "Do you want to delete unused custom formats from the workbook?"

    AnswerText = AnswerText & Chr(10) & "To get a list of used and unused formats only, choose No."

    Answer = MsgBox(AnswerText, 259)

    If Answer = vbCancel Then GoTo Finito

    On Error GoTo Finito

    Worksheets.Add.Move after:=Worksheets(Worksheets.Count)

    Worksheets(Worksheets.Count).Name = "CustomFormats"

    Worksheets("CustomFormats").Activate

    Set Buffer = Range("A2")

    Buffer.Select

    nFormat(0) = Buffer.NumberFormatLocal

    Counter = 1

    Do

        SaveFormat = Buffer.NumberFormatLocal

        Dummy = Buffer.NumberFormatLocal

        DoEvents

        SendKeys "{tab 3}{down}{enter}"

        Application.Dialogs(xlDialogFormatNumber).Show Dummy

        nFormat(Counter) = Buffer.NumberFormatLocal

        Counter = Counter + 1

    Loop Until nFormat(Counter - 1) = SaveFormat

    ReDim Preserve nFormat(0 To Counter - 2)

    Range("A1").Value = "Custom formats"

    Range("B1").Value = "Formats used in workbook"

    Range("C1").Value = "Formats not used"

    Range("A1:C1").Font.Bold = True

    StartRow = 3

    EndRow = 16384

    For Counter = 0 To UBound(nFormat)

        Cells(StartRow, 1).Offset(Counter, 0).NumberFormatLocal = nFormat(Counter)

        Cells(StartRow, 1).Offset(Counter, 0).Value = nFormat(Counter)

    Next Counter

    Counter = 0

    For Each Sh In ActiveWorkbook.Worksheets

        If Sh.Name = "CustomFormats" Then Exit For

        For Each c In Sh.UsedRange.Cells

            fFormat = c.NumberFormatLocal

            If Application.WorksheetFunction.CountIf(Range(Cells(StartRow, 2), Cells(EndRow, 2)), fFormat) = 0 Then

                Cells(StartRow, 2).Offset(Counter, 0).NumberFormatLocal = fFormat

                Cells(StartRow, 2).Offset(Counter, 0).Value = fFormat

                Counter = Counter + 1

            End If

        Next c

    Next Sh

    xFormat = Range(Cells(StartRow, 2), Cells(EndRow, 2)).Find("").Row - 2

    Counter2 = 0

    For Counter = 0 To UBound(nFormat)

        pPresent = False

        For Counter1 = 1 To xFormat

            If nFormat(Counter) = Cells(StartRow, 2).Offset(Counter1, 0).NumberFormatLocal Then

                pPresent = True

            End If

        Next Counter1

        If pPresent = False Then

            Cells(StartRow, 3).Offset(Counter2, 0).NumberFormatLocal = nFormat(Counter)

            Cells(StartRow, 3).Offset(Counter2, 0).Value = nFormat(Counter)

            Counter2 = Counter2 + 1

        End If

    Next Counter

    With ActiveSheet.Columns("A:C")

        .AutoFit

        .HorizontalAlignment = xlLeft

    End With

    If Answer = vbYes Then

        DataStart = Range(Cells(1, 3), Cells(EndRow, 3)).Find("").Row + 1

        DataEnd = Cells(DataStart, 3).Resize(EndRow, 1).Find("").Row - 1

        On Error Resume Next

        For Each c In Range(Cells(DataStart, 3), Cells(DataEnd, 3)).Cells

            ActiveWorkbook.DeleteNumberFormat (c.NumberFormat)

        Next c

    End If

Finito:

    Set c = Nothing

    Set Sh = Nothing

    Set Buffer = Nothing

End Sub

**********************************

EXCEL 2000 PROGRAMMING TIP

Created by David Hager

The Spreadsheet Component has an extensive object model similar to Excel itself, but one of the features it does not have is the ability to use array formulas. Presented below is a workaround to that deficiency that allows the entering of an array formula in a Spreadsheet cell and the calculation of that formula to afford the result in the cell. However, the calculation is actually performed on a worksheet named "slink" in the workbook containing this application. So, for this to work you need an UserForm with a CommandButton (named CommandButton1) and a visible Spreadsheet Component (named Spreadsheet1) and the worksheet named "slink". Place this code in the UserForm module. When you want to calculate an array formula in the Spreadsheet Component, you click the button and type your formula in a cell. You can change what is initially in the Spreadsheet Component at design time, and that data is updated on the slink worksheet at run time by the Initialize event. Subsequent changes are handled by the Calculate event of the Spreadsheet Component.

Public EAF As Boolean

Private Sub CommandButton1_Click()

    EAF = True

End Sub

Private Sub Spreadsheet1_Calculate(ByVal EventInfo As OWC.SpreadsheetEventInfo)

    Dim pRange As Range

    Dim aCell As Range

    If Not EAF Then Exit Sub

    On Error Resume Next

    Application.EnableEvents = False

     If Spreadsheet1.ActiveCell.Formula = "" Then

         EAF = False

        Exit Sub

     End If  

    Set pRange = ThisWorkbook.Worksheets("sLink").Range(Spreadsheet1.ActiveSheet.UsedRange.Address)

    Set aCell = ThisWorkbook.Worksheets("sLink").Range(Spreadsheet1.ActiveCell.Address)

    Spreadsheet1.ActiveSheet.UsedRange.Copy

    pRange.PasteSpecial

    Spreadsheet1.ActiveCell.Formula = Application.Evaluate(aCell.Formula)

    EAF = False

End Sub

Private Sub UserForm_Initialize()

    Dim pRange As Range

    Dim aCell As Range

    Application.EnableEvents = False

    Set pRange = ThisWorkbook.Worksheets("sLink").Range(Spreadsheet1.ActiveSheet.UsedRange.Address)

    Spreadsheet1.ActiveSheet.UsedRange.Copy

    pRange.PasteSpecial

End Sub

**********************************

DID YOU KNOW?...

that the Spreadsheet Component calculates dates differently than Excel. In fact, it works much better! See: http://support.microsoft.com/support/kb/articles/q210/7/82.asp for details and http://support.microsoft.com/support/kb/articles/Q216/5/78.asp for information on calculation differences between Excel 2000 and the Spreadsheet Component.

**********************************

**********************************

COMMENTS

Welcome to the 9th issue of the Excel Experts E-letter (or EEE), by David Hager. EEE is a semi-monthly publication. Feel free to distribute copies of EEE to your friends and colleagues. New issues are normally available on the 1st and 16th of each month.

**********************************

TOP EXCEL WEB SITES

An interesting product that allows Excel (version >5) to be used as an Internet data browser has just been released. You can find the details on this at: http://www.inventure.com

**********************************

WORKSHEET FORMULA TIP

Created by Laurent Longre

This formula counts the number of cells in the discontiguous range that contain a value greater than 20.

=INDEX(FREQUENCY((A1,A3,A5),20),2)

**********************************

POWER FORMULA TECHNIQUE

Created by Jan Karel Pieterse

This is an example of how to pass arguments to defined formulas. It consists of the named formula called Myref (see below), which evaluates the string of the formula of the "active" cell (the cell that calls one of the other named formulas in this example). In order to use these formulas, a trick is involved. The function must be called like this:

=IF(ROW(ref),NameOfTheDefinedFormula)

The Myref function finds the "Row(" part in the string of the formula and takes all text between that and the first closing paren to be a valid cell reference. Myref in part is used as an argument to the other functions.

Define Myref as:

=MID(GET.CELL(6,A1),FIND("ROW(",GET.CELL(6,A1))+4,FIND(")",GET.CELL(6,A1))-FIND("ROW(",GET.CELL(6,A1))-4)

Please take into account, that when defining this formula:

- your active cell HAS TO BE cell A1

- You should NOT use absolute refs (no dollar signs)

- You should NOT use sheetrefs ("Sheet1!")

- When editing Myref, you should REMOVE ALL sheetrefs Excel places there itself.

Note, that Myref has to be adapted for other language versions of Excel, which may have a different word for the ROW function (change the string "ROW(" and the two 4's in the formula accordingly).

IsFormula

=GET.CELL(48,INDIRECT(myref)))+0*now()

CellColor

=GET.CELL(63,indirect(MyRef))+0*now()

example:

=IF(ROW(D3),CellColor)

Shows the colournumber of the background of cell D3

RowIsHidden

=IF(GET.CELL(17,INDIRECT(Myref))=0,TRUE,FALSE)+0*now()

RowHeight

=GET.CELL(17,INDIRECT(Myref))+0*NOW()

**********************************

VBA CODE EXAMPLES 

By John Green

This procedure finds cells on a worksheet containing data displayed as #####...

Sub FindIncorrectDataDisplay()

  Dim rng As Range

  For Each rng In ActiveSheet.UsedRange

    If IsNumeric(rng.Value) And Left(rng.Text, 1) = "#" Then MsgBox "Column too narrow for " & rng.Address

  Next rng

End Sub

By Nick Hodge

This procedure prints out all cell comments from a workbook.

First, create a text file on your desktop, (or change the reference in the code), called test.txt and run the code below. This will write each comment, on all worksheets, with it's address and sheet no. to the txt file and close it.  

Sub writeComments()

Dim mycomment As Comment

Dim mySht As Worksheet

Open "C:\Windows\Desktop\Test.txt" For Output As #1

For Each mySht In Worksheets

    On Error Resume Next

    For Each mycomment In Worksheets(mySht.Name).Comments

Print #1, "From " & mycomment.Parent.Parent.Name & mycomment.Parent.Address & " comes the comment: " & mycomment.Text

    Next mycomment

    On Error GoTo 0

Next mySht

Close #1

End Sub
By Laurent Longre

This procedure looks up the Windows 95 serial number.

Declare Function RegOpenKeyExA Lib "Advapi32" (ByVal hkey As Long, ByVal lpszSubKey As String, _

  ByVal dwReserved As Long, ByVal samDesired As Long, phkResult As Long) As Long

Declare Function RegQueryValueExA Lib "Advapi32" (ByVal hkey As Long, ByVal lpszValueName As String, _

  lpwReserved As Long, lpdwType As Long, ByVal lpbData As String, lpcbData As Long) As Long

Declare Function RegCloseKey Lib "Advapi32" (ByVal hkey As Long) As Long

Sub Win95SerialNumber()

  Dim hkey As Long

  Dim Buffer As String

  Dim lgBuf As Long

  If RegOpenKeyExA(&H80000002, "Software\Microsoft\Windows" & "\CurrentVersion", 0, &H960277, hkey) Then Exit Sub

  RegQueryValueExA hkey, "ProductId", 0, 1, Buffer, lgBuf

  Buffer = Space(lgBuf)

  If RegQueryValueExA(hkey, "ProductId", 0, 1, Buffer, lgBuf) = 0 Then MsgBox "Serial number = " & Buffer

  RegCloseKey hkey

End Sub

By Mark Lundberg

The following procedure is a workaround to the lack of a straightforward programmatic way to turn off the 'Break on Unhandled Errors in Class Module' option in the VBE.

Sub AClassCanBeAPainInThe()

    Application.SendKeys "%{F11}%TO+{TAB}{RIGHT 2}%E~%{F4}"

End Sub

**********************************

POWER PROGRAMMING TECHNIQUES

Created by Laurent Longre

This example shows how to register functions into user-defined categories and provide descriptions for their arguments. The Auto_Open procedure registers the two functions, Multiply and Divide in two categories Multiplication and Division and provides descriptions of the input parameters.

Const Lib = """c:\windows\system\user32.dll"""

Option Base 1

Private Function Multiply(N1 As Double, N2 As Double) As Double

  Multiply = N1 * N2

End Function

Private Function Divide(N1 As Double, N2 As Double) As Double

  Divide = N1 / N2

End Function

Sub Auto_open()

  Register "DIVIDE", 3, "Numerator,Divisor", 1, "Division", "Divides two numbers", """Numerator"",""Divisor """, "CharPrevA"

  Register "MULTIPLY", 3, "Number1,Number2", 1, "Multiplication", "Multiplies two numbers", """First number"",""Second number """, "CharNextA"

End Sub

Sub Register(FunctionName As String, NbArgs As Integer, Args As String, MacroType As Integer, Category As String, Descr As String, DescrArgs As String, FLib As String)

  Application.ExecuteExcel4Macro "REGISTER(" & Lib & ",""" & FLib & """,""" & String(NbArgs, "P") & """,""" & FunctionName & """,""" & Args & """," & MacroType & ",""" & Category & """,,,""" & Descr & """," & DescrArgs & ")"

End Sub

Sub Auto_close()

  Dim FName, FLib

  Dim I As Integer

  FName = Array("DIVIDE", "MULTIPLY")

  FLib = Array("CharPrevA", "CharNextA")

  For I = 1 To 2

    With Application

      .ExecuteExcel4Macro "UNREGISTER(" & FName(I) & ")"

      .ExecuteExcel4Macro "REGISTER(" & Lib & _

        ",""CharPrevA"",""P"",""" & FName(I) & """,,0)"

      .ExecuteExcel4Macro "UNREGISTER(" & FName(I) & ")"

    End With

  Next  

End Sub

Created by Bob Umlas and adaptation to a Function procedure by John Walkenbach

This procedure allows for the access of information in a closed workbook with VBA.

From Bob:

Here's a routine you can incorporate into your programs if you ever have a need to retrieve data from a file without opening it. In the needed case, I had to read any number of files (depending on user's selection from a listbox) and determine whether range W11:W36 on sheet "General" totalled zero and do one thing if ANY file met that condition or another thing if not. At first, I thought I'd need to open each file, take the sum, then close it again. Not true. By building a string which you can pass into the ExecuteExcel4Macro, you can access this info directly, without opening the file, making it very fast:

Sub GetDataFromClosedFile()

    filepath = "G:\fsoft\sos\data\ley"

    FileName = "1cA10.sos"  '<==this could change in a loop

    sheetname = "General"

    Strg = "sum('" & filepath & "\[" & FileName & "]" & sheetname & "'!r11c23:r36c23)"

    MsgBox ExecuteExcel4Macro(Strg)

End Sub

'In reality, it looked like this:

Sub GetDataFromClosedFile()

    filepath = "G:\fsoft\sos\data\ley"

    sheetname = "General"

    For Each Fl In DialogSheets("DlgMulti").ListBoxes("MainList").List

        Strg = "sum('" & filepath & "\[" & Fl & "]" & sheetname & "'!r11c23:r36c23)"

        Ans = ExecuteExcel4Macro(Strg)

        If Ans > 0 Then Exit Sub

    Next

    'none > 0

    '...rest of code
End Sub

From John:

The GetValue function listed below takes four arguments:

path:  The drive and path to the closed file (e.g., "d:\files") 

file:  The workbook name (e.g., "99budget.xls") 

sheet: The worksheet name (e.g., "Sheet1") 

ref:   The cell reference (e.g., "C4") 

Private Function GetValue(path, file, sheet, range_ref)

'   Retrieves a value from a closed workbook

Dim arg As String

'   Make sure the file exists 


If Right(path, 1) <> "\" Then path = path & "\"

    
If Dir(path & file) = "" Then        



GetValue = "File Not Found"

        

Exit Function    


End If

'   Create the argument

    arg = "'" & path & "[" & file & "]" & sheet & "'!" & Range(range_ref).Range("A1").Address(, , xlR1C1)

'   Execute an XLM macro

    GetValue = ExecuteExcel4Macro(arg)

End Function

**********************************

EXCEL 2000 INFORMATION

A workbook containing a hyperlink to a blank worksheet in the same workbook, when saved as a Web page and viewed in a Web browser, will do nothing when you click the hyperlink. Excel does not save blank pages by design when a workbook is saved as a web page to conserve disk space. You must add some text to the blank page prior to saving it in this way for the hyperlink to work. For more information on this, see:

http://support.microsoft.com/support/kb/articles/q221/0/64.asp

**********************************

DID YOU KNOW?...

that aside from the new COM add-ins in Office 2000 and using complied xll/dll's, there is no good way to protect the code associated with applications built in Excel. There are password crackers/retrievers available that can hack the password from most if not all versions of Excel. However, there is a way to add a layer of protection to your password for a VB Project in Excel by using unprintable ASCII characters. The following list shows those characters that can be used:

Alt-0128, Alt-0129, Alt-0141, Alt-0142, Alt-0143, Alt-0144, Alt-0157, Alt-0158

There is no visual way to tell the difference among these characters, so a password of suitable length will provide additional protection. It is likely that code crackers can be made to display the corresponding ASCII code, but 

at least it makes the password-breaking process a bit more difficult.

**********************************

COMMENTS

Welcome to the 10th issue of the Excel Experts E-letter (or EEE), by David Hager. EEE is a semi-monthly publication. Feel free to distribute copies of EEE to your friends and colleagues. New issues are normally available on the 1st and 16th of each month.

**********************************

TOP EXCEL WEB SITES

See www.beyondtechnology.com for some great Excel tips and a free Excel newsletter.

**********************************

WORKSHEET FORMULA TIP

By George Simms

This array formula returns the sum of cells in the 3D range bounded by the sheets named in cells B2 and B3.

=SUM(N(INDIRECT(ROW(INDIRECT(B2&":"&B3))&"!A1")))

**********************************

POWER FORMULA TECHNIQUE

By Alan Beban

Here is an example of how to solve a set of simultaneous equations using Excel. Start with equations that are linearly independent so that there is, in fact, a solution; e.g.,

17 = 5x + 3y + 2z

13 = 2x + 4y +  z

22 = 3x + 2y + 5z

Put the coefficients of the unknowns in, e.g., A1:C3 (i.e., 5,3,2 in A1:C1, 2,4,1 in A2:C2, etc.);

Put the constants (17, 13, 22) in, e.g., D1:D3;

Highlight, e.g., E1:E3 and array enter (i.e., enter with Ctrl+Shift+Enter instead of just Enter)

=MMULT(MINVERSE(A1:C3,D1:D3)

and the solution vector (1,2,3) will appear in E1:E3; i.e., x=1, y=2, z=3

For a set of equations that does not have a solution, the #VALUE error will appear in E1:E3.

**********************************

VBA CODE EXAMPLES 

By Andrew Baker

Use this procedure to disable the Excel close button. Send in Me.Caption into either of the following routines. Make sure you do this on the initialise event for 'DisableActiveDialogMenuControls'

'-----------------------------Declarations to Remove Dialog Controls

Private Const MF_BYPOSITION As Long = &H400     

''' Deletes the menus byposition (this is our default)

Private Const MF_BYCOMMAND As Long = &H0        

''' Deletes the menu by Command ID. This is rarely used and is shown here for information purposes only.

Private Const mlNUM_SYS_MENU_ITEMS As Long = 9  

''' This is the number of items on the system menu

Private Declare Function GetSystemMenu Lib "user32" (ByVal hWnd As Long,

ByVal bRevert As Long) As Long

Private Declare Function DeleteMenu Lib "user32" (ByVal hMenu As Long, ByVal

nPosition As Long, ByVal wFlags As Long) As Long

Private Declare Function FindWindowA Lib "user32" (ByVal lpClassName As

String, ByVal lpWindowName As String) As Long

' Comments:   Deletes the system control menu of the specified window.

' Arguments:  DialogCaption The caption of the window whose control menu you want to delete. If not specified, Application.Caption is assumed.

Public Sub DisableActiveDialogMenuControls(DialogCaption As String)

    Dim lHandle As Long, lCount As Long

    On Error Resume Next

    DialogCaption = DialogCaption & vbNullChar

    lHandle = FindWindowA(vbNullString, DialogCaption)

    ' Only continue if the passed window handle isn't zero.

    If lHandle <> 0 Then

        ' There are 9 items on the application control menu.

        ' Loop through and disable each one.

        For lCount = 1 To mlNUM_SYS_MENU_ITEMS

            ' The nPosition of the DeleteMenu function will always be 0, because as we delete each menu item, the next one moves up into the first position (index of 0).

            DeleteMenu GetSystemMenu(lHandle, False), 0, MF_BYPOSITION

        Next lCount

    End If

End Sub

' Comments:   Restores the system control menu of the specified window.

' Arguments:  szCaption   (Optional) The caption of the window whose control menu you want to delete. If not specified, Application.Caption is assumed.

Public Sub EnableActiveDialogMenuControls(DialogCaption As String)

    Dim lHandle As Long

    On Error Resume Next

    DialogCaption = DialogCaption & vbNullChar

    lHandle = FindWindowA(vbNullString, DialogCaption)

    ' Passing True to the bRevert argument of the GetSystemMenu API restores

    ' the control menu of the specified window.

    GetSystemMenu lHandle, True

End Sub

By Robert Rosenburg

This routine clears only numbers from a selection (leaving the formulas & any cells containing text alone).

Sub ClearNumbersOnly()

    Dim iCalc As Integer

    Dim rngCell As Range

    On Error GoTo Error

    If LCase(TypeName(Selection)) = "range" Then

        iCalc = Application.Calculation

        Application.Calculation = xlCalculationManual

        For Each rngCell In Selection

            If Not rngCell.HasFormula Then If Application.IsNumber(rngCell) Then rngCell.ClearContents
        Next rngCell

        Application.Calculation = iCalc

    End If

Error:


Msgbox "Error in: ClearNumbersOnly"

End Sub

**********************************

POWER PROGRAMMING TECHNIQUE

By David Hager

The goal is to create a protected worksheet where filtering and formatting can be done without unprotecting the worksheet. That can be accomplished by placing the following event procedure in the corresponding sheet module.

Public bFlag As Boolean

Private Sub Worksheet_Calculate()

    If bFlag Then Exit Sub

    On Error Resume Next

    With Application

    
.EnableEvents = False

    
.Undo

    
.EnableEvents = True

    End With

End Sub

Now, this procedure will prevent changes to cell contents as long as the Calculate event is triggered. This can be done by placing the following formula in cell A1.

=COUNTA(A2:A65536,B:IV)

If you try to drag and drop data to an area that already contains data, an Excel message prompt appears, but either way it is answered does not affect the protection of the data. If the formula in A1 is moved, circular reference messages appear, but again, the formula is not affected. These prompts and messages occur before any worksheet-based event, so there does not seem to be a way to prevent their appearance. This technique works especially well for sheets containing a list used as a flat database. The filtering of the list does not trigger the Calculate event. 

You might want to have the option to update a worksheet protected in this way. This can be done by using the following procedure.

Sub ChangeTheSheet()

    bFlag = True

    ' some code here to change the worksheet

    bFlag = False

End Sub

By David Hager

There is a little-known effect for the display of charted data that can add considerable polish to a chart presentation. This applies to data ordered by either rows or columns. When records are arranged by rows, they can be hidden through filtering by using Data, Filter, AutoFilter. Columns of charted data can be hidden by using the Format, Column, Hide command. In either case, it turns out that the data that has been filtered or hidden no longer appears on the chart. This effect is quite useful for the viewing of data with a single chart, since what appears on the chart is controlled by the visible data on the worksheet. For example, you can have a chart with many data series and view them one at a time. Other descriptive fields or rows can be added to the data table that enhance to ability to 

filter the data in different ways. Due to the options available for the manipulation of data in the data filtering process, this effect works best when the data is ordered in rows, assuming that the data set is not larger than the number of columns.

One drawback to using this technique for the display on information in a meeting is that changes to the source data would have to be done by toggling between the chart and worksheet holding the data. Fortunately, there is an

easy way to overcome this problem. The desired filter settings can be stored in custom views. Then, a listbox with those views can be added to the chart sheet. Since an ActiveX listbox cannot be used on a chart sheet, you will have to use the native Excel listbox that is available from the Forms toolbar. You can add the desired custom view names programatically or by linking it to a worksheet range. The following procedure will add all of the custom views in a workbook to the listbox. It contains a workaround for a problem in Excel that prevents a normal looping process for the Custom Views collection.
See:

http://support.microsoft.com/support/kb/articles/q164/0/21.asp for more details (this article may not currently be available at the Microsoft site).

Function CreateArrayAndAddToListBox()

    Dim TheArrayCount As Integer

    Dim ListArray()

    With ActiveWorkbook


.CustomViews.Add "Temp"

        TheArrayCount = .CustomViews.Count - 1

        ReDim Preserve ListArray(TheArrayCount)

        For n = 1 To TheArrayCount

            .CustomViews(n).Show

            ListArray(n) = .CustomViews(n).Name & "  (" & ActiveSheet.Name & ")"

        Next

        For i = 1 To TheArrayCount        

            For j = i + 1 To TheArrayCount

                If ListArray(i) > ListArray(j) Then

                    tVar = ListArray(i)

                    ListArray(i) = ListArray(j)

                    ListArray(j) = tVar

                End If

            Next

        Next        


.Sheets("TheChart").ListBoxes("lbShow").List = ListArray

       .CustomViews("Temp").Delete

    End With

End Function

Be aware that there are some problems in running code that shows a chart as a view. I experienced several system crashes, so try to avoid this scenario. It might be preferable to use the worksheet list link, since you can include 

only the custom views you want for a given chart quite easily this way. Then, right-click on the listbox and assign the macro shown below to it.

Sub ChangeChartView()

    Application.ScreenUpdating = False

    ThisChart = ActiveSheet.Name

    With ActiveChart.ListBoxes("lbShow")

    
ActiveWorkbook.CustomViews(.List(.ListIndex)).Show

    End With

    Sheets(ThisChart).Activate

    Application.ScreenUpdating = True

End Sub

Then, by clicking on an item in the listbox, the custom view corresponding to the name of the item clicked will be shown. That will cause the filtering and/or the hiding of columns to be applied to the source data for the chart. That, in turn, will cause complete data points or complete data series to not appear on the chart. If you are using a legend on your chart, it will change to reflect only the data series currently appearing on the chart. 

**********************************

EXCEL 2000 TIP

Are you interested in the role XML plays in Excel 2000 file conversion and web data transmission? See: http://msdn.microsoft.com/isapi/msdnlib.idc?theURL=/library/officedev/offhtml9/shared/ofelexml.htm

**********************************

DO YOU KNOW?...

the steps for making an Office 2000 COM add-in? Here is Stephen Bullen's guide through that process.

Using MOD 2000:

1. Open FP (one instance)

2. switch to the VBE

3. Add a new addin project

Using VB6:

1. Start VB6, electing to create a new COM Addin

2. Do nothing
3. Do nothing

Both:

4. Add a normal module, containing:

Public oFP As FrontPage.Application

Public oEvents As New CEvents

5. Add a class module called CEvents, containing:

Public WithEvents oBtn As CommandBarButton

Private Sub oBtn_Click(ByVal Ctrl As Office.CommandBarButton, CancelDefault As Boolean)

MsgBox "Clicked in " & oFP.ActiveWebWindow.Caption

End Sub

6. Add code to the Designer's code module:

Private Sub AddinInstance_OnConnection(ByVal Application As Object, ByVal ConnectMode As AddInDesignerObjects.ext_ConnectMode, ByVal AddInInst As Object, custom() As Variant)

Dim oBar As CommandBar, oBtn As CommandBarButton

Set oFP = Application

Set oBar = oFP.CommandBars("Menu Bar")

RemoveMenu

Set oBtn = oBar.Controls.Add(msoControlButton)

With oBtn

    .Caption = "Test"

    .Tag = "FPT"

    .Style = msoButtonCaption

End With

Set oEvents.oBtn = oBtn

End Sub

Private Sub AddinInstance_OnDisconnection(ByVal RemoveMode As AddInDesignerObjects.ext_DisconnectMode, custom() As Variant)

RemoveMenu

End Sub

Private Sub RemoveMenu()

Dim oCtls As CommandBarControls, oCtl As CommandBarControl

Set oCtls = oFP.CommandBars.FindControls(Tag:="FPT")

If Not oCtls Is Nothing Then

    For Each oCtl In oCtls

        oCtl.Delete

    Next

End If

End Sub

7. Compile it into a DLL

8. Close the project / Addin

9. Swtich to / open FrontPage 2000

10. Click on Tools, Addins to start the addin

11. Click on the "Test" menu item added to the menu bar - see the message

12. Start another instance of FrontPage

13. Click on the "Test" menu item added to the menu bar - see the message

14. Use File/Open to open a new FP web

15. Click on the "Test" menu item added to the menu bar - see the message

**********************************

COMMENTS

Welcome to the 11th issue of the Excel Experts E-letter (or EEE), by David Hager. EEE is a semi-monthly publication. Feel free to distribute copies of EEE to your friends and colleagues. New issues are normally available on the 1st and 16th of each month. This publication resumes its normal schedule after 6 weeks of inactivity. I appreciate all of the positive comments I received during this time off.

**********************************

TOP EXCEL WEB SITES

Here is a list of web sites for products that will find/remove passwords from Excel workbooks/projects/worksheets. This list was compiled by Tom Ogilvy.

http://www.accessdata.com

http://www.crak.com

http://www.lostpassword.com

http://www.elkraft.unit.no/~huse/xlpassword.htm

http://www.elkraft.ntnu.no/~huse/xlpassword.htm

http://www.elcomsoft.com/ae97pr.html

http://home.telia.no/exceltips/

http://www.zuarin.de/sec_eng.htm

http://webdon.com/mso/

services:

http://www.pwcrack.com/

http://www.passwordservice.com/crack.html

**********************************

WORKSHEET FORMULA TIP

by Bob Umlas

This array formula is an example of a case-sensitive MATCH function.

=MATCH(TRUE,EXACT("A",MyRange),0)

by George Simms

This array formula will extract the phone number as text in the form of 123-45678 from examples as shown below.

234-5678PG

Result 234-5678

Array enter the formula and copy it down as far as needed for entries in column A.

=MID(A1,MATCH(FALSE,ISERROR(1*MID(A1,ROW(INDIRECT("1:20")),1)),0),21-

SUM(1*ISERROR(1*MID(A1,ROW(INDIRECT("1:20")),1))))

**********************************

POWER FORMULA TECHNIQUE

by Stephen Bullen

This formula perform 'bankers rounding' for a number (Num) to a given number (Plc) of significant digits.
=MROUND(Num,IF(VALUE(RIGHT(Num/10^(INT(LOG(ABS(Num)))-Plc+1),2))=0.5,2,1)*

SIGN(Num)*10^(INT(LOG(ABS(Num)))-Plc+1))

If you define 'Fact' as =10^(INT(LOG(ABS(Num)))-Plc+1), this reduces to:

=MROUND(Num,IF(VALUE(RIGHT(Num/Fact,2))=0.5,2,1)*SIGN(Num)*Fact)

**********************************

VBA CODE EXAMPLES 

by Jim Rech (and others)

This procedure selects the last used cell in a worksheet.

Sub GotoLast()

    On Error Resume Next

    Application.ScreenUpdating = False

    Cells(Cells.Find("*", Range("A1"), , , xlByRows, xlPrevious).Row, _

    Cells.Find("*", Range("A1"), , , xlByColumns, xlPrevious).Column).Select

    If Err.Number <> 0 Then MsgBox "No data in sheet"

    Application.ScreenUpdating = True

End Sub

by Stephen Bullen

This function returns the dimension order of an array (up to 4D).

Public Function fnGetDimension(vaArray)

    Dim i As Integer, l As Long

    On Error Resume Next

    Err.Clear

    For i = 1 To 4

        l = UBound(vaArray, i)

        If Err.Number <> 0 Then Exit For

        fnGetDimension = i

    Next

    Err.Clear

End Function

by John Green

This procedure brings data into a worksheet from an external source using ADO. Note that use of the Transpose function will introduce array size limitations in versions of Excel previous to Excel 2000.

Sub GetDataWithADOIn97()

  Dim cnt As New ADODB.Connection

  Dim rst As New ADODB.Recordset

  Dim ws As Worksheet

  Dim recArray As Variant

  Dim fldCount As Integer

  Dim iCols As Integer

  Dim recCount As Long

  Set ws = ActiveSheet

  cnt.Open "Provider=Microsoft.Jet.OLEDB.4.0;" & "Data Source=C:\My Documents\SalesDb.mdb;"

  rst.Open "Select * From SalesData", cnt

  fldCount = rst.Fields.Count

  For iCols = 0 To fldCount - 1

    ws.Cells(1, iCols + 1).Value = rst.Fields(iCols).Name

  Next

  'Copy records to array

  recArray = rst.GetRows

  recCount = UBound(recArray, 2)

  'Transpose array into worksheet

  ws.Range(ws.Cells(2, 1), ws.Cells(recCount + 1, fldCount)).Value = Application.Transpose(recArray)

End Sub

by John Walkenbach

This sub prints (in the Immediate window) the same list of files displayed by the Edit-Links menu command.

Sub ShowLinks()

    On Error Resume Next

    For Each Lnk In ActiveWorkbook.LinkSources(xlExcelLinks)

        Debug.Print Lnk

    Next Lnk

    For Each Lnk In ActiveWorkbook.LinkSources(xlOLELinks)

        Debug.Print Lnk

    Next Lnk

End Sub

by Rob Bovey

This simple procedure displays the chart wizard dialog box.

Sub ShowChartWizard()

    CommandBars("Standard").FindControl(,436).Execute

End Sub

by Jim Rech

Excel does not support automatically adjusting the row height of a merged cell with wrap text set.  This procedure serves as a workaround.

Sub AutoFitMergedCellRowHeight()

    Dim CurrentRowHeight As Single, MergedCellRgWidth As Single

    Dim CurrCell As Range

    Dim ActiveCellWidth As Single, PossNewRowHeight As Single

    If ActiveCell.MergeCells Then

       With ActiveCell.MergeArea

            If .Rows.Count = 1 And .WrapText = True Then

                Application.ScreenUpdating = False

                CurrentRowHeight = .RowHeight

                ActiveCellWidth = ActiveCell.ColumnWidth

                For Each CurrCell In Selection

                    MergedCellRgWidth = CurrCell.ColumnWidth + MergedCellRgWidth

                Next

                .MergeCells = False

                .Cells(1).ColumnWidth = MergedCellRgWidth

                .EntireRow.AutoFit

                PossNewRowHeight = .RowHeight

                .Cells(1).ColumnWidth = ActiveCellWidth

                .MergeCells = True

                .RowHeight = IIf(CurrentRowHeight > PossNewRowHeight, CurrentRowHeight, PossNewRowHeight)

            End If

        End With

    End If

End Sub

by Bernie Deitrick

This procedure returns the named ranges that include the active cell.

Sub DetermineRangeInclusion()

Dim myName As Name

Dim myAddress, myMessage As String

Dim InRange As Integer

myMessage = "Cell " & ActiveCell.Address & " is not in a Range"

InRange = 0

For Each myName In Application.Names

myAddress = myName.RefersTo

Set b = Intersect(ActiveCell, Range(myAddress))

If Not (b Is Nothing) Then

If InRange = 0 Then

InRange = 1

myMessage = "Cell " & b.Address & Chr(10) & Chr(13) & " is in " & myName.Name

Else: myMessage = myMessage & Chr(10) & Chr(13) & " and in " & myName.Name

End If

End If

Next myName

MsgBox myMessage

End Sub

by Jan Karel Pieterse

This procedure searches through all worksheets in a workbook.

Sub FindItAll()

    Dim oSheet As Object

    Dim Firstcell As Range

    Dim NextCell As Range

    Dim WhatToFind As Variant

    WhatToFind = Application.InputBox("What are you looking for ?", "Search", , 100, 100, , , 2)

        If WhatToFind <> "" And Not WhatToFind = False Then

            For Each oSheet In ActiveWorkbook.Worksheets

            oSheet.Activate

            oSheet.[a1].Activate

            Set Firstcell = Cells.Find(What:=WhatToFind, LookIn:=xlValues, LookAt:=xlPart, SearchOrder:=xlByRows, SearchDirection:=xlNext, MatchCase:=False)

                If Not Firstcell Is Nothing Then

                Firstcell.Activate

                MsgBox ("Found " & Chr(34) & WhatToFind & Chr(34) & " in " & oSheet.Name & "!" & Firstcell.Address)

                On Error Resume Next

                    While (Not NextCell Is Nothing) And (Not NextCell.Address = Firstcell.Address)

                    Set NextCell = Cells.FindNext(After:=ActiveCell)

                        If Not NextCell.Address = Firstcell.Address Then

                        NextCell.Activate

                        MsgBox ("Found " & Chr(34) & WhatToFind & Chr(34) & " in " & oSheet.Name & "!" & NextCell.Address)

                        End If

                    Wend

                End If

            Set NextCell = Nothing

            Set Firstcell = Nothing

            Next oSheet

        End If

End Sub

**********************************

EXCEL PRODUCTIVITY TIPS

by Rob Bovey

When the merge cells feature is used on a worksheet, it is difficult to make additional formatting changes to columns/rows that contain the merged cell(s).

The best workaround in this case is just not to use the merge cells feature. The old center across selection, which does the same thing for most purposes and causes no problems, is still available. It's just hidden under the 

Format/Cells/Alignment menu at the bottom of the Horizontal dropdown.

by David Hager

There is a quicker way to freeze formulas to values on a worksheet than using Edit Copy, then Edit Paste Special and choosing the Values option. After making a selection, right-click its edge and drag it away slightly.

Then, place it back in its original position. When you do that, a popup menu appears. Select the Copy Here as Values option and you are finished.

**********************************

DO YOU KNOW?...

that if you apply the Protect method with the UserInterfaceOnly argument set to True to a worksheet and then save the workbook, the entire worksheet (not just the interface) will be fully protected when you reopen the workbook. 

To unprotect the worksheet but re-enable user interface protection after the workbook is opened, you must again apply the Protect method with UserInterfaceOnly set to True.

discoverd by Vasant Nanavati in online help

**********************************

**********************************

COMMENTS

Welcome to the 12th issue of the Excel Experts E-letter (or EEE), by David Hager. EEE is a semi-monthly publication. Feel free to distribute copies of EEE to your friends and colleagues. New issues are normally available on the 1st and 16th of each month. See:

**********************************

TOP EXCEL WEB SITES

Lots of useful information at David McRitchie's web site.

http://members.aol.com/dmcritchie/excel/excel.htm

**********************************

POWER FORMULA TECHNIQUE

by David Hager

Give a range of values (in this case B1:D3), find the maximum value and return the corresponding character in the adjacent column (in this case A1:A3). For the example shown below the answer is "z".

x
1
4
7

y
2
5
8

z
3
6
9

The following array formula will return the desired result.

=INDEX(A1:A3,MAX((B1:D3=MAX(B1:D3))*ROW(A1:A3)))

**********************************

VBA CODE EXAMPLES 

by David Hager

This procedure works in a similar manner to the Edit Fill Across Worksheets command in that it operates on a selection and the selected sheets, but formulas in the selection containing relative references are filled in a 

sheet-relative manner.

Sub FillSpecial()

    Msg = "Do you want to add the sheet name to all references in your selection ?"

    Style = vbYesNo + vbDefaultButton2

    Title = "Add Sheet Name?"

    Response = MsgBox(Msg, Style, Title)

    If Response = vbYes Then

        Application.StatusBar = "Converting references..."

        Add_Sheet_Name_to_Formulas

    End If

    On Error GoTo EOP

    Application.StatusBar = "Starting fill special..."

    Application.ScreenUpdating = False

    Application.DisplayAlerts = False

    Dim SheetPosNum As Integer

    Dim SSPNum As Integer

    Dim wbrray()

    Dim ssrray()

    Dim m As Integer

    Dim n As Integer

    Dim y As Integer

    Dim z As Integer

    Dim sscount As Integer

    Dim CurSheet As String

    Dim ASName As String

    Dim RSName As String

    Dim errval As Variant

    ReDim wbrray(1 To ActiveWorkbook.Sheets.Count)

    ReDim ssrray(1 To Windows(1).SelectedSheets.Count)

    If Windows(1).SelectedSheets.Count = 1 Then

        Application.StatusBar = False

        Exit Sub

    End If

    n = 1

    For Each s In ActiveWorkbook.Sheets

        RSName = Application.Substitute(s.Name, " ", "")

        RTName = Application.Substitute(RSName, "(", "")

        RUName = Application.Substitute(RTName, ")", "")

        If s.Name <> RUName Then

            Msga = "The sheetname [" & s.Name & "] needs to be " & _


    "modified to workwith formulas. Is it OK?"

            Stylea = vbYesNo + vbDefaultButton1

            Titlea = "Change Sheet Name?"

            Responsea = MsgBox(Msga, Stylea, Titlea)

            If Responsea = vbYes Then

                Sheets(s.Name).Name = RUName

                wbrray(n) = RUName

            End If

        Else

            wbrray(n) = s.Name

        End If

        n = n + 1

    Next

    sscount = Windows(1).SelectedSheets.Count

    Application.StatusBar = "0 of " & sscount & " worksheets finished."

    ASName = ActiveSheet.Name

    SheetPosNum = Application.Match(ASName, wbrray, 0)

    ActiveWindow.SelectedSheets.FillAcrossSheets Range:=Selection, Type:=xlContents

    m = 1

    For Each s In Windows(1).SelectedSheets

        ssrray(m) = s.Name

        m = m + 1

    Next

    ActiveSheet.Select

   For t = 1 To sscount

     Application.StatusBar = t & " of " & sscount & " worksheets finished."

     CurSheet = Application.Index(ssrray, t)

     Worksheets(CurSheet).Activate

     SSPNum = Application.Match(ssrray(t), wbrray, 0)

     y = ActiveWorkbook.Sheets.Count

     Selection.Replace What:="=", Replacement:="(/)", LookAt:= xlPart, SearchOrder:=xlByRows, MatchCase:=False

     For Each r In wbrray

         Selection.Replace What:=wbrray(y), Replacement:="ZZZ00" & y, LookAt:= xlPart, SearchOrder:=xlByRows, MatchCase:=False

         y = y - 1

     Next

     z = ActiveWorkbook.Sheets.Count

     For Each q In wbrray

         Selection.Replace What:="ZZZ00" & z + SheetPosNum - SSPNum, Replacement:=wbrray(z), LookAt:= xlPart, SearchOrder:=xlByRows, MatchCase:=False

         z = z - 1

     Next

     Selection.Replace What:="(/)", Replacement:="=", LookAt:= xlPart, SearchOrder:=xlByRows, MatchCase:=False

     For Each CurCell In Selection

           If IsError(CurCell) Then

             errval = CurCell.Value

            Select Case errval

             Case CVErr(xlErrName)

              CurCell.Formula = ""

             Case CVErr(xlErrRef)

              CurCell.Formula = ""

            End Select

           End If

     Next

    Next

    Worksheets(SheetPosNum).Activate

    Application.StatusBar = False

    Exit Sub

EOP:

    MsgBox "Illegal formula reference attempted. Examine all " & "filled formulas and try again."

    Worksheets(SheetPosNum).Activate

    Application.StatusBar = False

End Sub

Sub Add_Sheet_Name_to_Formulas()

    Dim CurrentSheet As String

    On Error GoTo EOSH

    CurrentSheet = ActiveSheet.Name

    Application.ReferenceStyle = xlR1C1

    Application.ScreenUpdating = False

    With Selection

    .Replace What:="RA", Replacement:="ARZZ"

    .Replace What:="RE", Replacement:="ERZZ"

    .Replace What:="RI", Replacement:="IRZZ"

    .Replace What:="RO", Replacement:="ORZZ"

    .Replace What:="+R", Replacement:="+" & CurrentSheet & "!R"

    .Replace What:="-R", Replacement:="-" & CurrentSheet & "!R"

    .Replace What:="(R", Replacement:="(" & CurrentSheet & "!R"

    .Replace What:=",R", Replacement:="," & CurrentSheet & "!R"

    .Replace What:="/R", Replacement:="/" & CurrentSheet & "!R"

    .Replace What:="~*R", Replacement:="*" & CurrentSheet & "!R"

    .Replace What:="=R", Replacement:="=" & CurrentSheet & "!R"

    .Replace What:=" R", Replacement:=" " & CurrentSheet & "!R"

    .Replace What:="^R", Replacement:="^" & CurrentSheet & "!R"

    .Replace What:="&R", Replacement:="&" & CurrentSheet & "!R"

    .Replace What:="(C[", Replacement:="(" & CurrentSheet & "!C["

    .Replace What:=" C[", Replacement:=" " & CurrentSheet & "!C["

    .Replace What:="=C[", Replacement:="=" & CurrentSheet & "!C["

    .Replace What:="~*C[", Replacement:="*" & CurrentSheet & "!C["

    .Replace What:="/C[", Replacement:="/" & CurrentSheet & "!C["

    .Replace What:="ORZZ", Replacement:="RO"

    .Replace What:="IRZZ", Replacement:="RI"

    .Replace What:="ERZZ", Replacement:="RE"

    .Replace What:="ARZZ", Replacement:="RA"

    End With

    Application.ReferenceStyle = xlA1

    Exit Sub

EOSH:

    MsgBox "Not all references may have converted correctly."

    Application.ReferenceStyle = xlA1

End Sub

by Dana DeLouis 

This procedure converts normal formulas to those that show an empty cell

if an error condition exists in the original formula.

Sub ErrorTrapAddDDL()

' Adds =If(IsError() around formulas

  Dim cel As Range

  Dim rng As Range

  Dim Check As String

  Const Equ As String = "=IF(ISERROR(_x) ,"""", _x)"

  Check = Left$(Equ, 12) & "*" ' Check for =IF(ISERROR(

  On Error Resume Next

  Set rng = Selection.SpecialCells(xlFormulas, 23)

  If rng Is Nothing Then Exit Sub

  With WorksheetFunction

    For Each cel In rng

      If Not cel.Formula Like Check Then cel.Formula = .Substitute(Equ, "_x", Mid$(cel.Formula, 2))

    Next

  End With

End Sub

**********************************

DO YOU KNOW?...

that you can use defined names in a workbook that are defined in another workbook. For example, if TheValue is defined as 4 in BookB.xls, typing =BookB.xls!TheValue in a cell in another workbook will return the value 4.

However, the workbook containing the defined name formua must be open for this to work. This is NOT true for defined name ranges. These can be used to communicate with CLOSED workbooks! So, for example, if TheRange is defined as Sheet1!A1:A3 in BookB.xls, typing =SUM(BookB.xls!TheRange) in a cell in another workbook will return the value 17 (if that range contains the values 1,2 and 14). When the workbook containing the defined name range is closed, the full path of BookB.xls will be shown in the formula. Recalculation of that formula continues to return the value 17. Unfortunately, the range cannot be defined with the OFFSET function as an expanding range, such as:

=OFFSET(Sheet1!$A$1,,,COUNTA(Sheet1!$A:$A),)

since this fits into the category of defined name formulas described earlier which do not work with closed workbooks. However, it works fine when the workbook is open.

An important sidenote to the use of defined names is the transport of defined names to another workbook. If you type =MyIncrediblyComplexDFFormula in a cell in the same workbook it is defined in, then copy/paste that cell to 

another workbook, the defined name formula associated with that formula (along with any dependent defined name formulas) will be copied to that workbook as well. This is true even if the workbook and worksheet is completely protected. A method of preventing this from occurring is the attachment of an xlm function of your choosing to the formula (perhaps one that always returns 0).  Since xlm functions cannot be used directly on a worksheet, the destination workbook will not accept the paste operation.

**********************************

Excel Magic Consolidator(MagicCons.xls)

by David Hager

Copyright @1999 All Rights Reserved

Feel free to use this technique in your Excel projects, as long as you include a statement as to the original source.

There are no examples of the formulas referred to in this text in the working xl file, but you should be able to construct your own, based on the following information.

1) Basic Instructions

   a) What does it do?

It allows the user to write formulas on the consolidation worksheet (called "Summary" by default) that act on the same cell from every worksheet in the workbook. The results of the formulas change dynamically as sheets are 

added/deleted from the workbook. Also, the summary sheet can be located at any position within the workbook. There is no VBA or xlm macro code used in this solution. All of the work is done by defined name formulas.

   b) Writing the formulas

As an example, if you type the formula =SUM(cCell) in cell B4 on the Summary worksheet, that formula will return the sum of cell B4 for every worksheet in the workbook, since cCell as used in cell B4 returns the array of entries for those worksheets. Information about using arrays that return entries from cells offset from the cells they are used in can be found in 2a.

   c) Changing the consolidation sheet name

To change the consolidation sheet name, go to Insert, Name, Define in the menu. The named formula called TheSummarySheetName is defined as ="Summary". This means that the worksheet named "Summary" is the only sheet in the workbook that can be used with the consolidation formulas. If, for example, you want change the name to "ConsSheet", then you need to define TheSummarySheetName as ="ConsSheet". Of course, you must have a worksheet by that name as well.

   d) Exporting to an existing workbook

To export this functionality to another workbook, you need to use the Move or Copy menu item from the popup menu that is available when you right-click a worksheet tab. In this case, right-click the Summary tab (or whatever 

name you may have changed it to). Then,  select the desired workbook and sheet location from the dialog box and the checkbox named "Create a copy" and press Enter. All of the defined name formulas will copy over to the new workbook (and of course it is not necessary for your workbook to be named MagicCons.xls). Note that a new workbook must first be saved for this technique to work.

2) How does it work?

   a) Understanding the formulas

All of the formulas used to create the consolidation are defined name formulas. You can view them by selecting Insert, Name, Define from the menu. Do not change these formulas unless you understand how they work. There are 4 constants defined for use in the z-relative formulas. By default, the defined name formulas down, left, right and up have been assigned a value of 1.

TheSummarySheetName is defined as: ="Summary"

This is a defined name formula that sets the name of the worksheet to be used as the consolidation worksheet. 

ThisSheet is defined as:

=LEFT(GET.DOCUMENT(1),FIND("]",GET.DOCUMENT(1)))&TheSummarySheetName

This formula returns the sheet name of the consolidation worksheet in the form "[MagicCons.xls]Summary". This string will be different if used in another workbook and/or with a different consolidation worksheet. This string will be used to match the same string in the TheSheets formula.

TheSheets is defined as: =IF(GET.WORKBOOK(1)=ThisSheet,"",GET.WORKBOOK(1))

The GET.WORKBOOK(1) xlm macro function returns an array of names for the worksheets in the workbook. This formula modifies that array to return an array with an empty string for the array item corresponding to the 

consolidation worksheet. NOTE: You can modify this formula to exclude worksheets other than the "Summary" sheet (if you know how).

cCell is defined as:

=IF(ISERROR(N(INDIRECT(TheSheets&"!"&ADDRESS(ROW(),COLUMN())))),"",

N(INDIRECT(TheSheets&"!"&ADDRESS(ROW(),COLUMN()))))

The concatenated string in the formula INDIRECT(TheSheets&"!"&ADDRESS(ROW(),COLUMN())) creates an array of cell addresses for the cell in which the formula resides all of the worksheets in the workbook. The worksheet cell address for the position on the consolidation worksheet is constructed incorrectly by design so that a circular reference to that cell will not be created. When that string is acted on by the INDIRECT function, a 3-D or

z-range is created. Due to a glitch in how Excel returns this array, it must be acted on by the N function to produce a true array.

cCellDown is defined as:

=IF(ISERROR(N(INDIRECT(TheSheets&"!"&ADDRESS(ROW()+down,COLUMN())))),"",

N(INDIRECT(TheSheets&"!"&ADDRESS(ROW()+down,COLUMN()))))

cCellLeft is defined as:

=IF(ISERROR(N(INDIRECT(TheSheets&"!"&ADDRESS(ROW(),COLUMN()-left)))),"",

N(INDIRECT(TheSheets&"!"&ADDRESS(ROW(),COLUMN()-left))))

cCellRight is defined as:

=IF(ISERROR(N(INDIRECT(TheSheets&"!"&ADDRESS(ROW(),COLUMN()+right)))),"",

N(INDIRECT(TheSheets&"!"&ADDRESS(ROW(),COLUMN()+right))))

cCellUp is defined as:


=IF(ISERROR(N(INDIRECT(TheSheets&"!"&ADDRESS(ROW()-up,COLUMN())))),"",

N(INDIRECT(TheSheets&"!"&ADDRESS(ROW()-up,COLUMN()))))

Realize that in order to use offset arrays of differing dimensions, you will have to define you own hard-coded formulas, such as:

cCellUp4 is defined as:


=IF(ISERROR(N(INDIRECT(TheSheets&"!"&ADDRESS(ROW()-4,COLUMN())))),"",

N(INDIRECT(TheSheets&"!"&ADDRESS(ROW()-4,COLUMN()))))

   b) Using arrays with "non-3D enabled" Excel functions

There are quite a few Excel functions that do not work with the 3D ranges that are inherent to Excel. For example, the MATCH function cannot be as shown in the following formula.

=MATCH(2, Sheet1:Sheet7!C1, 0) 

However, this formula does work as expected.

=MATCH(2, cCell, 0)

In the former case, the 3D range reference Sheet1:Sheet7!C1 does not give an array that the MATCH function can operate on. The latter case contains the readable array cCell (which can be viewed by evaluating that portion of the formula in the formula bar) that MATCH does work with.

   c) Z-relative array formulas 

Since real arrays are returned by cCell and its cousins, they can be used just like any normal range is used in an array formula.

3) Problems

   a) Circular references

If you try to use the consolidation formulas on any other worksheet than the designated consolidation sheet, a circular reference will be created. Do not use these formulas on other worksheets!

   b) Sheets other than worksheets

The presence of charts and Excel5 dialog sheets do not interfere with the workings of the consolidation formulas. However, an Excel4 macro sheet will behave as if was a regular worksheet. This should not cause a problem

in most cases, but if you have entries in cells that correspond to the cell ranges you have chosen for consolidation, they will be used in the formulas.

   c) "Incorrect" result from formulas

The z-relative arrays contain the same number of items as the number of worksheets in your workbook, and that includes the consolidation worksheet. As such, the COUNTA function will always return that number when used with the cCell (and similar) arrays. The value zero is returned from empty cells and so the COUNT function will count those cells. For the same reason, the SMALL, AVERAGE AND MIN functions may not return the expected answer.  Thus, it is recommended that these functions not be used in the consolidation formulas, unless you are sure that each worksheet for a specified cell contains an entry. 

   d) Only returns values 

These formulas have been constructed to return only arrays of values. This was done by design, since consolidation is performed on numbers. All text entries are converted to zero. However, if you would prefer a solution that does include text entries in the arrays, follow these steps:

Define nCell as =N(INDIRECT(TheSheets&"!"&ADDRESS(ROW(),COLUMN())))

Define tCell as =T(INDIRECT(TheSheets&"!"&ADDRESS(ROW(),COLUMN())))

Define cCell as =IF(ISERROR(nCell),"",IF(tCell<>"",tCell,nCell))

Of course you would need to do this for the offset arrays as well. I leave that as an exercise to the reader.

**********************************

COMMENTS

Welcome to the 14th issue of the Excel Experts E-letter (or EEE), by David Hager. EEE is a semi-monthly publication. Feel free to distribute copies of EEE to your friends and colleagues. New issues are normally available on the 1st and 16th of each month. There will be periods when EEE is not published due to time and travel constraints.

**********************************

TOP EXCEL WEB SITES

See:

http://www.appspro.com

for a group of great free Excel utilities that have finally found a home.

**********************************

POWER FUNCTION TECHNIQUE

by Stephen Bullen and David Hager

These functions are modifications of an user-defined function made by Stephen Bullen and published in the Feb'99 issue of PC World magazine. All of these functions are primarily designed to be used as a condition for conditional formatting, as they are meant to be used with a single cell range. When used with multi-cell ranges, these functions will return True if the range argument intersects the filter range. Determining if the range argument is a subset of the filter range would require the comparison of the intersection of the range argument and the filter range to see if it was equal to the range argument.

Function InFilterList(Rng As Range) As Boolean

    On Error GoTo TheEnd

    InFilterList = False

      If Not Intersect(Rng, Rng.Parent.AutoFilter.Range) Is Nothing Then InFilterList = True

    Exit Function

TheEnd:

End Function

The InFilterList function returns True if the range in question is located in a filter range. This is the range where the Data, AutoFilter has been applied but no criteria has been chosen. The act of adding or removing the autofilter does not cause a recalculation of this function when it is used in a conditional formatting formula. Thus, a recalculation on the worksheet is needed for the conditional format to be applied.

Function InFilteredList(Rng As Range) As Boolean

    On Error GoTo TheEnd

    InFilteredList = False

    With Rng.Parent.AutoFilter

      If Not Intersect(Rng, .Range) Is Nothing Then

      For n = 1 To .Range.Columns.Count

        If .Filters(n).On Then

            InFilteredList = True

            Exit For

        End If

      Next

      End If

    End With

    Exit Function

TheEnd:

End Function

The InFilteredList function returns True if the range in question is located in a filtered range. Since the application of the filter is recognized by Excel as a change requiring a recalculation, this function will afford dynamic formatting changes to cells when used in conjunction with conditional formatting. 

Function InFilteredField(Rng As Range) As Boolean

    On Error GoTo TheEnd

    InFilteredField = False

    With Rng.Parent.AutoFilter

      If Not Intersect(Rng, .Range) Is Nothing Then

        If .Filters(Rng.Column - .Range.Column + 1).On Then InFilteredField = True

      End If

    End With

    Exit Function

TheEnd:

End Function

The InFilteredField function returns True if the range in question is located in a column to which a filter has been applied. If the entire filter range has been conditionally formatted, all of the columns that have a set criteria will display the desired formatting.

**********************************

VBA CODE EXAMPLES 

by Stephen Bullen

This procedure delinks all of the charts in a workbook.

Sub RemoveChartLinks()

Dim oSht As Worksheet, oCht As ChartObject, oSeries As Series

'From all embedded charts

For Each oSht In ActiveWorkbook.Worksheets

    For Each oCht In oSht.ChartObjects

        For Each oSeries In oCht.Chart.SeriesCollection

            With oSeries

                .Name = .Name

                .Values = .Values

                .XValues = .XValues

            End With

        Next

    Next

Next

'From all chart sheets

For Each oCht In ActiveWorkbook.Charts

    For Each oSeries In oCht.SeriesCollection

        With oSeries

            .Name = .Name

            .Values = .Values

            .XValues = .XValues

        End With

    Next

Next

End Sub

by Jim Rech

This procedure opens an application through the use of the Shell function and it allows for the lag time involved with the opening process.

Declare Function OpenProcess Lib "kernel32" _

(ByVal dwDesiredAccess As Long, ByVal bInheritHandle As Long, ByVal dwProcessId As Long) As Long

Declare Function GetExitCodeProcess Lib "kernel32" (ByVal hProcess As Long, lpExitCode As Long) As Long

Public Const PROCESS_QUERY_INFORMATION = &H400

Public Const STILL_ACTIVE = &H103

Sub Test()

    Dim StartTime As Double

    StartTime = Now

    ShellAndWait "calc.exe", 1

    MsgBox "Gone " & Format(Now - StartTime, "s") & " seconds"

End Sub

'Window States (Per Help for Shell function):

'   1, 5, 9 Normal with focus.

'   2   Minimized with focus.

'   3   Maximized with focus.

'   4, 8 Normal without focus.

'   6, 7 Minimized without focus.

Sub ShellAndWait(ByVal PathName As String, Optional WindowState)

    Dim hProg As Long

    Dim hProcess As Long, ExitCode As Long

    'fill in the missing parameter and execute the program

    If IsMissing(WindowState) Then WindowState = 1

    hProg = Shell(PathName, WindowState)

    'hProg is a "process ID under Win32. To get the process handle:

    hProcess = OpenProcess(PROCESS_QUERY_INFORMATION, False, hProg)

    Do

    'populate Exitcode variable

        GetExitCodeProcess hProcess, ExitCode

        DoEvents

    Loop While ExitCode = STILL_ACTIVE

End Sub

by Jim Rech

This procedure removes all code and related structures from a workbook.

''Needs a reference to the VB Extensibility library set

'Removes from active workbook all:

 ''Regular modules

 ''Class modules

 ''Userforms

 ''Code in sheet and workbook modules

 ''Non built-in references

 ''Excel 4 macro sheets

 ''Dialog sheets

Sub RemoveAllCode()

    'XL2K:

    'Dim VBComp As VBComponent, AllComp As VBComponents, ThisProj As

    VBProject

    'XL97 & XL2K:

    Dim VBComp As Object, AllComp As Object, ThisProj As Object

    Dim ThisRef As Reference, WS As Worksheet, DLG As DialogSheet

    If ActiveWorkbook.Name <> ThisWorkbook.Name Then

        Set ThisProj = ActiveWorkbook.VBProject

        Set AllComp = ThisProj.VBComponents

        For Each VBComp In AllComp

            With VBComp

                Select Case .Type

                    Case vbext_ct_StdModule, vbext_ct_ClassModule, vbext_ct_MSForm

                        AllComp.Remove VBComp

                    Case vbext_ct_Document

                        .CodeModule.DeleteLines 1, .CodeModule.CountOfLines

                End Select

            End With

        Next

        For Each ThisRef In ThisProj.References

            If Not ThisRef.BuiltIn Then ThisProj.References.Remove ThisRef

        Next

    End If

    Application.DisplayAlerts = False

    For Each WS In Excel4MacroSheets

        WS.Delete

    Next

    For Each DLG In DialogSheets

        DLG.Delete

    Next

End Sub

**********************************

POWER PROGRAMMING TECHNIQUES

by David Hager

Data normalization is a task that is commonly applied in a variety of data workups.  When normalized, the sum of the data equals some value that is set by an arbitrary or real constraint. In Excel, the normalization process is accomplished with a column (or row) of formulas appropriate to the task. The following technique provides a way to convert data to a normalized form without the use of formulas. 

Sub NormalizeRangeValues(Optional nRange As String, Optional nValue As Double = 1)

    If nRange = "" Then nRange = Selection.Address

    nSum = Application.WorksheetFunction.Sum(Range(nRange))

    If nSum = 0 Then Exit Sub

    For Each nCell In Range(nRange)

        With nCell

            If .Value <> "" Then .Value = (nValue / nSum) * .Value

        End With

    Next

End Sub

Sub NormalizeTableValues(tRange As String, _

        Optional nVal As Double = 1, Optional CoR As Boolean = True)

    Dim n As Integer

    If CoR Then

      CoR_Count = Range(tRange).Columns.Count

    Else

      CoR_Count = Range(tRange).Rows.Count

    End If

    For n = 1 To CoR_Count

        NormalizeRangeValues RangeSection(tRange, n, CoR), nVal

    Next

End Sub

Function RangeSection(tRange As String, _

         posNum As Integer, Optional ByCol As Boolean = True) As String

         Dim cOffset As Integer

         Dim rOffset As Integer

         Dim cSize As Integer

         Dim rSize As Integer

         cOffset = 0

         rOffset = 0

         cSize = 1

         rSize = 1

         Set mRange = Range(tRange)

         If ByCol Then

            cOffset = posNum - 1

            rSize = mRange.Rows.Count

         Else

            rOffset = posNum - 1

            cSize = mRange.Columns.Count

         End If

         Set sRange = mRange.Offset(rOffset, cOffset).Resize(rSize, cSize)

         RangeSection = sRange.Address

End Function

Apart from its use with the normalization technique, the RangeSection function can be useful for returning the address of a row or column within a specified range. The function is constructed to return a string, but it can just as easily be made to return a Range object.

Sub RunNormalizeTable()

   Application.EnableEvents = False

   NormalizeTableValues Selection.Address, 2.5, False

   Application.EnableEvents = False

End Sub

The procedure shown above will normalize the data in all of the rows in a selected data table to a value of 2.5. When writing a procedure that incorporates a general utility macro, it is a good idea to disable/enable events in that procedure if it triggers an event that is not inherent to the function of that utility. In the case of using the NormalizeRangeValues function, the cell values are changed, so that will start any application, workbook or worksheet level change event for each cell changed. If those event procedures contain code, that code will run with each change, which may not be the desired outcome.

by David Hager

The following event procedures work together to place the contents of a cell into a cell comment when another entry is made. For example, if a cell contains a value of 13, and 23 is entered in the cell, the cell comment will 

contain the statement:

"Previous entry was 13"

Public acVal

Private Sub Worksheet_Change(ByVal Target As Excel.Range)

    On Error Resume Next

    Target.AddComment

    Target.Comment.Text "Previous entry was " & acVal

End Sub

Private Sub Worksheet_SelectionChange(ByVal Target As Excel.Range)

    If ActiveCell.Address <> Target.Address Then Exit Sub

    acVal = Iif(Target.Value = "", "", Target.Value)
End Sub

The cell value is stored in a public variable when a cell is selected. Then, when a new value is added, the Worksheet_Change event procedure adds a cell comment (the error generated if the cell already has a comment is stepped over) and then uses the stored variable as part of the text string for the comment. This technique could be easily modified to add all of the changes made to a cell over time to the comment.

**********************************

Next issue scheduled for 16Nov1999.

BY David Hager

**********************************

COMMENTS

Y2K is nearly here!

Welcome to the 15th issue of the Excel Experts E-letter (or EEE), by David Hager. EEE is now a monthly publication. Feel free to distribute copies of EEE to your friends and colleagues. New issues are normally available on the 1st of each month. There will be periods when EEE is not published due to time and travel constraints.

**********************************

TOP EXCEL WEB SITES

See Ole P.'s web site for lots of great Excel stuff.

http://w1.2735.telia.com/~u273500023/english/index.htm

Go to Aaron Blood's growing Excel site at:

http://geocities.com/aaronblood

**********************************

WORKSHEET FORMULA TIP

by Harlan Grove

??? formula to reverse the sequence of elements in a range ???

This method makes use of matrix multiplication. The idea is pre or post multiply by a square matrix (N by N)of ones in the elements where the sum of the row and column indices equal N+1 and zeros elsewhere, eg, for 3 by 3

0  0  1

0  1  0

1  0  0

Call these matrices R(N), where N is the dimension (N by N), then for A a matrix with 4 rows and 3 columns, the matrix product  R(4) * A  reverses the row order of A while  A * R(3)  reverses the column order of A.

So if A is

11  12  13

21  22  23

31  32  33

41  42  43

then the array formulas

=MMULT(N(ROW(INDIRECT("1:"&ROWS(A)))=TRANSPOSE(

ROWS(A)+1-ROW(INDIRECT("1:"&ROWS(A))))),A)

and

=MMULT(A,N(ROW(INDIRECT("1:"&COLUMNS(A)))=TRANSPOSE

(COLUMNS(A)+1-ROW(INDIRECT("1:"&COLUMNS(A))))))

give

41  42  43

31  32  33

21  22  23

11  12  13

and

13  12  11

23  22  21

33  32  31

43  42  41

respectively.

**********************************

POWER FORMULA TECHNIQUE

by Bob Umlas

This array formula returns TRUE if the number in cell A1 is a prime number. 

=OR(A1=2,A1=3,ISNA(MATCH(TRUE,A1/ROW(INDIRECT("2:"&INT(SQRT(A1))))=

INT(A1/ROW(INDIRECT("2:"&INT(SQRT(A1))))),0)))

Use it as a conditional formatting formula, with A1 as the active cell in the selection to be formatted.

Here's how Bob's amazing formula works. In a nutshell, the number is divided by all potential prime factors, and the resulting array is tested to see whether it contains a whole number. If is does, you have a prime number. A limitation of this formula is that it cannot test numbers that are greater than 65535^2. This is due to the array size constraint in Excel 97/2000. 

**********************************

VBA CODE EXAMPLES 

by Jake Marx

??? read the names of all Sheets in a closed workbook ???

Here's a way to do it through ADO (ActiveX Data Objects) in Excel 2000.  To use this code, you must first set a reference to "Microsoft ActiveX Data Objects 2.1 Library" and "Microsoft ADO Ext. 2.1 for DDL and Security".

    Sub ReadSheetNames(TheCompleteFilePath As String)

    Dim cnn As New ADODB.Connection

    Dim cat As New ADOX.Catalog

    Dim tbl As ADOX.Table

    cnn.Open "Provider=MSDASQL.1;Data Source=" & "Excel Files;Initial Catalog=" & TheCompleteFilePath

    cat.ActiveConnection = cnn

    For Each tbl In cat.Tables

        MsgBox Left$(tbl.Name, Len(tbl.Name) - 1)

    Next tbl

    Set cat = Nothing

    cnn.Close

    Set cnn = Nothing

    End Sub

by Bill Manville

??? synchronise the horizontal scrolling of 2 windows onto the same worksheet ???

Place this event procedure in the worksheet module.

Private Sub Worksheet_SelectionChange(ByVal Target As Range)

Dim W As Window

Dim stCap as String

stCap = ActiveWindow.Caption

Application.ScreenUpdating = False

If Right(stCap, 2) = ":1" Then

  Set W = Windows(Left(stCap, Len(stCap) - 2) & ":2")

ElseIf Right(stCap, 2) = ":2" Then

  Set W = Windows(Left(stCap, Len(stCap) - 2) & ":1")

Else

  Exit Sub ' single window.

End If

W.ScrollColumn = ActiveWindow.ScrollColumn

Application.ScreenUpdating = True

End Sub

by Bill Manville

??? group multiple worksheets and print a selection from the selected sheets all on one page ???

Sub MultiSheetPrint()

' prints the selected area on each of a set of selected worksheets on a single sheet

  Dim oActive As Object

  Dim oSheet As Object

  Dim oSheets As Object

  Dim wsPrint As Worksheet

  Dim oLastPic As Object

  Dim iPics As Integer

  ' remember where we are

  Set oSheets = ActiveWindow.SelectedSheets

  If oSheets.Count = 1 Then

    Selection.PrintOut preview:=True

    Exit Sub

  End If

  Set oActive = ActiveSheet

  Application.ScreenUpdating = False

  oActive.Select ' otherwise we get lots of new sheets

  Set wsPrint = Worksheets.Add

  For Each oSheet In oSheets

    If TypeName(oSheet) = "Worksheet" Then

      iPics = iPics + 1

      oSheet.Activate

      Selection.CopyPicture

      wsPrint.Cells(iPics * 3 - 2, 1).Value = oSheet.Name

      wsPrint.Paste wsPrint.Cells(iPics * 3 - 1, 1)

      wsPrint.Rows(iPics * 3 - 1).RowHeight = _

            wsPrint.Pictures(iPics).Height

    End If

  Next

  wsPrint.PrintOut preview:=True

  Application.DisplayAlerts = False

  wsPrint.Delete

  Application.DisplayAlerts = True

  oSheets.Select

  oActive.Activate

  Application.ScreenUpdating = True

End Sub

**********************************

POWER FUNCTION TECHNIQUES

by Harlan Grove

This function evaluates first argument, v, and return replacement value, rep, depending on comparison given by cmp. If cmp is blank, replace all error values in v with rep. Otherwise, use Evaluate() with v and cmp, and if the result is True, then replace v with rep.

Function EvalReplace(v As Variant, Optional cmp As String = "", Optional rep As Variant = "") As Variant

    Dim i As Long, j As Long, ret() As Variant, x As Variant

    If TypeOf v Is Range Then v = v.Value

    If Not IsArray(v) Then v = Array(v)

    On Error Resume Next

    j = UBound(v, 2) - LBound(v, 2) + 1

    On Error GoTo 0

    If j = 0 Then

        ReDim ret(1 To 1, 1 To UBound(v, 1) - LBound(v, 1) + 1)

    Else

        ReDim ret(1 To UBound(v, 1) - LBound(v, 1) + 1, 1 To j)

    End If

    i = 1

    j = 1

    For Each x In v

        If cmp = "" Then

            If IsError(x) Then ret(i, j) = rep

        ElseIf Not IsError(x) Then

            If Evaluate("=" & x & cmp) Then ret(i, j) = rep

        End If

        If IsEmpty(ret(i, j)) Then ret(i, j) = x

        If i < UBound(ret, 1) Then

            i = i + 1

        Else

            i = 1

            j = j + 1

        End If

    Next

    EvalReplace = ret

End Function

This function is more efficient at replacing error values than it is at comparison replacements. Nevertheless, when the expression v is complex, this can be preferable to using v twice in IF().

Examples:

=EvalReplace(SQRT(-1))              returns a zero-length string

=EvalReplace(SQRT(-1),,0)         returns 0

=EvalReplace({1,2,3,4},"<2",2)    returns {2,2,3,4}

**********************************

POWER PROGRAMMING TECHNIQUES

by Stephen Bullen

??? assign a procedure to the Click event of a command button added to a form at run time ???

[Class CBtnEvents]

Public WithEvents oBtn As MSForms.CommandButton

Private Sub oBtn_Click()

'... Your code

End Sub

[In the Form]

Dim oEvents As New Collection

Private Sub Userform_Initialize()

Dim oBtnEvts As CBtnEvents

Set oBtnEvts = New CBtnEvents

Set oBtnEvts.oBtn = FrmFieldShow.Controls.Add(bstrprogid:="forms.commandbutton.1", Name:="CmdToG", Visible:=True)

        With oBtnEvts.oBtn

            .Top = 50

            .Height = 25

            .Width = 100

            .Left = (FrmFieldShow.Width / 2 - (100 / 2))

            .Caption = "Ok"

         End with

oEvents.Add oBtnEvts

End Sub

When you click the button, the routine in the class module will fire.

by Laurent Longre

??? add a Add-In path dynamically while it's loading, so the path can be adjusted according to the location of other applications ???

Since the calls to the XLA functions create a link to the XLA file in question, you can test if the path of the add-in is not the same as the path of the link. You should test this in all the workbooks which are already open at load-time of the add-in, and in all the workbooks which will be opened after the add-in is installed.

Dim WithEvents App As Application

Private Sub App_WorkbookOpen(ByVal Wb As Excel.Workbook)

  TestLink Wb

End Sub

Private Sub TestLink(Wb As Workbook)

  Dim Link, I As Integer

  If IsEmpty(Wb.LinkSources(xlExcelLinks)) Then Exit Sub

  For Each Link In Wb.LinkSources(xlExcelLinks)

    If Link = Me.FullName Then Exit Sub

    For I = Len(Link) To 1 Step -1

      If Mid$(Link, I, 1) = "\" Then Exit For

    Next I

    If Mid$(Link, I + 1) = Me.Name Then

      Wb.ChangeLink Link, Me.FullName, xlExcelLinks

      Exit Sub

    End If

  Next Link

End Sub

Private Sub Workbook_Open()

  Dim Wb As Workbook

  For Each Wb In Workbooks

    TestLink Wb

  Next Wb

  Set App = Application

End Sub

**********************************

SPECIAL VBA PROJECT: Custom Number Formats

by Guy Boertje

Finds all of the user-defined custom number formats in a workbook.

Here's how it works.

1)  Save a temporary copy of the workbook.

2)  Open the raw binary file.

3)  Find the bottom of file (BOF) marker in the Workbook globals stream of the compound OLE2 doc.

4)  Find the end of file (EOF) marker in the Workbook globals stream.

5)  Scan between the first byte and the EOF mark looking for number format records.

6)  If one is found, extract the number format string and add it to a collection.

7)  Stop when the EOF mark is reached.  Close the binary file.

8)  Convert the collection of strings to an array of strings.

9)  Return the array.

Option Explicit

Const csNNF As String = "not number format"

Sub RetrieveCustomNumbersFormats()

Dim v, i As Integer

    v = getCustomNumberFormats(ActiveWorkbook)

    For i = 0 To UBound(v)

        ActiveSheet.Cells(i + 1, 1) = v(i)

    Next

End Sub

Public Function getCustomNumberFormats(wb As Workbook) As Variant

'input  - a workbook object

'output - an array of strings

Const BOF_L As Byte = 9, BOF_U As Byte = 8

Const FMT_L As Byte = 30, FMT_U As Byte = 4, U_IDX As Integer = 160

Const EOF_L As Byte = 10, EOF_U As Byte = 0

'change these constants to suit the path to the temp folder on your system

Const drv As String = "C:", stp As String = "TEMP", 

fold As String = "WINDOWS"

Dim hFile As Long, lngLen As Long, i As Long, fIs97 As Boolean

Dim NumFormats() As String, s As String, sep As String

Dim sPath As String, c As New Collection, wbA As Workbook

Dim lngBegin As Long, lngEnd As Long

    'first we need to create a temporary copy of the file to scan

    sep = Application.PathSeparator

    '### adjust below to suit location of your temp folder

'   sPath = drv & sep & stp & sep & wb.Name

    sPath = drv & sep & fold & sep & stp & sep & wb.Name

    '###

    'set xl97 file format flag

    fIs97 = (wb.FileFormat = xlWorkbookNormal Or wb.FileFormat = xlExcel9795)

    wb.SaveCopyAs sPath

    hFile = FreeFile

    Open sPath For Binary Access Read As hFile

    lngLen = LOF(hFile) - 1

    If lngLen > 0 Then

        'find the beginning of the workbook globals stream

        lngBegin = FindBofGlobals(hFile, BOF_L, BOF_U, fIs97)

        'find the end of the workbook globals stream

        lngEnd = FindEofMarker(hFile, EOF_L, EOF_U)

        'sometimes there are number format records before the BOF

        'so scan from the first byte

        If lngBegin > 0 Then lngBegin = 1

        'were the workbook globals markers found?

        If lngBegin > 0 And lngEnd > 0 Then

            'reset the file position

            Seek hFile, lngBegin

            Do While Seek(hFile) < lngEnd

                'scan for a format record

                'i will be the ifmt field

                s = getFmtRec(hFile, FMT_L, FMT_U, i, fIs97)

                'was one found?

                If Not s = csNNF Then

                    'greater than U_IDX is a custom format

                    'use AddTo because if we find the same number format 

                    'in a different record we don't want to add it twice

                    'we might do because we are scanning from the

                    'start of the file not the start of the workbook



    'globals stream 

                    If i > U_IDX Then AddTo c, s

                End If

            Loop

        End If

    End If

    Close hFile

    lngLen = c.Count - 1

    'transfer the collection of strings to an array of strings

    'I think its better to return an array and keep the collection 

    'object local

    If lngLen >= 0 Then

        ReDim NumFormats(lngLen)

        For i = 0 To lngLen

            NumFormats(i) = c(i + 1)

        Next

        getCustomNumberFormats = NumFormats

    End If

    Set c = Nothing

    'get rid of the temp file

    If Len(Dir(sPath)) > 0 Then Kill sPath

End Function

Private Function getFmtRec(h As Long, Lbyte As Byte, Ubyte As Byte, _

i As Long, f97 As Boolean) As String

Dim rec(1) As Byte, l As Long, s As String, o As Long

Dim t As Long, f As Boolean, bytA As Byte, bytB As Byte

Dim j As Long

    'structure of a number format BIFF record

    '2 bytes    marker

    '2 bytes    size

    '2 bytes    ifmt

    '1 byte     length of the format string (can only be 255 characters long)

    'n bytes    format string (with two zero bytes if xl97 file fmt)

    getFmtRec = csNNF

    s = vbNullString

    'get the first byte

    Get h, , rec(0)

    'is it the first part of the number formats marker?

    If rec(0) = Lbyte Then

        'if so then get the next byte

        Get h, , rec(1)

        'is it the second part of the number formats marker?

        If rec(1) = Ubyte Then

            o = getTwoBytes(h) 'get the offset - the size of the record

            i = getTwoBytes(h) 'get the ifmt field - number format is 





'built-in or custom

            t = getOneByte(h) 'get the length of the format string

            'check that the offset and the length of the format string 


    'differ by 5 bytes

            l = o - 5

            If t <> l Then Debug.Print o; l; t 'if this bit executes then 







'there are corrupted records

            'in the xl97 file format there are two null bytes before 


    'the format string

            If f97 Then t = t + 2

            s = getFormatString(h, t, Ubyte, Lbyte)

            If f97 Then

                'strip the two null bytes away

                getFmtRec = Mid$(s, 3)

            Else

                getFmtRec = s

            End If

        End If

    End If

End Function

Private Function getFormatString(h As Long, l As Long, Ubyt As Byte, Lbyt As Byte) As String

Dim j As Long, byt(1) As Byte, s As String

    For j = 1 To l

        Get h, , byt(0)

        'while getting the string, make sure that

        'it is not the start of the next format record

        If byt(0) = Lbyt Then

            Get h, , byt(1)

            If byt(1) = Ubyt Then

                'if a number format record is found then

                'move the file pointer back two bytes and exit

                Seek h, Seek(h) - 2

                Exit For

            Else

                'otherwise move the file pointer back one byte

                'making sure that no bytes are skipped

                Seek h, Seek(h) - 1

            End If

        End If

        s = s & Chr$(byt(0))

    Next

    getFormatString = s

End Function

Sub AddTo(c As Collection, s As String)

    'will get an error if the key has been used before

    'this guarantees that each string in the collection is unique

    On Error Resume Next

    c.Add Item:=s, key:=s

End Sub

Private Function FindBofGlobals(h As Long, Lbyte As Byte, Ubyte As Byte, f97 As Boolean) As Long

Dim rec(1) As Byte, recA(5) As Byte, l As Long, s As String

Dim offs(1) As Byte, bifv(1) As Byte

Dim wgbl(1) As Byte, place As Long, f As Boolean

    If f97 Then

        'in xl97 the BOF record is 16 bytes long

        offs(0) = 16: offs(1) = 0

        'biff8 is indicated by a 6 in the upper byte

        bifv(0) = 0: bifv(1) = 6

    Else

        'previously it was 8 bytes long

        offs(0) = 8: offs(1) = 0

        'biff5or7 is indicated by a 5 in the upper byte

        bifv(0) = 0: bifv(1) = 5

    End If

    'the workgroup globals BOF is marked as 5

    'there are other BOFs records, marked differently

    wgbl(0) = 5: wgbl(1) = 0

    FindBofGlobals = -1

    Do

        'jump in 2 byte steps until the BOF record or the end of file 


'is reached

        Get h, , rec

        If Seek(h) >= LOF(h) - 7 Then Exit Function

        f = (rec(0) = Lbyte And rec(1) = Ubyte)

        If f Then

        'remember point where we have tested for BOF marker

        'now we test for the other elements for a valid wb global bof

            place = Seek(h)

            Get h, , recA

            'is the offset the correct size?

            'is the biff version correct?

            'is it a wb global bof?

            f = recA(0) = offs(0) And recA(1) = offs(1) And _

                recA(2) = bifv(0) And recA(3) = bifv(1) And _

                recA(4) = wgbl(0) And recA(5) = wgbl(1)

            If Not f Then Seek h, place 'move the file pointer back to the remembered point

        End If

    Loop Until f

    'return the start point of the bof record

    FindBofGlobals = place - 2

End Function

Private Function FindEofMarker(h As Long, Lbyte As Byte, Ubyte As Byte) As Long

Dim rec(1) As Byte, f As Boolean, place As Long

    FindEofMarker = -1

    Do

        'jump in two byte steps until the EOF record or the end of file is 


'reached

        Get h, , rec

        If Seek(h) >= LOF(h) - 5 Then Exit Function

        'is it an eof record?

        f = (rec(0) = Lbyte And rec(1) = Ubyte)

        If f Then

            'remember last eof tested point

            place = Seek(h)

            'are the next two bytes both zero?

            'they should be for a valid eof

            f = (getTwoBytes(h) = 0)

            If Not f Then Seek h, place

        End If

    Loop Until f

    'return the start point of the eof record

    FindEofMarker = place - 2

End Function

Private Function getTwoBytes(h As Long) As Long

Dim rec(1) As Byte, l As Long

    'returns the next two bytes in the file as a Long

    Get h, , rec

    getTwoBytes = CLng(rec(0)) + CLng(rec(1)) * 256

End Function

Private Function getOneByte(h As Long) As Integer

Dim rec As Byte

    'returns the next byte in the file as an Integer

    Get h, , rec

    getOneByte = rec

End Function

**********************************

COMMENTS

Welcome to the 16th issue of the Excel Experts E-letter (or EEE), by David Hager. EEE is now a monthly publication. Feel free to distribute copies of EEE to your friends and colleagues. New issues are normally available on the 1st of each month. There will be periods when EEE is not published due to time and travel constraints.

**********************************

TOP EXCEL WEB SITES

Visit Chip Pearson's growing and everchanging Excel web site at:

http://www.cpearson.com

A new addition to his site are two interesting Excel games (free with unprotected source code) made by yours truly.

http://www.cpearson.com/excel/games.htm

This web page provides a wealth of diverse Excel information.

http://www.mathtools.net/Excel/index.html

**********************************

WORKSHEET FORMULA TIP

by Harlan Grove

Needed: A formula to determine if the items contained in Range1 are contained in Range2. If not, then a comparison of Range1 will be made to another range and so on. For example:

Range1:

 A       B       C

PEAR    APPLE  ORANGE

Range2:

 A       B       C       D

PEAR    APPLE   ORANGE  BANANA

To check if everything in Range1 appears in Range2, you could use this array formula:

=AND(NOT(ISNA(MATCH(Range1,Range2,0))))

Trickier: if all single row ranges to check Range1 against are collected into a single table, for example, Range 3 as

pear    mango   orange

pear    mango   grapes  banana  dates

pear    grapes  orange  banana

grapes  mango   orange  banana

pear    apple   grapes  banana  dates   figs

apple   pear    orange  banana

grapes  apple   orange  banana

pear    apple   orange  banana  dates   figs    cheries

then the following array function will return the row index of the first (topmost) row in which there's a match for all entries in Range1:
=MATCH(COLUMNS(Range1),MMULT(COUNTIF(Range1,Range3),TRANSPOSE(COLUMN(Range3)^0)),0)

which takes advantage of COUNTIF's peculiar semantics when both of its arguments are arrays. This formula returns 6.

**********************************

POWER FORMULA TECHNIQUE

by David Hager

This array formula returns TRUE if the number in cell A1 is a Fibonacci number. A Fibonacci number is a member of the number series 1,1,2,3,5,8,13,21,34,55,89,... which is intimately linked to a variety of growth and life processes.

=OR(A1=ROUND((((SQRT(5)+1)/2)^ROW(1:73))/SQRT(5),0))

by Harlan Grove

This formula is a general two dimensional array reshaping formula for an array of size NewRows x NewCols, similar to APL's RHO array, that works for any worksheet array A.

=N(OFFSET(A,MOD(INT(((ROW(INDIRECT("1:"&NewRows))-1)*NewCols+

TRANSPOSE(ROW(INDIRECT("1:"&NewCols))-1))/COLUMNS(A)),

ROWS(A)),MOD(((ROW(INDIRECT("1:"&NewRows))-1)*NewCols+

TRANSPOSE(ROW(INDIRECT("1:"&NewCols))-1)),COLUMNS(A)),1,1))

For example, if A is {11,12;21,22;31,32;41,42;51,52;61,62}, NewRows has value 5 and NewCols has value 3, this formula gives {11,12,21;22,31,32;41,42,51;52,61,62;11,12,21}.

**********************************

VBA CODE EXAMPLES 

by David Hager

Use the first function to read a range from a closed workbook into an array and the second procedure for direct input into a range on the active worksheet.

'CWRIA is short for ClosedWorkbookRangeIntoArray

Function CWRIA(fPath As String, fName As String, sName As String, rng As String)

    Dim sRow As Integer

    Dim sColumn  As Integer

    Dim sRows As Integer

    Dim sColumns  As Integer

    Dim vrow As Integer

    Dim vcol  As Integer

    Dim fpStr As String

    Dim cArr()

    On Error GoTo NoArr

    If Right(fPath, 1) <> "\" Then fPath = fPath & "\"

    If Dir(fPath & fName) = "" Then

        CWA = CVErr(xlErrValue)

        Exit Function

    End If

    sRow = Range(rng).Row

    sColumn = Range(rng).Column

    sRows = Range(rng).Rows.Count

    sColumns = Range(rng).Columns.Count

    ReDim cArr(sRows, sColumns)

    For vrow = 1 To sRows

        For vcol = 1 To sColumns

            fpStr = "'" & fPath & "[" & fName & "]" & sName & "'!" & _

            "r" & sRow + vrow - 1 & "c" & sColumn + vcol - 1

            cArr(vrow, vcol) = ExecuteExcel4Macro(fpStr)

        Next

    Next

    CWRIA = cArr

    Exit Function

NoArr:

    CWRIA = CVErr(xlErrValue)

End Function

'CWRIR is short for ClosedWorkbookRangeIntoArray

Sub CWRIR(fPath As String, fName As String, sName As String, rng As String, destRngUpperLeftCell As String)

    Dim sRow As Integer

    Dim sColumn  As Integer

    Dim sRows As Integer

    Dim sColumns  As Integer

    Dim vrow As Integer

    Dim vcol  As Integer

    Dim fpStr As String

    Dim cArr()

    On Error GoTo NoArr

    If Right(fPath, 1) <> "\" Then fPath = fPath & "\"

    If Dir(fPath & fName) = "" Then

        CWA = CVErr(xlErrValue)

        Exit Function

    End If

    sRow = Range(rng).Row

    sColumn = Range(rng).Column

    sRows = Range(rng).Rows.Count

    sColumns = Range(rng).Columns.Count

    ReDim cArr(sRows, sColumns)

    Set destRange = ActiveSheet.Range(destRngUpperLeftCell) 

    For vrow = 1 To sRows

        For vcol = 1 To sColumns

            fpStr = "'" & fPath & "[" & fName & "]" & sName & "'!" & _

            "r" & sRow + vrow - 1 & "c" & sColumn + vcol - 1

            destRange.Offset(vrow - 1, vcol - 1) = ExecuteExcel4Macro(fpStr)

        Next

    Next

NoArr:

End Sub

The following procedure copies the values from the range A1:C3 from Sheet1 of the closed workbook cellDataVal.xls located at D:\EXCEL97\xlformulas to the range F9:H11 on the active worksheet.   

Sub InsertRangeFromClosedWorkbook()

    CWRIR "D:\EXCEL97\xlformulas", "cellDataVal.xls", "Sheet1", "a1:c3", "f9"

End Sub

**********************************

POWER PROGRAMMING TECHNIQUES

by xxxxxx

Here is a method for counting instances of Excel application and storing the handles for each instance in an array. 

Option Explicit

Private Declare Function FindWindow Lib "user32" Alias "FindWindowA"(ByVal lpClassName As String, ByVal lpWindowName As String) As Long

Private Declare Function GetNextWindow Lib "user32" Alias "GetWindow"(ByVal hwnd As Long, ByVal wFlag As Long) As Long

Private Declare Function GetClassName Lib "user32" Alias "GetClassNameA"(ByVal hwnd As Long, ByVal lpClassName As String, ByVal nMaxCount As Long) As Long

Private Const GW_HWNDNEXT = 2

Sub xlInstances()

  Dim hwnd As Long, lRet As Long

  Dim hWndArray() As Long

  Dim i As Integer

  Dim sClassBuffer As String

  i = 0

  hwnd = FindWindow("XLMAIN", vbNullString)

  If hwnd <> 0 Then

    ReDim hWndArray(i)

    hWndArray(i) = hwnd

    Do

      hwnd = GetNextWindow(hwnd, GW_HWNDNEXT)

      If hwnd = 0 Then Exit Sub

      sClassBuffer = String(255, 0)

      lRet = GetClassName(hwnd, sClassBuffer, Len(sClassBuffer))

      sClassBuffer = Left(sClassBuffer, InStr(1, sClassBuffer, Chr(0),

      vbTextCompare) - 1)

      If UCase(sClassBuffer) = "XLMAIN" Then

        i = i + 1

        ReDim Preserve hWndArray(i)

        hWndArray(i) = hwnd

      End If

    Loop

  End If

End Sub

Keep the Array hWndArray global, so that you can access it outside the search function. The handle is valid as long as the instance exists and will die if you quit Excel.

by Tom Ogilvy and David Braden

A FAST method for building an unique list from data in column A.

Sub BuildUnique1()

Dim vArr As Variant

Dim vArr1 As Variant

Set RNG = Range(Cells(1, "A"), Cells(1, "A"). End(xlDown))

vArr = Application.Transpose(RNG)

ShellSort vArr

ReDim vArr1(1 To 1)

vArr1(1) = vArr(1)

j = 1

For i = LBound(vArr, 1) + 1 To UBound(vArr, 1)

If vArr(i) <> vArr1(j) Then

   j = j + 1

   ReDim Preserve vArr1(1 To j)

   vArr1(j) = vArr(i)

End If

Next

End Sub

Using David Braden's implementation of ShellSort:

Sub ShellSort(list As Variant, Optional ByVal LowIndex As Variant, Optional HiIndex As Variant)

   'Translation of Shell's Sort as described in "Numerical Recipes in C", 2nd edition, Press et al. For large arrays, consider Quicksort. This algorithm is at least as good up to about 100 or so elements. But with 500 randomized elements it is about 27% slower than QSort, and looks increasingly worse as the array size increases.

   'Dec 17, '98 - David J. Braden

   Dim i As Long, j As Long, inc As Long

   Dim var As Variant

   If IsMissing(LowIndex) Then LowIndex = LBound(list)

   If IsMissing(HiIndex) Then HiIndex = UBound(list)

   inc = 1

   Do While inc <= HiIndex - LowIndex: inc = 3 * inc + 1: Loop

   Do

      inc = inc \ 3

      For i = LowIndex + inc To HiIndex

         var = list(i)

         j = i

         Do While list(j - inc) > var

            list(j) = list(j - inc)

            j = j - inc

            If j <= inc Then Exit Do

         Loop

         list(j) = var

      Next

   Loop While inc > 1

End Sub

by Laurent Longre

VBA code for placing a shortcut on the desktop.

Declare Function SHGetSpecialFolderLocation Lib "Shell32" (ByVal hwnd As Long, ByVal nFolder As Long, ppidl As Long) As Long

Declare Function SHGetPathFromIDList Lib "Shell32" (ByVal Pidl As Long, ByVal pszPath As String) As Long

Declare Function SetWindowPos Lib "User32" (ByVal hwnd As Long, ByVal hWndInsertAfter As Long, _

  ByVal X As Long, ByVal Y As Long, ByVal cx As Long, ByVal cy As Long, ByVal uFlags As Long) As Long

Declare Function SetForegroundWindow Lib "User32" (ByVal hwnd As Long) As Long

Declare Function GetForegroundWindow Lib "User32" () As Long

Function ShortCut(Target As String, Optional Target_Type As Long) As Boolean

  Dim hwnd As Long

  Dim Pidl As Long

  Dim Bureau As String

  If Dir(Target & IIf(Target_Type = vbDirectory, "\", ""), Target_Type) = "" Then Exit Function

  SHGetSpecialFolderLocation 0, 0, Pidl

  Bureau = Space(260)

  SHGetPathFromIDList Pidl, Bureau

  Bureau = Left(Bureau, InStr(1, Bureau, vbNullChar) - 1)

  hwnd = GetForegroundWindow

  SetWindowPos hwnd, -1, 0, 0, 0, 0, 3

  Shell "RunDLL32 AppWiz.Cpl,NewLinkHere " & Bureau & "\"

  SendKeys """" & Target & """~~", True

  SetForegroundWindow hwnd

  ShortCut = True

End Function

Sub Test()

  ' Creates a shortcut to the directory "C:\Temp"

  MsgBox IIf(ShortCut("C:\Temp", vbDirectory), "Shortcut created", "Can't find the directory")

  ' Creates a shortcut to the file "C:\Temp\Zaza.xls"

  MsgBox IIf(ShortCut("C:\Temp\Zaza.xls"), "Shortcut created", "Can't find the file")

End Sub

**********************************

EXCEL DEVELOPER'S TIP

by Jim Rech

How to duplicate your VBE setup.

So you've got your new PC and you've copied over your Excel.xlb and Personal.xls from the old machine.  Now you go into the VBE and... oh yeah, how do you copy over your VBE preferences, customizations and toolbars?  

Here's how:

- Run RedEdit.exe

- Navigate to the key HKEY_CURRENT_USER\Software\Microsoft\VBA\6.0\Common for Office 2000 or HKEY_CURRENT_USER\Software\Microsoft\VBA\Office for Office 97.

- From the file menu pick Export Registry File and select a file name.

- Copy the resulting REG file to the new machine.

- On the new machine you can run RegEdit and pick Import Registry File or from Windows Explorer right click on the file and select Merge.

**********************************

Top Excel Sites

Go to this site for a great index of Excel information.

http://www.mathtools.net/Excel/index.html

**********************************

WEB INFORMATION ON:

ExecuteExcel4Macro method

Access to closed workbooks:

http://eva.dc.lsoft.com/scripts/wa.exe?A2=ind9908D&L=excel-l&P=R8921&m=28403

http://eva.dc.lsoft.com/scripts/wa.exe?A2=ind9812C&L=excel-l&P=R8589&m=28403

http://x43.deja.com/=dnc/getdoc.xp?AN=370019007&CONTEXT=936234727.222691395&hitnum=103

http://x30.deja.com/=dnc/getdoc.xp?AN=394587442&CONTEXT=936236165.1461321770&hitnum=69

Excel charts:

http://peach.ease.lsoft.com/scripts/wa.exe?A2=ind9810B&L=excel-g&P=R5489

http://x30.deja.com/=dnc/getdoc.xp?AN=497422618&CONTEXT=936236165.1461321770&hitnum=37

http://x43.deja.com/=dnc/getdoc.xp?AN=362117773&CONTEXT=936234727.222691395&hitnum=121

http://x43.deja.com/=dnc/getdoc.xp?AN=247283361&CONTEXT=936234727.222691395&hitnum=187

Printing:

http://x43.deja.com/=dnc/getdoc.xp?AN=240180079&CONTEXT=936234727.222691395&hitnum=191

http://x43.deja.com/=dnc/getdoc.xp?AN=296331018&CONTEXT=936234727.222691395&hitnum=160

http://support.microsoft.com/support/kb/articles/Q139/4/05.asp

Formula.Find:

http://www.mailbase.ac.uk/lists/excel-vb-discuss/1999-03/0001.html

Custom Views:

http://eva.dc.lsoft.com/scripts/wa.exe?A2=ind9606&L=excel-l&P=R8426

http://eva.dc.lsoft.com/scripts/wa.exe?A2=ind9606&L=excel-l&P=R8508

Delete Rows:

http://x30.deja.com/=dnc/getdoc.xp?AN=484149794&CONTEXT=936236165.1461321770&hitnum=97

Set.Update.Status

http://support.microsoft.com/support/kb/articles/Q108/3/84.asp

Close:

http://x43.deja.com/=dnc/getdoc.xp?AN=209988463&CONTEXT=936234727.222691395&hitnum=202

ShowBar:

http://x43.deja.com/=dnc/getdoc.xp?AN=364967596&CONTEXT=936234727.222691395&hitnum=120

**********************************

WORKSHEET FORMULA TIP

by Harlan Grove

Question from Microsoft Excel worksheet formula newsgroup:

Is there a formula that I could use to count the number of rows that contain data based on my criteria... Using this data as an example

 A    B   C    D    E    F

 1    3    5    8    12    13

 3    2    6    5    7    9

 5    7    4    8    12    3

I would like to identify how many times 3 and 5 appear together and I might even want to identify how many times 3, 5 and 12 appear together. From the above example data you can see that the numbers 3, 5 and 12 wont always be in the same column, using DCOUNT restricts me to identifying the column heading in my field criteria... My criteria my look like this

Column headings----    Find 1  Find 2  Find 3  Find 4  Find 5  Find 6

and I would enter----           3          5         12

the formula / function -  whatever it is that would do what I require would return a count on the number of rows that contain 3,5 and 12 any order...

Answer:

With your sample data range named MyData and the 'criteria' entry cells (all 6) range above named MyCrit, try this array formula.

=COUNT(IF(MMULT(COUNTIF(MyCrit,MyData),TRANSPOSE(COLUMN(MyData)^0))=COUNT(MyCrit),1))

Note: this assumes no duplicate 'criteria' entries. 

**********************************

POWER FORMULA/FUNCTION TECHNIQUES

by David Hager

I wrote this array formula to combine the functionality of the XIRR and MIRR functions. This formula returns the internal rate of return for a schedule of cash flows that is not necessarily periodic while considering both the cost of the investment and the interest received on reinvestment of cash. The fields used in the formula are defined below.

=POWER((SUM(IF(values>0,values*(POWER(1+rRate,(MAX(dates)-dates)/daybase)),0)))/(SUM(IF(values<0,values/(POWER(1+iRate,(MAX(dates)-dates)/daybase)),0)))*-1,1/((MAX(dates)-MIN(dates))/daybase))-1

where:

values is the row or column range of cashflows

dates is the row or column range of corresponding dates

iRate is the interest rate you pay on the money used in the cash flows

rRate is the interest rate you receive on the cash flows as you reinvest them

daybase is days-in-year basis to use (usually 360 or 365).

The following UDF provides the same functionality as the array formula.

Function XMIRR(TheValues As Range, TheDates As Range, iRate, rRate, daybase)

    Dim rCount As Integer

    Dim cCount As Integer

    Dim rCounter As Integer

    Dim cCounter As Integer

    Dim TheVal As Double

    Dim TheDate As Double

    Dim MaxDate As Double

    Dim MinDate As Double

    Dim PosSum As Double

    Dim NegSum As Double

    On Error GoTo eFunction

    rCount = TheValues.Rows.Count

    cCount = TheValues.Columns.Count

    PosSum = 0

    NegSum = 0

    MinDate = TheDates.Offset(0, 0).Resize(1, 1).Value

    If rCount > cCount Then

        MaxDate = TheDates.Offset(rCount - 1, 0).Resize(1, 1).Value

        For rCounter = 0 To rCount - 1

            TheVal = TheValues.Offset(rCounter, 0).Resize(1, 1).Value

            TheDate = TheDates.Offset(rCounter, 0).Resize(1, 1).Value

            If TheVal < 0 Then

                NegSum = NegSum + TheVal / ((1 + iRate) ^ ((TheDate - MinDate) / daybase))

            Else

                PosSum = PosSum + TheVal * ((1 + rRate) ^ ((MaxDate - TheDate) / daybase))

            End If

        Next

    Else

        MaxDate = TheDates.Offset(0, cCount - 1).Resize(1, 1).Value

        For cCounter = 0 To cCount - 1

            TheVal = TheValues.Offset(0, cCounter).Resize(1, 1).Value

            TheDate = TheDates.Offset(0, cCounter).Resize(1, 1).Value

            If TheVal < 0 Then

                NegSum = NegSum + TheVal / ((1 + iRate) ^ ((TheDate - MinDate) / daybase))

            Else

                PosSum = PosSum + TheVal * ((1 + rRate) ^ ((MaxDate - TheDate) / daybase))

            End If

        Next

    End If

    XMIRR = ((PosSum / NegSum * -1) ^ (1 / ((MaxDate - MinDate) / daybase))) - 1

    Exit Function

eFunction:

    XMIRR = CVErr(2015)

End Function

by Laurent Longre

This VB function returns the same result as Excel's WEEKNUM function.

Function WKNUM(D As Date) As Long

  D = Int(D)

  WKNUM = DateSerial(Year(D + (8 - WeekDay(D)) Mod 7 - 3), 1, 1)

  WKNUM = ((D - WKNUM - 3 + (WeekDay(WKNUM) + 1) Mod 7)) \ 7 + 1

End Function 

**********************************

VBA CODE EXAMPLES 

by Jim Rech

This procedure changes the font size in all cell comments on a worksheet.

Sub ChgAllComments()

    Dim Cell As Range

    For Each Cell In Cells.SpecialCells(xlCellTypeComments)

        Cell.Comment.Shape.TextFrame.Characters.Font.Size = 9

    Next

End Sub

by Stephen Bullen

This procedure open shows Excel's DataForm in New Record mode.

Sub ShowDataFormWithNewRecord()

   'Send a keystroke 

   SendKeys "+{TAB 6} "

   'This is the same as ActiveSheet.ShowDataForm, but without the International Issues

   CommandBars.FindControl(Id:=860).Execute

End Sub

**********************************

EXCEL DEVELOPER'S TIPS

Pull in correct values from Internet:

When importing stock data into a sheet using a web query, fractional stock prices less than 1 may be interpreted by Excel as dates. However, selecting Tools, Options, Transition and then clicking "Transition formula entry" coerces Excel into accepting the desired value.

Use class modules from another project:

By John Green

You can create an instance of a class in another project by creating a function in the referenced project containing the class module. In the project containing the class module, include something like the following code, in a standard module:

Function GetClass() As Class1

  Set GetClass = New Class1

End Function

In the project that wants to access the class, use something like the following code:

Dim cls As Object

Sub Test()

  Set cls = GetClass()

End Sub

**********************************

Top Excel Sites

For a list of fixed problems in Microsoft Office 2000 Service Release 1 (SR-1), go to:

http://support.microsoft.com/support/kb/articles/q245/0/21.ASP

**********************************

WORKSHEET FORMULA TIPS

by Bernie Deitrick

Question: I have a formula

=COUNTIF('Sheet1'!Ddd2346, "=0")

in a cell, where Ddd2346 refers to a named range. What I would like to do is to have an easy way to copy this formula down a column of cells, and have the Ddd2346 number increment by one each time.  The next cell needs to be Ddd2347, then Ddd2348 etc.
Answer: In this specific case, use:

=COUNTIF(INDIRECT("Ddd"&2345+ROW(A1)),"=0")

When this formula is filled down the column, the numeric suffixes of the named ranges increment by one.

**********************************

POWER FORMULA/FUNCTION TECHNIQUES

by George Simms

Question: I have the following problem:

In cells A1:E1:   1, 2, 3, 4, 5

In cells A2:E2:   6, 7, 8, 9, 10

In cells A5:E5:   21, 22, 23, 24, 25

Now I want to total diagonally (A1, B2, C3, D4 and E5) and I do that by the formula:

=SUM((ROW(A1:A5)=COLUMN(A1:E1))*(A1:E5)) - array entered.

This works fine (sum =65), but I wondered how can I total diagonally "the other way" (here A5, B4, C3, D2 and E1)?

Answer: To sum A5:E1 diagonally, use: =SUM(N(OFFSET(E1,ROW(1:5)-1,-1*ROW(1:5)+1)))

Or (not Array Entered):

=SUMPRODUCT(N(OFFSET(E1,ROW(INDIRECT("1:5"))-1,-1*ROW(INDIRECT("1:5"))+1)))

It would be better to use ROW(INDIRECT("1:5") in all the formulas, like

=SUM(N(OFFSET(E1,ROW(INDIRECT("1:5"))-1,-1*ROW(INDIRECT("1:5"))+1)))

as inserting a row above row 5 will change the reference. To sum across sheets (Sheet1!A1  Sheet2!B2  Sheet3!C3..etc) use:

=SUM(N(INDIRECT("Sheet"&ROW(1:5)&"!"&ADDRESS(ROW(1:5),ROW(1:5)))))

Or (not Array Entered):

=SUMPRODUCT(N(INDIRECT("Sheet"&ROW(1:5)&"!"&ADDRESS(ROW(1:5),ROW(1:5)))))

**********************************

VBA CODE EXAMPLES 

by Tom Ogilvy and Dana DeLouis

Question: I'd like to expand the Custom Autofilter to 3 or more entries.

This procedure assumes you want to display cells that have a,b, and c in the Cell in column A. If you are looking for multiple "Or" conditions, then use Union instead of Intersect.

Sub MultCustomAutoFilter()

  Dim rng1 As Range

  Dim rng2 As Range

  Dim rngAll3 As Range

  Range("A1").AutoFilter

  With [_FilterDatabase].Offset(1)

    Range("A1").AutoFilter Field:=1, Criteria1:="*a*", Operator:=xlAnd, Criteria2:="*b*"

    Set rng1 = .SpecialCells(xlVisible)

    Range("A1").AutoFilter Field:=1, Criteria1:="*c*"

    Set rng2 = .SpecialCells(xlVisible)

    ActiveSheet.AutoFilterMode = False

    Set rngAll3 = Application.Intersect(rng1, rng2)

    .EntireRow.Hidden = True

    rngAll3.EntireRow.Hidden = False

  End With

End Sub

This procedure works for Excel 2000.  For prior version change the range object in the With statement to: ActiveSheet.AutoFilter.Range.Offset(1, 0)

by Bill Manville

Question: Is there a way to reliably code the show detail and hide detail commands off of the "Data" menu into a VB macro?

Answer: This feature is not well supported by VBA. Thus, to hide the detail for the block within which the cursor sits, use:

ExecuteExcel4Macro "SHOW.DETAIL(1," & ActiveCell.Row & ",FALSE)"

Answer: by Stephen Bullen

Question: How can I change the name of the vbcomponent based on the name of the related worksheet?
Answer: The CodeName of a sheet can be changed with:

 Sub RenameCodeName(oWks As Sheet, sNewName As String)

 oWks.Parent.VBProject.vbComponents(oWks.CodeName).Properties("_CodeName") = sNewName

 End Sub

**********************************

POWER PROGRAMMING TECHNIQUE

by Bill Manville

Question: I have to check the contents of a large spreadsheet against a second more up to date spreadsheet in another workbook.  The structure of the 2 workbooks is the same. How can I identify which cells differ so I can investigate those individually.

Answer: This procedure creates a new workbook which lists the comparison results for each worksheet in the two workbooks of interest. Each of the two workbooks should be open prior to running this procedure. Replace the dummy names in the the DoCompare sub with appropriate filenames.

Sub DoCompare()

  Dim WS As Worksheet

  Workbooks.Add

  For Each WS In WorkBooks("SomeBook.xls").Worksheets

    CompareSheets WS, Workbooks("SomeOther.xls").Worksheets(WS.Name)

  Next

End Sub

Sub CompareSheets(WS1 As Worksheet, WS2 As Worksheet)

  Dim iRow As Integer, iCol As Integer

  Dim R1 As Range, R2 As Range

  Worksheets.Add.Name = WS1.Name ' new book for the results

  Range("A1:D1").Value = Array("Address", "Difference", WS1.Parent.Name, WS2.Parent.Name)

  Range("A2").Select

  For iRow = 1 To Application.Max(WS1.Range("A1").SpecialCells(xlLastCell).Row, _

                      WS2.Range("A1").SpecialCells(xlLastCell).Row)

    For iCol = 1 To Application.Max(WS1.Range("A1").SpecialCells(xlLastCell).Column, _

                      WS2.Range("A1").SpecialCells(xlLastCell).Column)

      Set R1 = WS1.Cells(iRow, iCol)

      Set R2 = WS2.Cells(iRow, iCol)

      ' compare the types to avoid getting VBA type mismatch errors.

      If TypeName(R1.Value) <> TypeName(R2.Value) Then

        NoteError R1.Address, "Type", R1.Value, R2.Value

      ElseIf R1.Value <> R2.Value Then

        If TypeName(R1.Value) = "Double" Then

          If Abs(R1.Value - R2.Value) > R1.Value * 10 ^ (-12) Then NoteError R1.Address, "Double", R1.Value, R2.Value

        Else

            NoteError R1.Address, "Value", R1.Value, R2.Value

        End If

      End If

      ' record formulae without leading "=" to avoid them being evaluated

      If R1.HasFormula Then

        If R2.HasFormula Then

          If R1.Formula <> R2.Formula Then NoteError R1.Address, "Formula", Mid(R1.Formula, 2), Mid(R2.Formula, 2)

        Else

          NoteError R1.Address, "Formula", Mid(R1.Formula, 2), "**no formula**"

        End If

      Else

        If R2.HasFormula Then

          NoteError R1.Address, "Formula", "**no formula**", Mid(R2.Formula, 2)

        End If

      End If

      If R1.NumberFormat <> R2.NumberFormat Then

        NoteError R1.Address, "NumberFormat", R1.NumberFormat, R2.NumberFormat

      End If

    Next iCol

  Next iRow

  With ActiveSheet.UsedRange.Columns

    .AutoFit

    .HorizontalAlignment = xlLeft

  End With

End Sub

Sub NoteError(Address As String, What As String, V1, V2)

  ActiveCell.Resize(1, 4).Value = Array(Address, What, V1, V2)

  ActiveCell.Offset(1, 0).Select

  If ActiveCell.Row = Rows.Count Then

    MsgBox "Too many differences", vbExclamation

    End

  End If

End Sub

**********************************

EXCEL TIPS

by John Green

Question: I need a simple macro to take the cell text in a selected cell(s) and add characters such as "." after the text until the cell is filled to its width. You get something like:

Text_here.....

and................

here..............

Answer: All you need to do is apply a custom format (Format|Cells - Number) to the cell such as:


@*.

@ is a place marker for the text and the character after the asterisk is repeated to fill the cell. "*.@" fills to the left. 

If you want to do this in code:


Range("A1:A10").NumberFormat = "@*."

Note: If this is done with cells containing numbers, they cannot be operated on as numbers since they are formatted as text. As a workaround to this problem, use:

=SUM(VALUE(range))  ' array-entered, instead of =SUM(range)

by George Simms

Question: The problem I have is that in order to add a value to an existing value field I need to insert an "=" at the beginning of the cell before it displays the solution. If I simply type the "+" or "-" after the value Excel displays

the formula (obviously interpreting it as text). Can I perform this task without inserting the "=" each time?

Answer: There is a way to do what you want, if you use it to only edit your data. From the menu > Tools > Options > Transition tab > Check the "Transition formula entry" box. It is recommend that once you have edited the data, go back and uncheck the box.  If left checked this can produce some odd results, entering dates etc.....

**********************************

Top Excel Sites

See http://www.officevba.com for VBA articles and downloadables files of the highest order.

**********************************

POWER FORMULA TECHNIQUES

by Leo Heuser and Eero Tibar

Question: How can you get a list of unique entries in an n * m array by using a worksheet formula?

Answer: Example: Assuming data in B5 : GR10, enter this array formula in e.g. G12. G11 must be empty or, if it has a value, this value must not occur in B5 : GR10.

=OFFSET($B$5,MIN(IF(COUNTIF($G$11:G11,$B$5:$GR$10)=0,ROW($B$5:$GR$10)-

ROW($B$5))),MOD(MIN(IF(COUNTIF($G$11:G11,$B$5:$GR$10)=0,ROW($B$5:$GR$10)-

ROW($B$5) +(COLUMN($B$5:$GR$10)-COLUMN($B$5))/1000)),1)*1000)

Drag down until the value in G12 begins repeating.

Here is slightly different approach to extract unique items from a N*M table (named as "tbl" in the formula). Type "Unique items from the table" in A1 and enter the following formula as an array into A2 and copy it down.

=INDEX(tbl,MIN(IF(COUNTIF($A$1:A1,tbl)=0,ROW(tbl)-MIN(ROW(tbl))+1)),

MATCH(0,COUNTIF($A$1:A1,INDEX(tbl,MIN(IF(COUNTIF($A$1:A1,tbl)=0,ROW(tbl)

-MIN(ROW(tbl))+1)),,1)),0),1) 

**********************************

VBA CODE EXAMPLES 

by David Hager

Question: I like using the Pick List feature in Excel 97 (and later), but I don't like having to select the menu item for that feature everytime I go to a new cell. How can I solve this problem?

Answer: Place this event procedure in the ThisWorkbook module. Then, any time you select a cell where the pick list would pop up when called from a menu item, it will instead pop up automatically.

Private Sub Workbook_SheetSelectionChange(ByVal Sh As Object, ByVal Target As Excel.Range)

    On Error Resume Next

    With Target

        If .Value <> "" Then Exit Sub

        If .Row = 1 Or .Row = Rows.Count Then Exit Sub

        If .Offset(-1).Value = "" And .Offset(1).Value = "" Then Exit Sub

        If Selection.Address <> .Address Then Exit Sub

    End With

    Application.EnableEvents = False

    SendKeys "+{F10}k"

    Application.EnableEvents = True

End Sub

by Bob Umlas

Question: How can you give users a Print Preview and not allow them to change any of the settings?
Answer: Use the following procedure. It disables key buttons at the top of the preview window.

Sub PrtPvw()

    ActiveSheet.PrintPreview False
'"False"==> no changes allowed

    ActiveWindow.View = xlNormalView
' In case user switched to Page Brake Preview.

End Sub

by Bob Umlas

Question: How can I step through a For-Next loop by using a custom step process?


Answer: If you need to loop through code with an index which takes on specific values like 1,4,5,18,28,33,34,85 instead of the more commom 1,2,3,4,5,6.. or 1,3,5,7,..., then you can use this technique:

Sub OddLoop()

   For i = 1 to 8

       j=Array(1,4,5,18,28,33,34,85)(i)

       'Now use j as your subscript

   Next

End Sub

**********************************

POWER PROGRAMMING TECHNIQUE

by Bob Umlas

This procedure contains VBA code to add to your existing VBA code. Basically, it puts one statement at the beginning of each procedure in every module (class modules and event procedures not included). This statement is

a call to a routine (which YOU need to write) and passes the sub/function name. For example:
Before:

Sub ABC()

   Dim i as Integer

   For each x in sheets

   Next

End Sub

Sub xyz()

End Sub

After:

Sub ABC()

MyProc "ABC"

   Dim i as Integer

   For each x in sheets

   Next

End Sub

Sub xyz()

MyProc "xyz"

End Sub

Notice that afterwards, there's a new line immediately after the sub. It calls MyProc (this is changeable) and passes the name of the procedure it's in. You can use MyProc to trace flow, track the time, etc -- you can get creative here!

There are 2 main routines: Addit, and Deleteit. Running Addit will insert the one-liner, running Deleteit will remove this one-liner. The code is inserted into the active workbook.

The first line inside the VBE for AddALine.xls is:

Public Const TheProcName As String = "MyProc" '==CHANGE THIS LINE

and whatever you change "MyProc" to will be the routine called inside each procedure of your code.

Public Const TheProcName As String = "MyProc" '==CHANGE THIS LINE

Sub Addit()

'=====
'RUN THIS CODE TO INSERT THE LINE INTO THE ACTIVE WORKBOOK's CODE

'=====

    AddALine

    MsgBox "Done....Don't forget to write procedure " & TheProcName & "!", vbExclamation

End Sub

Sub Deleteit()

'=====
'RUN THIS CODE TO DELETE THE LINE

'=====

    DelALine

    MsgBox TheProcName & " has been deleted from each procedure."

End Sub

Sub AddALine()

Dim ProcName As String, ProcNames() As String, Boo As Boolean

Dim LngR As Long, TheLine As Long, LngI As Long

    Set VBP = ActiveWorkbook.VBProject

    nocomponents = VBP.VBComponents.Count

    On Error Resume Next

    For i = 1 To nocomponents

        If VBP.VBComponents(i).Type = 1 Then 'module

            With VBP.VBComponents(i).CodeModule

                If .Name = "ModInserter" Then GoTo NextOne

                col = .CountOfLines

                codl = .CountOfDeclarationLines

                ProcName = .ProcOfLine(codl + 1, LngR)

                If ProcName = "" Then GoTo NextOne

                If LngR <> 0 Then GoTo NextOne

                TheLine = .ProcBodyLine(ProcName, vbext_pk_Proc)

                thetext = .Lines(TheLine, 1)

                If Right(thetext, 1) = "_" Then j = 2 Else j = 1

                .InsertLines TheLine + j, TheProcName & """" & ProcName & """"

                LngI = codl + 1

                col = col + 1

2:

                If LngI > col Then GoTo 1

                If ProcName <> .ProcOfLine(LngI, LngR) Then

                    ProcName = .ProcOfLine(LngI, LngR)

                    If LngR <> 0 Then GoTo 3

                    TheLine = .ProcBodyLine(ProcName, vbext_pk_Proc)

                    thetext = .Lines(TheLine, 1)

                    If Right(thetext, 1) = "_" Then j = 2 Else j = 1

                    .InsertLines TheLine + j, TheProcName & """" & ProcName & """"

                    col = col + 1

                End If

3:

                LngI = LngI + 1

                GoTo 2

1:             

            End With

        End If

NextOne:

    Next

End Sub

Sub DelALine()

Dim ProcName As String, ProcNames() As String, Boo As Boolean

Dim LngR As Long, TheLine As Integer, LngI As Integer

    If MsgBox("Are you sure you want to delete " & TheProcName & " from each procedure?", vbYesNo + vbQuestion) = vbNo Then Exit Sub

    Set VBP = ActiveWorkbook.VBProject

    nocomponents = VBP.VBComponents.Count

    On Error Resume Next

    For i = 1 To nocomponents

        If VBP.VBComponents(i).Type = 1 Then 'module

            With VBP.VBComponents(i).CodeModule

                If .Name = "ModInserter" Then GoTo NextOne

                col = .CountOfLines

                codl = .CountOfDeclarationLines

                ProcName = .ProcOfLine(codl + 1, LngR)

                If ProcName = "" Then GoTo NextOne

                If LngR <> 0 Then GoTo NextOne

                TheLine = .ProcBodyLine(ProcName, vbext_pk_Proc)

                thetext = .Lines(TheLine, 1)

                If Right(thetext, 1) = "_" Then j = 2 Else j = 1

                If Left(.Lines(TheLine + j, 1), 5) <> Left(TheProcName, 5) Then

'                    MsgBox TheProcName & " not found in procedure """ & ProcName & """... ignoring"

                    GoTo 22

                End If

                .DeleteLines TheLine + j, 1

22:

                LngI = codl + 1

2:

                If LngI > col Then GoTo 1

                If ProcName <> .ProcOfLine(LngI, LngR) Then

                    ProcName = .ProcOfLine(LngI, LngR)

                    If LngR <> 0 Then GoTo 3

                    If ProcName = "" Then GoTo 3

                    TheLine = .ProcBodyLine(ProcName, vbext_pk_Proc)

                    thetext = .Lines(TheLine, 1)

                    If Right(thetext, 1) = "_" Then j = 2 Else j = 1

                    If Left(.Lines(TheLine + j, 1), 5) <> Left(TheProcName, 5) Then

'                        MsgBox TheProcName & " not found in procedure """ & ProcName & """... ignoring"

                        GoTo 3

                    End If

                    .DeleteLines TheLine + j, 1

                End If

3:

                LngI = LngI + 1

                GoTo 2

1:                

            End With

        End If

NextOne:

    Next

End Sub

Sub Showcode()

    MsgBox "Before running ""Addit"", activate the workbook whose code " & "this routine will update."

    MsgBox "Change ""TheProcName"" to the name of the procedure you want " & "to run for each sub."

    Application.SendKeys "{up}{up}"

    Application.Goto "Addit"

End Sub

**********************************

DEVELOPER TIPS

by Chip Pearson and Stephen Bullen

Question: Why use class modules?

Answer: (Chip Pearson) Basically, a Class is the definition of an Object. The word "object" is deliberately vague. And object is anything that you want to design. It is defined entirely (mostly) by its properties, methods, and events. In Excel, there are hundreds of "built-in" objects, all defined by class modules. The "class" is the definition of an "object". For example, a Worksheet is an object. And there is a class module which defines just what

a Worksheet really is. There are various properties of a Worksheet object (e.g., Visible). Properties simply define and set various attributes. Think of properties as "Adjectives" which describe an object. An object also has Methods. Methods are the "Verbs" of objects. For example, a Worksheet object has an Activate method. This causes something to happen. Finally there are Events. I can't think of a good grammatical analogy for events. Essentially, Events are how an object tells the rest of the world that something happened. For example, in a Worksheet object, there is a Change event. This is the Worksheet object's way of telling the rest of world, "Hey, look at me, I changed". The rest of the world can ignore that event, or it may take action. But the world has been told that object has done something (or had done something to it).

Now, you use Class Modules to create your own objects. Suppose you were writing an application that was used for employee tracking. Using a class module, you would define your own object called "Employee". This class

would define a single, generic, employee. With the DIM and SET statement, you can create a specific employee, based on the "template" or "definition" of a generic employee. The Employee class would have several Properties,

such as Name, Address, and Salary. It could also have methods, such as Promote, GiveRaise, and Fire. In your application, the Promote method would do the same things -- e.g., increasing the Salary property, updating a

central database, sending an email to another department to buy him a nicer computer, etc. These actions are all the same whenever you Promote any employee. By using a Class Module to define a "generic" employee, you only have to write the code once. Then to work with a *specific* employee, you just call the methods for that employee:

Dim ThisEmp As CEmployee

' more code

Set ThisEmp = New CEmployee

ThisEmp.Name = "John Smith"

' more code

ThisEmp.Promote

All of the code related to the Promote event is contained in the Class modules (the definition of any employee), so you can simply call the Promote method. Once you've defined the Class, you never have to worry about what

Promote actually does.

Here's another way to think about it. In the Worksheet object, there is a PrintOut method. Within the PrintOut method, there is all the code that actually formats the worksheet for printing, determines what printer you have, and actually does all the work of printing the sheet. As a VBA programmer, you don't have to worry about any of that. You simply call PrintOut, and let that do all the work for you. You don't have to worry about what sort of printer the user has, whether it can print color, and a hundred other things. You just call PrintOut and let the Object do all the work.

Class Modules let you create you own objects, or extend the functionality of other, existing objects. They are very useful because they allow you to write the code once, and then simply create new objects based on the class (think of it like a blueprint for a house). It is write the code once, and use it many times. For example, I have a class module that extends the functionality of a standard list box. The standard list box doesn't have a MoveUp method,

which simply moves the selected item one row up in the list. By using a Class Module, I added a MoveUp method (as well as MoveDown, MoveToTop, MoveToBottom, etc). I wrote that class one time. Now, whenever I need to

use "better" list boxes in my applications, I just use that Class. I don't have to "re-invent the wheel" for every application I write.

This just scratches the surface of what a Class is and how to use them. If you've ever heard the term "object oriented", Classes are the foundation of this entire design philosophy.

Answer (Stephen Bullen)
Just to provide the opposite end of the spectrum to Chip's excellent answer, class modules can also be though of as user-defined types (UDT) on steroids.  A simple UDT can be used to store related information about a particular thing, such as Chip's employee:

Type Employee

     Name As String

     DOB As String

     Grade As String

     Salary As Double

End Type

If you wanted to do stuff with an employee, you'd use a normal procedure somewhere:

Sub RaiseEmployee(uEmp As Employee, sNewGrade As String)

     'Validate Grade

     '...

     uEmp.Grade = sNew Grade

     'Do stuff to work out new salary etc.

     '...

End Sub

Sub FireEmployee(uEmp As Employee)

     uEmp.Grade = "F"

     uEmp.Salary = 0

End Sub

Sub SetSalary(uEmp As Employee, dNewSalary As Double)

     'Validate Salary

     '...

     'Does new salary mean a new grade?

     '...

End Sub     

etc.

That's fine as far as it goes and you can create some great programs without ever using class modules.  The main thing wrong with it is that the *data* for the object (i.e. the contents of your UDT) is separated from the *actions* that are performed on the data (the RaiseEmployee and FireEmployee subs).  Hence, you have to be very careful that the same validation is performed in each sub and that one sub doesn't alter the data in a way that will cause another sub to fail; this is often the cause of some of the hardest bugs to find - logic problems.

If you use a class module instead, you can include the validation and other functionality *with* the data; to the extent that the data can *not* be changed unless it's valid.  You can think of it as that the 'Grade' property of the Employee (for example) can validate *itself* and can refuse to be updated, or it can know *itself* that when it changes to a valid new grade, it needs to change the salary too.

In the example above, with two simple procedures, think what would happen if we had to add another check before changing the grade, or introduce a new action to be performed if the grade is changed (such as notifying their 

manager).  In the procedural approach, we'd have to change two or three routines to handle it - i.e. wherever the grade is set.  In the clas module approach, it is simply another action to be performed by the 'grade' property *itself* - none of the other code needs to know about it.

i.e:

Dim msGrade As String    'Data that only code in the class can 'see'

'Property to read the grade

Public Property Get Grade() As String: Grade = msGrade: End property

'Property to set the grade

Public Property Let Grade(sNew As String)

     If Not sNew Is Valid Then

         Err.Raise "Not a valid grade"

         Exit Property

     End If

     'Grade is valid, so we can safely store it

     msGrade = sNew

     'Now what else do we need to do when the grade changes?

     Select Case sNew

     Case "F"

         'Being fired, better ask for a redundancy slip

         Salary = 0

     Case "M"

         'Being made a manager, better ask for a better car

         'Increase the Salary too

     Case "D"

         'Being demoted, schedule for more frequent reviews

         'Decrease the Salary

     'etc

     End Select

End Property

Public Sub Fire()

     Grade = "F"

End Sub

Now, everything that needs to be done when the grade is changed has been 

made an *integral* part of changing the grade - there's no way that the 

grade can be changed by anywhere else in the system without those checks 

and actions happening.

Really, though, it just boils down to a different design and development style, and one that hopefully takes us further down the road of improved code reuse, more stability and fewer opportunities for bugs to creep in.

The hardest thing to work out, though, is to decide which functionality should be 'in' the class module and which should be on the outside, but using the class module.  For example, do we have a '.Fire' method within the class, or a Fire(oEmp As Employee) procedure outside that just sets the grade to "F"?

I find that I'm using class modules more and more; it's almost at the stage where if I'm asked "Why use a class module", my reply is "Why not?"

Welcome to the 20th issue of the Excel Experts E-letter (or EEE), by David Hager. EEE used to be a monthly publication. It's been a long time since the last issue, and I cannot say when the next issue will be. Feel free to distribute copies of EEE to your friends and colleagues and to contribute your Excel gems to EEE so that others can benefit from your work. Due to problems associated with distribution lists, I cannot mail EEE directly to individuals anymore.
**********************************

Top Excel Sites

See:

http://home.pacbell.net/beban

for a great collection of array UDFs.

**********************************

POWER FORMULA TECHNIQUES

by David Hager

How can I find the count of unique items in a filtered column?
Define a column range in your table (excluding header) as Rge.

Define unRge as:

=IF(SUBTOTAL(3,OFFSET(Rge,ROW(Rge)-MIN(ROW(Rge)),,1)),Rge,"")

Then, the array formula to return the # of unique occurrences in a filtered column is:

=SUM(N(IF(ISNA(MATCH("",unRge,0)),MATCH(Rge,Rge,0),IF(MATCH(unRge,unRge,0)

=MATCH("",unRge,0),0,MATCH(unRge,unRge,0)))=ROW(Rge)-MIN(ROW(Rge))+1))

by Tom Ogilvy

How can I set validation so no spaces are allowed?
Select A1:C20 with A1 as the active cell in the selection.

Pick Data > Validation from the menu and select the custom option.

Use the following formula:

=LEN(A1)=LEN(SUBSTITUTE(A1," ",""))

Since you are using relative cell references, the validation formula will adjust to address each of the cells in the selection.

by John Walkenbach and John Green

How can I locate cells containing formulas with literal values?
Use the following UDF as your conditional formatting formula.

Function CellUsesLiteralValue(Cell As Range) As Boolean

    If Not Cell.HasFormula Then

        CellUsesLiteralValue = False

    Else

        CellUsesLiteralValue = Cell.Formula Like "*[=^/*+-/()><, ]#*"

    End If

End Function

It accepts a single cell as an argument. It returns True if the cell's formula contains an operator followed by a numerical digit. In other words, it identifies cells that have a formula which contains a literal numeric value. You can test each cell in the range, and highlight it if the function returns True.

by George Simms

If the NETWORKDAYS function (found in the Analysis Toolpak) cannot be used, is there a formula that will perform the same function?
If the Start date is in A1 and the End date is in B1, then use:

=(INT(B1/7)-INT(A1/7))*5+MAX(0,MOD(B1,7)-1)-MAX(0,MOD(A1,7)-2)

**********************************

VBA CODE EXAMPLES 

by Bill Manville

The objective is to prevent people cutting/copying and pasting when your workbook is open.
Run DisableCutAndPaste from a suitable event procedure (e.g. Workbook_Open or Worksheet_Activate) and EnableCutAndPaste from another (e.g. Workbook_Close or Worksheet_Deactivate).  

Sub DisableCutAndPaste()

  EnableControl 21, False   ' cut

  EnableControl 19, False   ' copy

  EnableControl 22, False   ' paste

  EnableControl 755, False  ' pastespecial

  Application.OnKey "^c", ""

  Application.OnKey "^v", ""

  Application.OnKey "+{DEL}", ""

  Application.OnKey "+{INSERT}", ""

  Application.CellDragAndDrop = False

End Sub

Sub EnableCutAndPaste()

  EnableControl 21, True   ' cut

  EnableControl 19, True   ' copy

  EnableControl 22, True   ' paste

  EnableControl 755, True  ' pastespecial

  Application.OnKey "^c"

  Application.OnKey "^v"

  Application.OnKey "+{DEL}"

  Application.OnKey "+{INSERT}"

  Application.CellDragAndDrop = True

End Sub

Sub EnableControl(Id As Integer, Enabled As Boolean)

  Dim CB As CommandBar

  Dim C As CommandBarControl

  For Each CB In Application.CommandBars

    Set C = CB.FindControl(Id:=Id, recursive:=True)

    If Not C Is Nothing Then C.Enabled = Enabled

  Next

End Sub

by Chip Pearson

Is is possible to disable certain menu items on both the toolbar and the right-click pop-up that wil prevent the user from either deleteing/renaming, a sheet without protecting the entire workbook structure?
You can disable them with:

Dim Ctrl As Office.CommandBarControl

For Each Ctrl In Application.CommandBars.FindControls(ID:=847)

    Ctrl.Enabled = False

Next Ctrl

For Each Ctrl In Application.CommandBars.FindControls(ID:=889)

    Ctrl.Enabled = False

Next Ctrl

by Chip Pearson

How can I search through all the cell formulas on a worksheet and find out the cells that reference a specific named range?
Use the following procedure:

Dim Rng As Range

Dim NameRange As Range

Set NameRange = ActiveWorkbook.Names("TheName").RefersToRange

On Error Resume Next

For Each Rng In ActiveSheet.UsedRange.SpecialCells(xlCellTypeFormulas)

    Err.Clear

    If Not Application.Intersect(Rng.DirectPrecedents, NameRange) Is Nothing

Then

        If Err.Number = 0 Then Debug.Print "Cell: " & Rng.Address & " refers to " & NameRange.Address

    End If

Next Rng

**********************************

POWER PROGRAMMING TECHNIQUE

by Jim Rech

Can I change the Excel logo to something else?
This code shows you how to change the Excel icon:

Declare Function GetActiveWindow32 Lib "USER32" Alias "GetActiveWindow" () As Integer

Declare Function SendMessage32 Lib "USER32" Alias "SendMessageA" (ByVal hWnd As Long, ByVal wMsg As Long, ByVal wParam As Long, ByVal lParam As Long) As Long

Declare Function ExtractIcon32 Lib "SHELL32.DLL" Alias "ExtractIconA" (ByVal hInst As Long, ByVal lpszExeFileName As String, ByVal nIconIndex As Long) As Long

Sub ChangeXLIcon()

    Dim h32NewIcon As Long

    Dim h32WndXLMAIN As Long

    h32NewIcon = ExtractIcon32(0, "Notepad.exe", 0)

    h32WndXLMAIN = GetActiveWindow32()

    SendMessage32 h32WndXLMAIN, &H80, 1, h32NewIcon 'Icon big

    SendMessage32 h32WndXLMAIN, &H80, 0, h32NewIcon 'Icon small

End Sub

by Leo Heuser

I would like to create a Excel template which when you open a document from it, it assigns a unique sequential number to the new document. Is there a way of doing this?

Below find two routines to do, what you want. They are both inserted in "ThisWorkbook" (<Alt><F11>) for the template and are fired, when a new invoice is created. The first one saves the current invoice number to the registry, and can be used, if you are the sole user of the system. The second solution saves the number in an INI-file, which you can place, where you please. This solution is useful, if more persons are using the invoice system.

Private Sub Workbook_Open()

'From the template, in the VBA editor, set a reference to Microsoft Visual Basic for Applications Extensibility 5.3 in the menu Tools

Dim WorksheetName As String

Dim WorksheetCell As String

Dim SettingName As String

Dim lLine As Long

Dim InvoiceNumber As Variant

Dim InvoiceNumberCell As Object

Dim TemplateName As String

    TemplateName = "John.xlt"

    WorksheetName = "Invoice"

    WorksheetCell = "F7"

    SettingName = "John"

    Set InvoiceNumberCell = Worksheets(WorksheetName).Range(WorksheetCell)

    If UCase(ActiveWorkbook.Name) = UCase(TemplateName) Then GoTo Finito

    InvoiceNumber = GetSetting(SettingName, WorksheetName, "InvoiceNumber")

    If InvoiceNumber = "" Then

        InvoiceNumber = 1

    Else

        InvoiceNumber = InvoiceNumber + 1

    End If

    SaveSetting SettingName, WorksheetName, "InvoiceNumber", InvoiceNumber

    InvoiceNumberCell.Value = InvoiceNumber

    With

ActiveWorkbook.VBProject.VBComponents(ActiveWorkbook.CodeName).CodeModule

        lLine = .ProcBodyLine("Workbook_Open", vbext_pk_Proc)

        .InsertLines lLine + 1, "Exit Sub"

    End With

Finito:

Set InvoiceNumberCell = Nothing

End Sub

Private Declare Function GetPrivateProfileString Lib "kernel32" Alias "GetPrivateProfileStringA" (ByVal lpApplicationName As String, ByVal lpKeyName As String, ByVal lpDefault As String, ByVal lpReturnedString As String, ByVal nSize As Long, ByVal lpFileName As String) As Long

Private Declare Function WritePrivateProfileString Lib "kernel32" Alias "WritePrivateProfileStringA" (ByVal lpApplicationName As String, ByVal lpKeyName As String, ByVal lpString As Any, ByVal lpFileName As String) As Long

Private Sub Workbook_Open()

'From the template, in the VBA editor, set a reference to Microsoft Visual Basic for Applications Extensibility 5.3 in the menu Tools

Dim WorksheetName As String

Dim WorksheetCell As String

Dim Section As String

Dim kKey As String

Dim lLine As Long

Dim InvoiceNumber As Long

Dim InvoiceNumberCell As Object

Dim TemplateName As String

Dim IniFileName As String

Dim Dummy As Variant

    TemplateName = "John2.xlt"

    WorksheetName = "Invoice"

    WorksheetCell = "F7"

    Section = "Invoice"

    kKey = "Number"

    IniFileName = "C:\Windows\Temp\InvoiceNumber.txt"

    Set InvoiceNumberCell = Worksheets(WorksheetName).Range(WorksheetCell)

    If UCase(ActiveWorkbook.Name) = UCase(TemplateName) Then GoTo Finito

    Dummy = GetString(Section, kKey, IniFileName)

    If Left(Dummy, 1) = Chr$(0) Then

        InvoiceNumber = 1

    Else

        InvoiceNumber = CLng(Dummy) + 1

    End If

    WritePrivateProfileString Section, kKey, CStr(InvoiceNumber),

IniFileName

    InvoiceNumberCell.Value = InvoiceNumber

    With

ActiveWorkbook.VBProject.VBComponents(ActiveWorkbook.CodeName).CodeModule

        lLine = .ProcBodyLine("Workbook_Open", vbext_pk_Proc)

        .InsertLines lLine + 1, "Exit Sub"

    End With

Finito:

Set InvoiceNumberCell = Nothing

End Sub

Function GetString(Section As String, Key As String, File As String) As String

    Dim KeyValue As String

    Dim Characters As Long

    KeyValue = String(255, 0)

    Characters = GetPrivateProfileString(Section, Key, "", KeyValue, 255, File)

    If Characters > 1 Then KeyValue = Left(KeyValue, Characters)

    GetString = KeyValue

End Function

by Jim Rech

Is there a way to delete all name ranges in a selection at one time?

Be careful to not break references to other formulas when using this procedure.

Sub Dename()

    Dim Cell As Range

    ActiveSheet.TransitionFormEntry = True

    For Each Cell In Selection.SpecialCells(xlFormulas)

        Cell.Formula = Cell.Formula

    Next

    ActiveSheet.TransitionFormEntry = False

End Sub

**********************************

DEVELOPER TIPS

by Chip Pearson

Notes on an interesting and useful debugging technique. Suppose you are developing some application, and you have some global variable such as:

Public  NumberOfUnits As Long

In your app, the only reasonable value for this is, say, between 1 and 100. For debugging purposes, you can "trap" your errors, when you assign an invalid value to this, as follows.

In your standard code module (NOTE: This does NOT have to be in a class module!) do the following:

Dim p_NumberOfUnits As Long

Property Get NumberOfUnits() As Long

    NumberOfUnits = p_NumberOfUnits

End Property

Property Let NumberOfUnits(Value As Long)

    If (Value >=1 ) And (Value <=100) Then

        p_NumberOfUnits = Value

    Else

        Err.Raise 5

    End If

End Property

Then, in the rest of your code, you'd access the variable in the normal way:

Sub AAA()

NumberOfUnits = 10

NumberOfUnits = 123

Msgbox "Units: " & NumberOfUnits

End Sub

These standard access methods will indeed take you through the get/let/set property procedures. And yes, standard code modules (BAS files) do support Property Get/Let/Set procedures.  You're code will blow up on the statement

NumberOfUnits = 123

(You must raise an error.  The specific error is, of course, you choice.)

Then, just use the View Call Stack to see where you called this from.  Of course, this adds some overhead, so in the production version of the code, you'd remove the Property Get/Let pair, and rename

Dim p_NumberOfUnits As Long

to

Dim NumberOfUnits As Long

Or, of course, you could do everything with conditional compilation. In the end, the really interesting thing is that you can use property get/let/set procedures in a standard code module, not just in a class module.

